Christopher Lum
lum@uw.edu

Lecture01d
Working with Matrices in Matlab and Mathematica

The YouTube videos that cover this lecture are:

'Working with Matrices in Matlab' located at https://youtu.be/ZBafH5fss1E?si=UZkpsUuXe_vty-9-

'Working with Matrices in Mathematica' located at https://y-outu.be/182m2cSyuUs?si=U82uz5F3mX018huv

---

Matrices

A matrix is a 2D array of numbers.

$$A = \begin{pmatrix} a_{11} & a_{12} & \dots & a_{1n} \\ a_{21} & a_{22} & \dots & a_{2n} \\ \vdots & \vdots & \ddots & \vdots \\ a_{m1} & a_{m2} & \dots & a_{mn} \end{pmatrix}$$

Note that the book uses a bold face to denote a matrix. This is done for emphasis and most literature simply uses capital letters to denote matrices. We will follow the nomenclature of using capital letters to denote matrices.

If a matrix only has a single row or a single column, it is called a vector. Typically, vectors are denoted using lower case letters and a bold or over bar to denote that it is a vector.

$\overline{v}$ = vector

Mathematica has the construct of a list which can be used to represent a 1 dimensional set of values such as a vector. A list can be created using the curly brace operators or the 'List' function

```
In[2]:= v = {1, -2, a}
        v = List[1, -2, a]
Out[2]= {1, -2, a}
Out[3]= {1, -2, a}
```

Note that this list does not have an "orientation" in the sense that we cannot specify whether it is a row

or column vector.

In order to specify orientation, we need to generate a full matrix. You can create matrices in Mathematica using lists of lists. Each inner list corresponds to a row of the matrix.

In[4]:=
```
A = {{a, b, c}, {d, e, f}}
```

Out[4]=
```
{{a, b, c}, {d, e, f}}
```

This is difficult to visualize, so we can ask Mathematica to display this in a MatrixForm

In[5]:=
```
A // MatrixForm
```

Out[5]//MatrixForm=

$$\begin{pmatrix} a & b & c \\ d & e & f \end{pmatrix}$$

Warning: Never assign a variable to a MatrixForm object. In other words, the follow is not advisable

In[6]:=
```
(*This is bad!  DO NOT DO THIS IN ACTUAL CODE*)
B = A // MatrixForm
```

Out[6]//MatrixForm=

$$\begin{pmatrix} a & b & c \\ d & e & f \end{pmatrix}$$

B is now stored in Mathematica as a MatrixForm object and we can no longer use it in calculations. For example, if we now try to perform $A + B$

In[7]:=
```
A + B
```

Out[7]=

$$\left\{\left\{a + \begin{pmatrix} a & b & c \\ d & e & f \end{pmatrix}, b + \begin{pmatrix} a & b & c \\ d & e & f \end{pmatrix}, c + \begin{pmatrix} a & b & c \\ d & e & f \end{pmatrix}\right\}, \left\{d + \begin{pmatrix} a & b & c \\ d & e & f \end{pmatrix}, e + \begin{pmatrix} a & b & c \\ d & e & f \end{pmatrix}, f + \begin{pmatrix} a & b & c \\ d & e & f \end{pmatrix}\right\}\right\}$$

This is clearly garbage.

In[8]:=
```
Clear[A, B]
```

An alternative method for defining matrices in Mathematica is to use the Basic Math Input Palette (Palettes > Other > Basic Math Input)



Clicking on this inserts a 2x2 matrix. You can add additional rows by clicking on an entry in the matrix

and pressing 'Ctrl + Enter'. You can add additional columns by clicking on an entry in the matrix and pressing 'Ctrl + ,'

Alternatively, you can quickly create matrices by

1. Type and open then closed parenthesis ()
2. Place cursor between the parenthesis
3. Press 'Ctrl + Enter' to create rows
4. Press 'Ctrl + ,' to create columns

Math Joke:
Question: How is Culture Club (Boy George) related to a 6 column Mathematica matrix?
Answer: You have to press 'Ctrl + , , , , , ' (chameleon)

Matrix Multiplication

We should note that matrix multiplication can mean different things in different contexts. In general, there are two types of matrix multiplication that we will typically utilize

1. Standard matrix multiplication
2. Matrix element-wise multiplication

Let us examine each of these cases in examples.

1. Standard Matrix Multiplication

Standard matrix multiplication multiplies a matrix A (of dimensions m – by – p) with a matrix B (of dimensions (p – by – n) to produce a matrix C (which will have dimensions m – by – n).

The entry in the i^{th} row and j^{th} column of C is obtained by taking the dot product of the entire i^{th} row of A with the entire j^{th} column of B.

Consider the example below which illustrates matrix multiplication by hand

$$\begin{pmatrix} 1 & 3 & 5 \\ 7 & 9 & 11 \end{pmatrix}\begin{pmatrix} 2 & 4 \\ 6 & 8 \\ 10 & 12 \end{pmatrix} = \begin{pmatrix} (1*2+3*6+5*10) & (1*4+3*8+5*12) \\ (7*2+9*6+11*10) & (7*4+9*8+11*12) \end{pmatrix}$$

$$= \begin{pmatrix} 70 & 88 \\ 178 & 232 \end{pmatrix}$$

Mathematica provides the . operator to perform matrix multiplication. Note: the * symbol should not be used for matrix multiplication in Mathematica as this means element-wise multiplication as we will show soon.

In[9]:=
```
(*Define two matrices*)
A = {{1, 2, 4}, {-3, 3, 4.5}};
B = {{1, 2, 3}, {1, 5, 4}, {-3, 1, -1}};

(*Multiply matrices using the Dot function*)
Print["Using the 'Dot' function"]
Dot[A, B] // MatrixForm

(*Multiply matrices using the . shorthand*)
Print["Multiply matrices using the . notation"]
A.B // MatrixForm

(*Clear these variables*)
Clear[A, B]
```

```
Using the 'Dot' function
```

Out[12]//MatrixForm=

$$\begin{pmatrix} -9. & 16. & 7. \\ -13.5 & 13.5 & -1.5 \end{pmatrix}$$

```
Multiply matrices using the . notation
```

Out[14]//MatrixForm=

$$\begin{pmatrix} -9. & 16. & 7. \\ -13.5 & 13.5 & -1.5 \end{pmatrix}$$

2. Matrix Element-wise Multiplication

Matrix element-wise multiplication multiplies a matrix A (of dimensions m – by – n) with a matrix B (of dimensions (m – by – n) to produce a matrix C (which will have dimensions m – by – n).

The entry in the i^{th} row and j^{th} column of C is obtained by simply multiplying the entry in the i^{th} row and j^{th} column of A with the entry in the i^{th} row and j^{th} column of B

In[16]:=
```
A = {{1, 2}, {3, 4}};
B = {{5, 6}, {7, 8}};
```

In Mathematica, we can perform element-wise multiplication operations using the * symbol (note that it is commutative)

In[18]:=
```
A * B // MatrixForm
B * A // MatrixForm
```

Out[18]//MatrixForm=

$$\begin{pmatrix} 5 & 12 \\ 21 & 32 \end{pmatrix}$$

Out[19]//MatrixForm=

$$\begin{pmatrix} 5 & 12 \\ 21 & 32 \end{pmatrix}$$

Similarly, in Mathematica, we can element-wise raise each element to the same power using the standard power notation (this is somewhat confusing as many textbooks will use this to denote matrix powers).

In[20]:=
```
A^2 // MatrixForm
A^3 // MatrixForm
```

Out[20]//MatrixForm=

$$\begin{pmatrix} 1 & 4 \\ 9 & 16 \end{pmatrix}$$

Out[21]//MatrixForm=

$$\begin{pmatrix} 1 & 8 \\ 27 & 64 \end{pmatrix}$$

Note that Matlab's behavior is exactly opposite in the sense that the A^2 means $A\,A$ (matrix multiplication). For example the code

```
%Define the matrix
A=  [1 2; 3 4]

%Perform matrix power
A^2
A^3
```

Produces output of

```
ans =


   7  10

  15  22




ans =


  37  54

  81 118
```

Note: The YouTube video on 'Getting Started with Matlab' (https://youtu.be/_M0mOHn0ink) is assigned as part of HW01

We need to be careful when we write expressions like A^3, this typically means $A\,A\,A$, so we need to perform matrix multiplication and cannot use the power symbol in Mathematica

In[22]:=
```
A.A.A // MatrixForm
```

Out[22]//MatrixForm=

$$\begin{pmatrix} 37 & 54 \\ 81 & 118 \end{pmatrix}$$

Alternatively, we can use the 'MatrixPower' function

In[23]:=
```
MatrixPower[A, 3] // MatrixForm
```

Out[23]//MatrixForm=

$$\begin{pmatrix} 37 & 54 \\ 81 & 118 \end{pmatrix}$$

In Matlab, the equivalent syntax is

* = matrix multiplication
.* = element wise multiplication

<Illustrate this in Matlab>

In general, when we refer to element-wise multiplication, we use subscripts to denote the specific indices.

element-wise multiplication = $C_{mn} = A_{mn} * C_{mn}$

This is typically for a coding application (like writing pseudo code) and not for performing a mathematical application.

Summary

desired operation	Mathematica	Matlab
standard matrix multiplication	. or Dot	*
element – wise multiplication	*	.*
matrix power	MatrixPower	^

Transpose

The operation of switching the rows and columns of a matrix is called a **transpose** operation. This is denoted by a T superscript

$B = A^T$ (B is the transpose of A)

Example: Transpose

Consider the matrix

$$A = \begin{pmatrix} a & b & c \\ d & e & f \end{pmatrix}$$

It's transpose is given by

$$A^T = \begin{pmatrix} a & d \\ b & e \\ c & f \end{pmatrix}$$

Mathematica provides the 'Transpose' function to perform this action.

In[24]:=
```
A = ( a  b  c
      d  e  f );
A // MatrixForm

AT = Transpose[A];
AT // MatrixForm
```

Out[25]//MatrixForm=

$$\begin{pmatrix} a & b & c \\ d & e & f \end{pmatrix}$$

Out[27]//MatrixForm=

$$\begin{pmatrix} a & d \\ b & e \\ c & f \end{pmatrix}$$

Note that 'Transpose' can also be called with the '//' notation

In[28]:=
```
A // Transpose // MatrixForm
```

Out[28]//MatrixForm=

$$\begin{pmatrix} a & d \\ b & e \\ c & f \end{pmatrix}$$

In[29]:=
```
Clear[A, AT]
```

Matlab uses the apostrophe (') to denote a transpose operation

```
A = [1 2]';          %A is a vertical, column vector
```