Christopher Lum
lum@uw.edu

Lecture 04b
Resonant Frequency of a Dynamic System

Lecture is on YouTube

The YouTube video entitled ‘Resonant Frequency of a Dynamic System’ that covers this lecture is located at https://youtu.be/0ZUp07xP--A.

Outline

-Resonant Frequency

Resonant Frequency

Consider a bode plot of a system as shown below



Note that the resonant frequency is not necessarily the natural frequency nor is it necessarily the damped natural frequency. The resonant frequency is simply the frequency where the maximum amplification occurs. Note that this definition is convenient as it applies to systems of order higher than 2 whereas the concept of natural frequency and damped natural frequency can typically only be applied to 2nd order systems.

We can show mathematically that $\omega_r \mp \omega_n \mp \omega_d$ for a 2nd order system. Consider a second order system in standard form

$$G(s) = \frac{\omega_n^2}{s^2 + 2\,\zeta\,\omega_n\,s + \omega_n^2}$$

So $G(j\,\omega)$ is given by

$$G(j\,\omega) = \frac{\omega_n^2}{-\omega^2 + 2\,\zeta\,\omega_n\,\omega\,j + \omega_n^2}$$

$$= \frac{\omega_n^2}{\omega_n^2 - \omega^2 + 2\,\zeta\,\omega_n\,\omega\,j} \qquad \text{let: } a = \omega_n^2 - \omega^2,\ b = 2\,\zeta\,\omega_n\,\omega$$

```
a = ωn^2 - ω^2;
b = 2 ζ ωn ω;
```

$$= \frac{\omega_n^2}{a + b\,j}$$

$$= \frac{\omega_n^2}{a + b\,j}\left(\frac{a - b\,j}{a - b\,j}\right)$$

```
num = ωn^2 (a - b I) // Expand
den = (a + b I) (a - b I) // Expand
```

$-\omega^2\,\omega n^2 - 2\,i\,\zeta\,\omega\,\omega n^3 + \omega n^4$

$\omega^4 - 2\,\omega^2\,\omega n^2 + 4\,\zeta^2\,\omega^2\,\omega n^2 + \omega n^4$

So we have

$$G(j\,\omega) = \alpha + \beta\,i$$

where $\alpha = \dfrac{\omega_n^4 - \omega_n^2\,\omega^2}{\omega^4 - 2\,\omega^2\,\omega_n^2 + 4\,\zeta^2\,\omega^2\,\omega_n^2 + \omega_n^4}$

$\beta = -\dfrac{2\,\zeta\,\omega\,\omega_n^3}{\omega^4 - 2\,\omega^2\,\omega_n^2 + 4\,\zeta^2\,\omega^2\,\omega_n^2 + \omega_n^4}$

```
α = (-ω^2 ωn^2 + ωn^4)/den
β = (-2 ξ ω ωn^3)/den
```

$$\frac{-\omega^2\,\omega n^2+\omega n^4}{\omega^4-2\,\omega^2\,\omega n^2+4\,\zeta^2\,\omega^2\,\omega n^2+\omega n^4}$$

$$-\frac{2\,\zeta\,\omega\,\omega n^3}{\omega^4-2\,\omega^2\,\omega n^2+4\,\zeta^2\,\omega^2\,\omega n^2+\omega n^4}$$

So the magnitude is given by

$$|\,G(\,j\,\omega)\,| = \sqrt{\alpha^2+\beta^2} \qquad \textbf{\textit{(Eq.1)}}$$

```
MagGjω[ω_] = Sqrt[α^2 + β^2] // Simplify
```

$$\sqrt{\frac{\omega n^4}{\omega^4+2\times\left(-1+2\,\zeta^2\right)\,\omega^2\,\omega n^2+\omega n^4}}$$

We can solve for the peak by taking the derivative and setting it equal to zero.

```
dMagGjωdω[ω_] = D[MagGjω[ω], ω] // Simplify
```

$$-\frac{2\,\left(\omega^3+\left(-1+2\,\zeta^2\right)\,\omega\,\omega n^2\right)\,\left(\frac{\omega n^4}{\omega^4+2\times\left(-1+2\,\zeta^2\right)\,\omega^2\,\omega n^2+\omega n^4}\right)^{3/2}}{\omega n^4}$$

We can note here that the slope is not zero at ω_n or $\omega_d = \omega_n\,\sqrt{1-\zeta^2}$

```
dMagGjωdω[ωn] // Simplify
dMagGjωdω[ωn Sqrt[1 - ξ^2]] // Simplify
```

$$-\frac{\sqrt{\frac{1}{\zeta^2}}}{2\,\omega n}$$

$$\frac{2\,\sqrt{1-\zeta^2}\,\sqrt{\frac{1}{4\,\zeta^2-3\,\zeta^4}}}{\left(-4+3\,\zeta^2\right)\,\omega n}$$

Although it is difficult to directly solve for when this is zero, we can show that $|\,G(j\,\omega)\,|$ is maximum at the resonant frequency of ω_r which is given by

$$\omega_r = \omega_n\,\sqrt{1-2\,\zeta^2} \qquad \textbf{\textit{(Eq.2)}}$$

```
ωr = ωn √(1 - 2 ζ^2);
dMagGjωdω[ωr] // Simplify
0
```

Note that although $\omega_r = \omega_n \sqrt{1 - 2\,\zeta^2}$ appears to solve $\frac{\partial |G(j\,\omega)|}{\partial\,\omega} = 0$ for all values of ω_n and ζ, we can note that in order for ω_r to be real, we need the term $1 - 2\,\zeta^2$ to be greater than 0.

$$1 - 2\,\zeta^2 > 0$$

$$1 > 2\,\zeta^2$$

$$\frac{1}{\sqrt{2}} > \zeta \qquad \textbf{(Eq.3)}$$

This is a somewhat important result. It states that the plot of $\frac{\partial |G(j\,\omega)|}{\partial\,\omega}$ vs. ω is equal to zero at a real value of ω only if $\zeta < 1/\sqrt{2}$. Following this logic, this implies that the magnitude plot of $|G(j\,\omega)|$ vs. ω has a maximum/minimum at a real value of ω only if $\zeta < 1/\sqrt{2}$. Extending this logic one more step, this means that if $\zeta > 1/\sqrt{2}$, the magnitude plot of $|G(j\,\omega)|$ vs. ω must be monotonically increasing or decreasing. This means that there is no resonant frequency for system with $\zeta > 1/\sqrt{2}$.

Often the damping ratio of $\zeta = 1/\sqrt{2}$ is called the critical damping value (not to be confused with a critically damping system with $\zeta = 1$) because of this phenomenon.

Example $\left(\zeta < 1/\sqrt{2}\right)$

Again, consider the mass, spring, damper system we used in our previous lectures

$$G(s) = \frac{Z(s)}{U(s)} = \frac{3}{s^2 + \frac{1}{2}\,s + 4} \qquad \text{(recall: } m = 1/3,\ c = 1/6,\ k = 4/3)$$

Recall the bode plot of this system was as shown below



We can see what the resonant frequency of the system is by looking at the peak of the bode plot. We see that the maximum amplification occurs at approximately

$\omega_r \approx 1.97\ \mathrm{rad}/s$

Let us rescale the output (AKA multiply the output by 4/3) so the DC gain is 1

$\tilde{G}(s) = \frac{\tilde{Z}(s)}{U(s)} = \frac{4}{s^2+\frac{1}{2}s+4}$

Solving for the damping ratio and natural frequency we obtain

```
temp = Solve[{2 ξ ωn == 1 / 2, ωn^2 == 4}, {ξ, ωn}];
ξgiven = ξ /. temp[[2]]
ωngiven = ωn /. temp[[2]]
```

$\frac{1}{8}$

2

We can plot the magnitude plot, $|G(j\omega)|$ vs. ω and $\frac{\partial |G(j\omega)|}{\partial \omega}$ vs. ω

```
Plot[MagGjω[ω] /. {ξ → ξgiven, ωn → ωngiven},
 {ω, 1 / 10, 10}, PlotRange → All, AxesLabel → {"ω", "|G(jω)|"}]
```



As can be seen, the plot has a maximum, and therefore we should be able to find a real value of $\omega = \omega_r$ where the response is maximum.

```
Plot[dMagGjωdω[ω] /. {ξ → ξgiven, ωn → ωngiven},
 {ω, 1 / 10, 10}, PlotRange → All, AxesLabel → {"ω", "∂ | G (jω) | / ∂ ω"}]
```




```
temp = Solve[(dMagGjωdω[ω] /. {ξ → ξgiven, ωn → ωngiven}) == 0, ω];
ωrGiven = ω /. temp[[3]]
ωrGiven // N
```

$$\frac{\sqrt{\frac{31}{2}}}{2}$$

1.9685

So we see that the resonant frequency of this system is

$$\omega_r = \frac{\sqrt{31/2}}{2} \approx 1.9685 \text{ rad}/s$$

We can verify that this corresponds with our previous result that $\omega_n \sqrt{1 - 2\zeta^2}$

```
ωr /. {ωn → ωngiven, ξ → ξgiven}
```

$$\frac{\sqrt{\frac{31}{2}}}{2}$$

We can compare the resonant frequency with the damped natural frequency. We know that the damped natural frequency of the system is

$$\omega_d = \omega_n \sqrt{1 - \zeta^2}$$

```
ωd = ωngiven √(1 - ξgiven²)
ωd // N
```

$$\frac{3\sqrt{7}}{4}$$

```
1.98431
```

So in summary, we have

$\omega_n = 2$	(natural frequency)
$\omega_d = \omega_n \sqrt{1 - \zeta^2} = \frac{3\sqrt{7}}{4} \approx 1.9843$	(damped natural frequency)
$\omega_r = \omega_n \sqrt{1 - 2\zeta^2} = \frac{\sqrt{31/2}}{2} \approx 1.9685$	(resonant frequency)

So we see that in general, $\omega_r \neq \omega_d \neq \omega_n$

The difference between ω_r and ω_d is evident at larger ζ

```
Plot[√(1 - ξ²) - √(1 - 2 ξ²), {ξ, 0, 1/√2},
 (*Plot Options*)
 AxesLabel → {"ξ", "√(1 - ξ²) - √(1 - 2 ξ²)"}]
```



Example $\left(\zeta = 1/\sqrt{2}\right)$

We can contrast this with the situation if $\zeta = 1/\sqrt{2}$

$$\zeta = 1/\sqrt{2}$$
$$\omega_n = 2$$

```
ξgiven = 1/√2;
ωngiven = 2;
```

We can plot the magnitude plot, $|G(j\omega)|$ vs. ω

```
Plot[MagGjω[ω] /. {ξ → ξgiven, ωn → ωngiven},
 {ω, 1/10, 10}, PlotRange → All, AxesLabel → {"ω", "|G(jω)|"}]
```



As can be seen, the plot is monotonically decreasing and therefore, there is no maximum response

(except for $\omega = 0$)

```
ωr /. {ξ → ξgiven, ωn → ωngiven}
```

0

Example $\left(\zeta > 1 \middle/ \sqrt{2}\right)$

We can contrast this with the situation if $\zeta > 1 \big/ \sqrt{2}$

$\zeta = 0.8$
$\omega_n = 2$

```
ξgiven = 8 / 10;
ωngiven = 2;
```

We can plot the magnitude plot, $|\,G(j\,\omega)\,|$ vs. ω

```
Plot[MagGjω[ω] /. {ξ → ξgiven, ωn → ωngiven},
 {ω, 1 / 10, 10}, PlotRange → All, AxesLabel → {"ω", "|G(jω)|"}]
```



As can be seen, the plot is monotonically decreasing and therefore, there is no maximum response (except for $\omega = 0$). The "resonant frequency" is imaginary.

```
ωr /. {ξ → ξgiven, ωn → ωngiven}
```

$$\frac{2\,i\,\sqrt{7}}{5}$$