Christopher Lum
lum@uw.edu

Lecture 07b
PDEs: 1D Wave Equation

The YouTube videos that cover this lecture are located at:

-‘Standing Waves Demonstration’ at https://youtu.be/42WBuhVJ7sA.

-‘Derivation of the 1D Wave Equation’ at https://youtu.be/IAut5Y-Ns7g.

-‘Solving the 1D Wave Equation’ at https://youtu.be/lMRnTd8yLeY.

Modeling: Vibrating String, Wave Equation

Let us derive the equation governing small transverse vibrations of an elastic string.

To illustrate a wave, go to YouTube and search for ‘Standing Waves Demonstration’ at https://youtu.be/42WBuhVJ7sA.

In this scenario, we seek to understand how the deflection of the string at a horizontal location $x \in [0, L]$ will deflect in the vertical direction at a given time t. In this case, the function u describes this vertical deflection.

$u(x, t)$ = deflection of the string in the vertical direction at horizontal location x at time t.



We make some simplifying assumptions.

1. The mass of the string per unit length, ρ (with units of kg/m), is constant (homogeneous string).

2. The string is perfectly elastic and does not offer any resistance to bending.

3. Gravitational force on the string is negligible (the tension force dominates or the string is vibrating in a plane normal to gravity vector).

4. The string performs small transverse motions (in the *x* direction). Every particle of the string moves only vertically

5. The deflection and the slope at every point of the string always remains small in absolute value.

Derivation of the PDE of the Model ("Wave Equation") from Forces

With these assumptions, we can derive the governing equation of the string. We can analyze a small section of the string from *P* to *Q* (see previous figure). We see that the tension on the left and right side of the string are T_1 and T_2, respectively. From assumption 3 to hold, the horizontal components must be equal. Note that we do not need to assume angles are small at this point.

$$T_1 \cos(\alpha) = T_2 \cos(\beta) = T = \text{constant} \qquad \textbf{\textit{(Eq.1)}}$$

So summing forces in the vertical direction for a small section string with mass $\rho\,\Delta x$ yields

$$\sum F_{\text{vertical}} = m\, a_{\text{vertical}}$$

$$T_2 \sin(\beta) - T_1 \sin(\alpha) = \rho\,\Delta x \frac{\partial^2 u}{\partial t^2}$$

where $\quad \rho$ = linear density of string (kg/m)

Let us divide both sides by T

$$\frac{T_2 \sin(\beta)}{T} - \frac{T_1 \sin(\alpha)}{T} = \frac{\rho\,\Delta x}{T} \frac{\partial^2 u}{\partial t^2} \qquad \text{recall: } T = T_1 \cos(\alpha) = T_2 \cos(\beta)$$

$$\frac{T_2 \sin(\beta)}{T_2 \cos(\beta)} - \frac{T_1 \sin(\alpha)}{T_1 \cos(\alpha)} = \frac{\rho\,\Delta x}{T} \frac{\partial^2 u}{\partial t^2}$$

$$\tan(\beta) - \tan(\alpha) = \frac{\rho\,\Delta x}{T} \frac{\partial^2 u}{\partial t^2}$$

From the figure, we see that $\tan(\alpha)$ is actually the slope of the string at x and $\tan(\beta)$ is the slope of the string at $x + \Delta\, x$. More precisely

$$\tan(\alpha) = \frac{\partial u}{\partial x}\Big|_{x=x} \qquad \text{and } \tan(\beta) = \frac{\partial u}{\partial x}\Big|_{x=x+\Delta x}$$

So our expression becomes

$$\frac{\partial u}{\partial x}\Big|_{x=x+\Delta x} - \frac{\partial u}{\partial x}\Big|_{x=x} = \frac{\rho\,\Delta x}{T} \frac{\partial^2 u}{\partial t^2}$$

$$\frac{1}{\Delta x}\left(\frac{\partial u}{\partial x}\Big|_{x=x+\Delta x}-\frac{\partial u}{\partial x}\Big|_{x=x}\right)=\frac{\rho}{T}\frac{\partial^2 u}{\partial t^2}$$

In the limit as Δx approaches 0, we see that the left side of this expression is effectively asking how does the $\frac{\partial u}{\partial x}$ change as a function of x. In other words, the left side is effectively the second partial of u w.r.t. x

$$\frac{\partial^2 u}{\partial x^2}=\frac{\rho}{T}\frac{\partial^2 u}{\partial t^2}$$

Rearranging terms

$$\frac{\partial^2 u}{\partial t^2}=\frac{T}{\rho}\frac{\partial^2 u}{\partial x^2}$$

$$\frac{\partial^2 u}{\partial t^2}=c^2\frac{\partial^2 u}{\partial x^2} \qquad \textbf{\textit{(Eq.3)}}$$

where $\quad c^2=\frac{T}{\rho}$

This is the **one-dimensional wave equation**. The notation c^2 (instead of c) is chosen to indicate that this constant is positive.

Note that while parameters such as the length do not appear directly in the PDE, they will become critical when we apply initial and boundary conditions to find actual solutions to this PDE.

Solution by Separating Variables. Use of Fourier Series

In the last section, we derived the 1-D wave equation

$$\frac{\partial^2 u}{\partial t^2}=c^2\frac{\partial^2 u}{\partial x^2} \qquad \textbf{\textit{(Eq.1)}}$$

As we mentioned previously, this general PDE may have many solutions. We can narrow the solution space by applying **boundary conditions**. In this situation, the ends of the string are fixed and cannot deflect. Mathematically, this translates to

$$u(0,t)=u(L,t)=0 \quad \forall\, t \qquad \textbf{\textit{(Eq.2)}}$$

Furthermore, we can apply **initial conditions** to further constrain the problem. We use these to characterize the initial deflection, $f(x)$, and velocity, $g(x)$, of the string

$$u(x,0)=f(x) \qquad \text{(initial deflection)} \qquad \textbf{\textit{(Eq.3.a)}}$$

$$\frac{\partial u(x,0)}{\partial t}=\frac{\partial u}{\partial t}\Big|_{t=0}=g(x) \qquad \text{(initial velocity)} \qquad \textbf{\textit{(Eq.3.b)}}$$

The procedure to solve this particular type of PDE with boundary and initial conditions is as follows

Step 1. Apply method of separating variables (aka product method) to obtain two ordinary differential equations.

Step 2. Determine solutions of those two equations that satisfy the boundary conditions.

Step 3. Use Fourier series to compose those solutions to get a solution to the wave equation that also satisfies the initial conditions.

During these steps, there are many variables we will introduce. To help with notation, a glossary of relevant symbols is given here

<u>Initial Problem Statement</u>
T =tension in string (N)
ρ =density per unit length (kg/m)
L = length of string (m)
$c = (T/\rho)^{1/2}$ = given parameter of PDE
$u(x, t)$ = vertical displacement of string (m)
x = horizontal position (m)
t = time (s)
$f(x)$ = initial string displacement (m)
$g(x)$ = initial string velocity (m/s)

<u>Step 1:</u>
$u(x, t) = F(x)\,G(t)$
$k = -p^2$ = separation constant (must be negative) (will show it must take specific values)

<u>Step 2:</u>
$F_n(x)$ = general solution at specific $n \in \mathbb{Z}$ value
$G_n(t)$ = general solution at specific $z \in \mathbb{Z}$ value
$u_n(x, t)$ = eigenfunction at specific $n \in \mathbb{Z}$ value
λ_n = eigenvalue at specific $n \in \mathbb{Z}$ values

<u>Step 3:</u>
B_n, B_n^* = coefficients of Fourier series

Step 1. Two ODEs from the Wave Equation

In the method of separating variables (aka product methods, separation of variables), we assume that solutions to the wave equation (Eq.1) take the form of

$$u(x, t) = F(x)\,G(t) \qquad \textbf{\textit{(Eq.4)}}$$

In other words, we assume that $u(x, t)$ can be written as the product of two functions, F and G which are only functions of x and t, respectively.

We can differentiate Eq.4 with respect to t twice to obtain

$$\frac{\partial^2 u}{\partial t^2} = F(x)\frac{d^2 G(t)}{d t^2} = F\ddot{G}$$ note: dot notation implies derivative with respect to time

We can do the same with respect to x

$$\frac{\partial^2 u}{\partial x^2} = \frac{d^2 F(x)}{d x^2} G(t) = F''G$$ note: prime notation implies derivative with respect to x

Inserting these results into the 1D wave equation ($\frac{\partial^2 u}{\partial t^2} = c^2 \frac{\partial^2 u}{\partial x^2}$) yields

$$F\ddot{G} = c^2 F''G$$

Dividing by $c^2 F G$, we obtain

$$\frac{\ddot{G}(t)}{c^2 G(t)} = \frac{F''(x)}{F(x)}$$

We notice that the left side is only a function of t and the right side is only a function of x. Since these variables are independent, one of them can change without the other. Therefore, the only way for the above equation to be true is if both sides are equal to the same constant. Unless both sides were constant, changing t would change the left side but not change the right (and similarly for x) thereby violating the equality relationship. This constant k is referred to as the **separation constant**.

$$\frac{\ddot{G}(t)}{c^2 G(t)} = \frac{F''(x)}{F(x)} = k$$

This realization leads to two, separate, ordinary differential equations

$$F''(x) - k F(x) = 0 \qquad \textbf{\textit{(Eq.5)}}$$

$$\ddot{G}(t) - c^2 k G(t) = 0 \qquad \textbf{\textit{(Eq.6)}}$$

Step 2. Satisfying the Boundary Conditions

We now would like to determine solutions for $F(x)$ and $G(t)$ such that $u(x, t) = F(x) G(t)$ satisfies the boundary conditions previously described in Eq.2. In other words

$$u(0, t) = F(0) G(t) = 0 \,\forall\, t \qquad \textbf{\textit{(Eq.7)}}$$

$$u(L, t) = F(L) G(t) = 0 \,\forall\, t$$

Solving the First ODE (Eq.5)

We see that an immediate solution which satisfies these boundary conditions is $G(t) \equiv 0$. However, since $u = F G$, this means that $u = 0$ so this is a trivial solution and is of little interest. Therefore, the other way that $u(0, t) = F(0)\,G(t) = 0$ and $u(L, t) = F(L)\,G(t) = 0$ are true for all t is if

$$F(0) = F(L) = 0 \qquad \textbf{\textit{(Eq.8)}}$$

So solving Eq.5 becomes a question of solving

$$F''(x) - k\,F(x) = 0$$

with $F(0) = F(L) = 0$

In a homework assignment, we will investigate why k must be negative. To illustrate this and remind ourselves that k is negative, we make the substitution of $k = -p^2$, and the problem can be restated as

$$F''(x) + p^2\,F(x) = 0$$

with $F(0) = F(L) = 0$

Note that this is a second order, linear, homogeneous, ordinary differential equation with an initial <u>AND</u> final condition. Since the function's independent variable is x, or a spatial variable, perhaps a better name for these are boundary conditions instead of initial/final conditions.

From our studies of mechanical systems, we see that this corresponds to an undamped oscillator (this corresponds to a damping ratio of 0 and a natural frequency of p). The general solution to this type of equation is

$$F(x) = A\cos(p\,x) + B\sin(p\,x)$$

We can now apply the boundary conditions to find A and B

<u>Apply BC#1: $F(0) = 0$</u>

$$F(0) = A\cos(0) + B\sin(0)$$

$$0 = A$$

<u>Apply BC#2: $F(L) = 0$</u>

$$F(L) = A\cos(p\,L) + B\sin(p\,L) \qquad \text{note: } A = 0$$

$$0 = B\sin(p\,L)$$

We notice that $B = 0$ yields the trivial solution of $F(x) = 0$, so we need to find another solution such that $0 = B\sin(p\,L)$ is satisfied but we do not obtain a trivial solution. We see that this is only possible if $\sin(p\,L) = 0$. We notice that any p such that $p\,L = n\,\pi$ for $n \in \mathbb{Z}$ (integer) will satisfy this equation. Therefore, we see that

$$p L = n \pi$$

$$p = \frac{n \pi}{L} \qquad n \in Z \qquad \textbf{(Eq.9)}$$

Therefore, if Eq.9 is satisfied, then B can be arbitrary and we obtain the following general solution to the spatial function, $F(x)$.

$$F(x) = B \sin\left(\frac{n \pi}{L} x\right) \qquad n \in Z$$

Since this is a function of the integer n, we adopt notation of

$$F_n(x) = B \sin\left(\frac{n \pi}{L} x\right) \qquad n \in Z \qquad \textbf{(Eq.10)}$$

We can verify that this satisfies the original ordinary differential equation and the boundary conditions. Note that these are not yet the eigenfunctions of the entire problem as $u = F G$ is the solution to the entire problem.

```
In[1]:= Fn[x_] = B Sin[(n π)/L x];

Print["Satisfies PDE?"]
D[Fn[x], {x, 2}] + p^2 Fn[x] == 0
```

Satisfies PDE?

Out[3]= $B p^2 \mathrm{Sin}\left[\frac{n \pi x}{L}\right] - \frac{B n^2 \pi^2 \mathrm{Sin}\left[\frac{n \pi x}{L}\right]}{L^2} == 0$

We see that this does not satisfy the original ODE for a general value of p. However we know p only takes on specific values. In this case, it does satisfy the original PDE.

```
In[4]:= Print["Satisfies PDE?"]
(D[Fn[x], {x, 2}] + p^2 Fn[x] /. {p → (n π)/L}) == 0
```

Satisfies PDE?

Out[5]= True

We now check if this satisfies the boundary conditions for specific values of p

```
In[6]:= Print["Satisfies boundary conditions?"]
Fn[0] == 0
(Fn[L] /. {p → n π / L}) == 0
```

Satisfies boundary conditions?

Out[7]= True

Out[8]= B Sin[n π] == 0

Again, we see that this does not satisfy the BCs for an arbitrary n but as long as n is an integer value, this will work.

In[9]:=
```
Print["Satisfies boundary conditions?"]
Fn[0] == 0
Simplify[Fn[L] /. {p → n π / L}, n ∈ Integers] == 0
```

```
Satisfies boundary conditions?
```

Out[10]= True

Out[11]= True

So we see that we satisfy all the requirements

Solving the Second ODE (Eq.6)

We now turn our attention to Eq.6, which governs the proposed temporal function and is repeated here for convenience

$$\ddot{G}(t) - c^2\, k\, G(t) = 0 \qquad \textbf{\textit{(Eq.6)}}$$

Recall that the separation constant k must be negative and we substituted $k = -p^2$. Furthermore, from the previous section, we determined that p must take specific values, namely $p = \frac{n\pi}{L}$ with $n \in \mathbb{Z}$ (Eq.9). So valid values of k are

$$k = -p^2 \qquad \text{recall: } p = \frac{n\pi}{L} \text{ with } n \in \mathbb{Z}$$

$$k = -\left(\frac{n\pi}{L}\right)^2 \qquad n \in \mathbb{Z}$$

Therefore, we can rewrite Eq.6 as

$$\ddot{G}(t) - c^2\left(-\left(\frac{n\pi}{L}\right)^2\right) G(t) = 0$$

$$\ddot{G}(t) + \left(\frac{c\, n\, \pi}{L}\right)^2 G(t) = 0$$

$$\ddot{G}(t) + \lambda_n{}^2\, G(t) = 0$$

where $\lambda_n = c\, n\, \pi / L$

Once again, we recognize this as a second order, undamped, ordinary differential equation with general solution of

$$G_n(t) = B_n \cos(\lambda_n t) + B_n{}^* \sin(\lambda_n t)$$

We will temporarily defer finding the coefficients B_n and $B_n{}^*$

So the solution is given by $u_n(x, t) = F_n(x)\, G_n(t)$

$$u_n(x,\,t) = B\sin\left(\frac{n\pi}{L}\,x\right)[B_n\cos(\lambda_n\,t) + B_n{}^*\sin(\lambda_n\,t)]$$

$$= [B\,B_n\cos(\lambda_n\,t) + B\,B_n{}^*\sin(\lambda_n\,t)]\sin\left(\frac{n\pi}{L}\,x\right)$$

Note the the product $B\,B_n$ and $B\,B_n{}^*$ are simply another constant, so for ease of notation, we can simply incorporate the B coefficient into the B_n and $B_n{}^*$ coefficients and obtain the final expression of

$$u_n(x,\,t) = (B_n\cos(\lambda_n\,t) + B_n{}^*\sin(\lambda_n\,t))\sin\left(\frac{n\pi}{L}\,x\right) n \in \mathbb{Z} \qquad \textit{\textbf{(Eq.11)}}$$

where $\quad \lambda_n = c\,n\,\pi/L \quad$ (eigenvalues)

In[12]:=
```
(*Input λn and un(x,t)*)
λn[n_] = c n π / L;

Print["un(x,t)"]
un[x_, t_, n_] = (Bn Cos[λn[n] t] + Bnstar Sin[λn[n] t]) Sin[n π/L x]

(*verify we satisfy original PDE*)
Print["Satisfies original PDE?"]
D[un[x, t, n], {t, 2}] == c^2 D[un[x, t, n], {x, 2}] // Simplify

(*verify that we satisfy the boundary condition*)
Print["Satisfies boundary conditions?"]
un[0, t, n] == 0
Simplify[un[L, t, n], n ∈ Integers] == 0
```

u_n(x,t)

Out[14]= $\left(\text{Bn Cos}\left[\frac{c\,n\,\pi\,t}{L}\right] + \text{Bnstar Sin}\left[\frac{c\,n\,\pi\,t}{L}\right]\right)\text{Sin}\left[\frac{n\,\pi\,x}{L}\right]$

Satisfies original PDE?

Out[16]= True

Satisfies boundary conditions?

Out[18]= True

Out[19]= True

Let us also verify that the superposition principle holds in this case. We can do this by generating solutions which are linear combinations of each eigenfunction

In[20]:=
```
uSuper[x_, t_] = 2 un[x, t, 2] - 3 un[x, t, 3] + π un[x, t, 23]

(*Verify this satisfies PDE and boundary conditions*)
Print["Satisfies original PDE?"]
D[uSuper[x, t], {t, 2}] == c^2 D[uSuper[x, t], {x, 2}] // Simplify

(*verify that we satisfy the boundary condition*)
Print["Satisfies boundary conditions?"]
uSuper[0, t] == 0
uSuper[L, t] == 0
```

Out[20]=
$$2\left(\text{Bn}\,\text{Cos}\left[\frac{2\,c\,\pi\,t}{L}\right]+\text{Bnstar}\,\text{Sin}\left[\frac{2\,c\,\pi\,t}{L}\right]\right)\text{Sin}\left[\frac{2\,\pi\,x}{L}\right]-3\left(\text{Bn}\,\text{Cos}\left[\frac{3\,c\,\pi\,t}{L}\right]+\text{Bnstar}\,\text{Sin}\left[\frac{3\,c\,\pi\,t}{L}\right]\right)\text{Sin}\left[\frac{3\,\pi\,x}{L}\right]+\pi\left(\text{Bn}\,\text{Cos}\left[\frac{23\,c\,\pi\,t}{L}\right]+\text{Bnstar}\,\text{Sin}\left[\frac{23\,c\,\pi\,t}{L}\right]\right)\text{Sin}\left[\frac{23\,\pi\,x}{L}\right]$$

```
Satisfies original PDE?
```

Out[22]= True

```
Satisfies boundary conditions?
```

Out[24]= True

Out[25]= True

The family of functions described by Eq.11 are known as **eigenfunctions,** or characteristic functions and the values $\lambda_n = c\,n\,\pi/L$ are called the **eigenvalues**, or characteristic values of the vibrating string. The set $\{\lambda_1, \lambda_2, \ldots\}$ are called the **spectrum**.

Discussion of eigenfunctions. We note that λ_n governs the temporal “angular speed” of the function as it is the angular velocity coefficient in the cos and sin terms. In this sense, λ_n describes how fast the function “oscillates” in the temporal variable.

From this, we see that each u_n represents a harmonic motion having a frequency of $\lambda_n/2\,\pi$ Hz.

Let us choose some example physical constants to illustrate some example values of λ_n.

In[26]:=
```
(*Define problem constants*)
LGiven = 3.5;
cGiven = 200;
```

What are the λ_n values with these constants for different values of n?

In[28]:=
```
λnGiven[n_] = λn[n] /. {c → cGiven, L → LGiven};

(*Create table data*)
nmax = 5;
nData = Table[n, {n, 1, nmax}];
λnData = Table[λnGiven[n], {n, 1, nmax}];
λnDataHz = Table[λnGiven[n] / (2 π), {n, 1, nmax}];

(*Now place them in a grid so it looks like a standard
 table (note the transpose so the orientation is vertical)*)
gridData = Transpose[{nData, λnData, λnDataHz}];

(*Add headers*)
gridDataWithHeaders = Prepend[gridData, {"n", "λn (rad/s)", "λn (Hz)" }];

(*Draw the table*)
Grid[gridDataWithHeaders, Frame → All]
```

Out[35]=

n	λ_n (rad/s)	λ_n (Hz)
1	179.52	28.5714
2	359.039	57.1429
3	538.559	85.7143
4	718.078	114.286
5	897.598	142.857

We can evaluate u_n with these given values (let us assume $B_n = B_n^* = 1$ to make the plotting simpler)

In[36]:=
```
unGiven[x_, t_, n_] = un[x, t, n] /. {L → LGiven, c → cGiven, Bn → 1, Bnstar → 1}
```

Out[36]=
```
(Cos[179.52 n t] + Sin[179.52 n t]) Sin[0.897598 n x]
```

We can animate how this function varies with t between $x \in [0, L]$. We can also animate for the various modes $n = 1, 2, \ldots$

Go to Animation Lecture (lecture07c_animation.nb)

Let us use the 'Manipulate' function to easily view an animation of the function (click on the + symbol to open the animation pane)

In[37]:=
```
(*Define plotting constants*)
tMax = 2 π / λnGiven[1];
yMin = -1.5;
yMax = 1.5;

(*Animate the scenario*)
Manipulate[
 Plot[unGiven[x, t, n], {x, 0, LGiven},
  PlotRange → {{0, LGiven}, {yMin, yMax}},
  PlotLabel → "un(x,t) with n=" + n,
  AxesLabel → {"x", "un(x,t)"}],
 {t, 0, tMax, AppearanceElements → All},
 {n, 1, 5, 1, AppearanceElements → All}]
```

Out[40]=



Step 3. Solution of the Entire Problem. Fourier Series

As we saw previously, $u_n(x, t)$ is a solution which satisfies the boundary conditions and generates standing waves. Clearly, this is not a general solution and does not solve the initial conditions of $u_n(x, 0) = f(x)$ for an arbitrary $f(x)$. However, since the PDE is linear and homogeneous, we can apply the Fundamental Theorem of Superposition (section 11.1). This states that the sum of solutions is also a solution. So the following infinite series is a solution

$$u(x, t) = \sum_{n=1}^{\infty} u_n(x, t) = \sum_{n=1}^{\infty} (B_n \cos(\lambda_n t) + B_n^* \sin(\lambda_n t)) \sin\left(\frac{n\pi}{L} x\right) \qquad \textbf{(Eq.12)}$$

Satisfying Initial Condition (Position)

We now would like to satisfy the initial condition of $u(x, 0) = f(x)$. Eq.12 becomes

$u(x, 0) = \sum_{n=1}^{\infty} B_n \sin\left(\frac{n\pi}{L} x\right) = f(x)$ ***(Eq.13)***

Therefore, we need to choose the coefficients B_n such that $u(x, 0) = f(x)$. Recall from Section 11.2, if we extend a function with period L as an odd function, the half range expansion is given as

$f(x) = \sum_{n=1}^{\infty} b_n \sin\left(\frac{n\pi}{L} x\right)$ ***(Eq.5** in Section 11.2)***

where $b_n = \frac{2}{L} \int_0^L f(x) \sin\left(\frac{n\pi}{L} x\right) dx$ $n = 1, 2 \ldots$ ***(Eq.6** in Section 11.2)***

Comparing Eq.13 with this half range expansion, we see that they are identical so we see that we need to choose coefficients B_n using the half range expansion of

$B_n = \frac{2}{L} \int_0^L f(x) \sin\left(\frac{n\pi}{L} x\right) dx$ $n = 1, 2 \ldots$ ***(Eq.14)***

Satisfying Initial Condition (Velocity)

In a similar fashion, we would like to satisfy the initial condition of

$$\frac{\partial u}{\partial t} \Big|_{t=0} = g(x)$$

$$\frac{\partial}{\partial t}\left[\sum_{n=1}^{\infty} (B_n \cos(\lambda_n t) + B_n^* \sin(\lambda_n t)) \sin\left(\frac{n\pi}{L} x\right)\right] \Big|_{t=0} = g(x)$$

$$\left[\sum_{n=1}^{\infty} (-B_n \lambda_n \sin(\lambda_n t) + B_n^* \lambda_n \cos(\lambda_n t)) \sin\left(\frac{n\pi}{L} x\right)\right] \Big|_{t=0} = g(x)$$

$$\sum_{n=1}^{\infty} B_n^* \lambda_n \sin\left(\frac{n\pi}{L} x\right) = g(x)$$

Once again, we need to choose the coefficients B_n^* so the left side equals $g(x)$. Once again, from Section 11.2 the odd expansion of a function ($g(x)$ in this case) can be written as

$g(x) = \sum_{n=1}^{\infty} b_n \sin\left(\frac{n\pi}{L} x\right)$ (Repeat of Eq.5** for convenience)

where $b_n = \frac{2}{L} \int_0^L g(x) \sin\left(\frac{n\pi}{L} x\right) dx$ $n = 1, 2 \ldots$ (Repeat of Eq.6** for convenience)

Comparing this to our previous expression we have

$$\sum_{n=1}^{\infty} B_n^* \lambda_n \sin\left(\frac{n\pi}{L} x\right) = \sum_{n=1}^{\infty} b_n \sin\left(\frac{n\pi}{L} x\right)$$

$$B_n^* = \frac{b_n}{\lambda_n}$$

where $b_n = \frac{2}{L} \int_0^L g(x) \sin\left(\frac{n\pi}{L} x\right) dx$ $n = 1, 2 \ldots$

$\lambda_n = c\,n\,\pi / L$

With these substitutions the equation for the coefficients $B_n{}^*$ can be written as

$$B_n{}^* = \frac{2}{c\,n\,\pi} \int_0^L g(x) \sin\left(\frac{n\,\pi}{L}\,x\right) d\,x \quad n = 1,\,2\,\ldots \qquad \textbf{\textit{(Eq.15)}}$$

So we have now have a general solution to the 1-D wave equation.

Example 1: Vibrating String if the Initial Deflection is Triangular

Consider an initial condition of

$$f(x) = \begin{cases} f_1(x) = \frac{2\,q}{L}\,x & \text{if } x \in [0,\,L/2] \\ f_2(x) = \frac{2\,q}{L}\,(L - x) & \text{if(} x \in [L/2,\,L] \end{cases}$$

$$g(x) \equiv 0$$

This corresponds to a triangular deflection with no initial velocity.

We can now find the solution $u(x,\,t)$ for this situation

In[41]:=
```
f1[x_, q_, L_] = (2 q)/L x;
f2[x_, q_, L_] = (2 q)/L (L - x);
f[x_, q_, L_] := If[
   x ≤ L / 2,
   f1[x, q, L],
   f2[x, q, L]
  ]

(*Visualize the function*)
LGiven = 3.5;
Manipulate[
 Plot[f[x, q, LGiven], {x, 0, LGiven},
  AxesLabel → {"x", "f(x)"}],
 {q, 0.01, 1, AppearanceElements → All}
]
```

Out[45]=



Recall that the general solution is given as

$$u(x, t) = \sum_{n=1}^{\infty} u_n(x, t) = \sum_{n=1}^{\infty} (B_n \cos(\lambda_n t) + B_n^* \sin(\lambda_n t)) \sin\left(\frac{n\pi}{L} x\right)$$

where $$B_n = \frac{2}{L} \int_0^L f(x) \sin\left(\frac{n\pi}{L} x\right) d x$$

$$B_n^* = \frac{2}{c n \pi} \int_0^L g(x) \sin\left(\frac{n\pi}{L} x\right) d x$$

We immediately see that $B_n^* = 0 \; \forall \; n$ since $g(x) \equiv 0$.

We can calculate the B_n coefficients using

$$B_n = \frac{2}{L} \int_0^L f(x) \sin\left(\frac{n\pi}{L} x\right) dx$$

$$= \frac{2}{L} \left(\int_0^{L/2} f_1(x) \sin\left(\frac{n\pi}{L} x\right) dx + \int_{L/2}^L f_2(x) \sin\left(\frac{n\pi}{L} x\right) dx \right)$$

$$= \frac{2}{L} \left(\int_0^{L/2} \frac{2q}{L} x \sin\left(\frac{n\pi}{L} x\right) dx + \int_{L/2}^L \frac{2q}{L} (L - x) \sin\left(\frac{n\pi}{L} x\right) dx \right)$$

In[46]:=
```
Bn[n_, q_] = 2/L (Integrate[f1[x, q, L] Sin[(n π)/L x], {x, 0, L/2}] +
    Integrate[f2[x, q, L] Sin[(n π)/L x], {x, L/2, L}]) // Simplify
```

Out[46]=
$$\frac{32\, q \,\text{Cos}\left[\frac{n\pi}{4}\right] \text{Sin}\left[\frac{n\pi}{4}\right]^3}{n^2 \pi^2}$$

So we have

$$B_n = \frac{32\, q \cos\left(\frac{n\pi}{4}\right) \sin\left(\frac{n\pi}{4}\right)^3}{n^2 \pi^2}$$

We can create a table of the first few coefficients

In[47]:=
```
nmax = 9;
t1 = Table[n, {n, 1, nmax}];
t2 = Table[Bn[n, L], {n, 1, nmax}];
gridData = Transpose[{t1, t2}];
gridDataWithHeaders = Prepend[gridData, {"n", "Bn" }];
Grid[gridDataWithHeaders, Frame → All]
```

Out[52]=

n	B_n
1	$\frac{8 L}{\pi^2}$
2	0
3	$-\frac{8 L}{9 \pi^2}$
4	0
5	$\frac{8 L}{25 \pi^2}$
6	0
7	$-\frac{8 L}{49 \pi^2}$
8	0
9	$\frac{8 L}{81 \pi^2}$

So the solution takes the form of

$$u(x, t) = \sum_{n=1}^{\infty} B_n \cos(\lambda_n t) \sin\left(\frac{n\pi}{L} x\right)$$

For computational purposes, we can once again take partial sums

$u_N(x, t) = \sum_{n=1}^{N} B_n \cos(\lambda_n t) \sin\left(\frac{n\pi}{L} x\right) \approx u(x, t)$ recall: $\lambda_n = c\, n\, \pi / L$

In[53]:=
```
uN[x_, t_] := Sum[Bn[n, k] Cos[λn[n] t] Sin[(n π)/L x], {n, 1, Nmax}]
```

We can visualize this example

In[54]:=
```
(*Define plotting and other constants*)
yMin = -3;
yMax = 3;
qGiven = 2.5;

(*Animate the scenario*)
Manipulate[
 Plot[Sum[Bn[n, q] Cos[λn[n] t] Sin[(n π)/L x], {n, 1, Nmax}] /.
    {c → cGiven, L → LGiven, q → qGiven}, {x, 0, LGiven},
   PlotRange → {{0, LGiven}, {yMin, yMax}},
   PlotLabel → "uN(x,t) with N=" + Nmax,
   AxesLabel → {"x", "uN(x,t)"}],
 {t, 0, tMax, AppearanceElements → All},
 {Nmax, 1, 20, 1, AppearanceElements → All}]
```

Out[57]=