Christopher Lum
lum@uw.edu

Lecture 02a
Introduction to Ordinary Differential Equations (ODEs)

The YouTube video entitled ‘Introduction to Ordinary Differential Equations’ that covers this lecture is located at https://youtu.be/yI7UX76tLeY.

Outline

-Intro to ODES
-Mathematical Description of ODEs
-Examples of ODEs
-Roadmap of ODE Lectures

Intro to ODES

ODEs can be used to describe a wide variety of engineering and scientific phenomena. They are incredibly useful in modeling various physical phenomena

Mathematical Description of ODEs

Let us first consider a normal, ordinary equation. For example

$$x^2 + 2x = 3x + 2$$

```
LHS = x^2 + 2 x;
RHS = 3 x + 2;
Plot[{LHS, RHS}, {x, -4, 4}]
```



We are asking for single values of *x* that make the left hand side equal the right hand side.

```
Solve[LHS == RHS, x]
```

```
{{x → -1}, {x → 2}}
```

```
Clear[LHS, RHS]
```

An ODE is similar in the sense that we are still trying to make the left hand side equal the right hand side except now, the *x* that we are searching for is a function, not necessarily a single value and the LHS and RHS involve the function $x(t)$ and its derivatives w.r.t. the independent parameter, typically time, *t*.

An example of an ODE is

$$a \frac{d^2 x(t)}{dt^2} + b \frac{d x(t)}{dt} + c\, x(t) = g(t)$$

You may see alternative notation of

$$a\, \ddot{x}(t) + b\, \dot{x}(t) + c\, x(t) = g(t)$$

In many cases, we will need to specify boundary conditions

$$a \frac{d^2 x(t)}{dt^2} + b \frac{d x(t)}{dt} + c\, x(t) = g(t) \qquad \textbf{\textit{(Eq.1)}}$$

with $x(0) = x_0$

$\dot{x}(0) = \dot{x}_0$

Eq.1 is an example of a linear ODE

Example: Linear ODE

Consider

$$\ddot{x}(t) + 3\,\dot{x}(t) + 2\,x(t) = 4 - t + 3\,t^2$$

where $\quad x(0) = 3$

$\qquad\qquad \dot{x}(0) = -2$

We claim that the solution to this ODE is the function

$$x(t) = 2\,e^{-2\,t} - 7\,e^{-t} + 8 - 5\,t + \tfrac{3}{2}\,t^2$$

```
x[t_] = 2 Exp[-2 t] - 7 Exp[-t] + 8 - 5 t + 3/2 t^2
Plot[x[t], {t, 0, 10}, AxesLabel → {"t", "x(t)"}]
```

$$8 + 2\,e^{-2\,t} - 7\,e^{-t} - 5\,t + \frac{3\,t^2}{2}$$



We can verify that this satisfies the ODE. Let us compute the left hand side

$$\ddot{x}(t) + 3\,\dot{x}(t) + 2\,x(t) =$$

$$\frac{d^2}{d\,t^2}\left[2\,e^{-2\,t} - 7\,e^{-t} + 8 - 5\,t + \tfrac{3}{2}\,t^2\right] + 3\,\frac{d}{d\,t}\left[2\,e^{-2\,t} - 7\,e^{-t} + 8 - 5\,t + \tfrac{3}{2}\,t^2\right] + 2\left[2\,e^{-2\,t} - 7\,e^{-t} + 8 - 5\,t + \tfrac{3}{2}\,t^2\right]$$

$$= \left[8\,e^{-2\,t} - 7\,e^{-t} + 3\right] + 3\left[-4\,e^{-2\,t} + 7\,e^{-t} - 5 + 3\,t\right] + 2\left[2\,e^{-2\,t} - 7\,e^{-t} + 8 - 5\,t + \tfrac{3}{2}\,t^2\right]$$

$$= 8\,e^{-2\,t} - 7\,e^{-t} + 3 - 12\,e^{-2\,t} + 21\,e^{-t} - 15 + 9\,t + 4\,e^{-2\,t} - 14\,e^{-t} + 16 - 10\,t + 3\,t^2$$

$$= 4 - t + 3\,t^2$$

This is exactly the right hand side that we expected, so this satisfies the ODE

We can verify that is also satisfies the ICs

$$x(0) = \left[2\,e^{-2\,t} - 7\,e^{-t} + 8 - 5\,t + \tfrac{3}{2}\,t^2\right]\,|_{\,t=0}$$

$$= 2\,e^{-2*0} - 7\,e^{-0} + 8 - 5*0 + \frac{3}{2}\times 0^2$$

$$= 2 - 7 + 8$$

$$x(0) = 3$$

For velocity

$$\dot{x}(0) = \left[-4\,e^{-2\,t} + 7\,e^{-t} - 5 + 3\,t\right]\,|_{t=0}$$

$$= -4\,e^{-2*0} + 7\,e^{-0} - 5 + 3*0$$

$$= -4 + 7 - 5$$

$$\dot{x}(0) = -2$$

```
D[x[t], {t, 2}] + 3 D[x[t], t] + 2 x[t] == 4 - t + 3 t^2 // Simplify
x[0] == 3
(D[x[t], t] /. {t → 0}) == -2
```

```
True
```

```
True
```

```
True
```

Check this solution using ‘DSolve’

```
temp = DSolve[{z''[t] + 3 z'[t] + 2 z[t] == 4 - t + 3 t^2, {z[0] == 3, z'[0] == -2}}, z[t], t];
xDSolve[t_] = z[t] /. temp[[1]]
```

$$\frac{1}{2}\,e^{-2\,t}\left(4 - 14\,e^{t} + 16\,e^{2\,t} - 10\,e^{2\,t}\,t + 3\,e^{2\,t}\,t^2\right)$$

```
x[t] == xDSolve[t] // Simplify
```

```
True
```

```
Clear[xDSolve, x]
```

ODEs can be more general. For example, consider Bessel’s equation

$$t^2\,\frac{d^2\,x(t)}{d\,t^2} + t\,\frac{d\,x(t)}{d\,t} + \left(t^2 - v^2\right)x(t) = 0 \qquad \textbf{\textit{(Eq.2)}}$$

where v = given real number which is positive or zero.

In this case of integer n, a particular solution of Bessel’s equation is (see text for derivation)

$$J_n(t) = t^n \sum_{m=0}^{\infty} \frac{(-1)^m\,t^{2\,m}}{2^{2\,m+n}\,m!\,(n+m)!}\; n \geq 0,\, n \in \mathrm{Z}$$

There are **Bessel function of the first kind of order *n***.

```
x[t_] = BesselJ[n, t];
FullSimplify[t^2 D[x[t], {t, 2}] + t D[x[t], t] + (t^2 - n^2) x[t], {n ∈ Integers}] == 0
```

```
True
```

```
Plot[{(x[t] /. n → 0), (x[t] /. n → 1), (x[t] /. n → 2)}, {t, 0, 35}, PlotRange → All]
```



Another solution is function is **Bessel functions of the second kind** $Y_v(x)$.

$$Y_0(t) = \frac{2}{\pi}\left[J_0(t)\left(\ln\left(\frac{t}{2}\right) + \gamma\right) + \sum_{m=0}^{\infty} \frac{(-1)^{m-1} h_m}{2^{2m} (m!)^2} t^{2m}\right] \qquad \textbf{(Eq.6)}$$

where $h_m = 1 + \frac{1}{2} + ... + \frac{1}{m}$

γ = Euler-Mascheroni constant and is defined as

$\gamma = \lim_{n\to\infty}\left(\sum_{k=1}^{\infty} \frac{1}{k} - \ln(n)\right)$ the limit of $1 + \frac{1}{2} + ... + \frac{1}{s} - \ln(s) \approx 0.57721566490$

```
x2[t_] = BesselY[n, t]
FullSimplify[t^2 D[x2[t], {t, 2}] + t D[x2[t], t] + (t^2 - n^2) x2[t], {n ∈ Integers}] == 0
```

```
BesselY[n, t]
```

```
True
```

```
Plot[{(x2[t] /. n → 0), (x2[t] /. n → 1), (x2[t] /. n → 2)}, {t, 0, 35}]
```



In a more general format, ODEs can be expressed as

$$F\left(t, x, \frac{dx}{dt}, \frac{d^2 x}{dt}, \ldots, \frac{d^n x}{dt^n}\right) = 0$$

Examples of ODEs

Mass/Spring/Damper

Dropping a ball

$$F = m\,a$$

$$F = m\,\ddot{x}(t)$$

If we only model gravity, we have

$$m\,g = m\,\ddot{x}(t)$$

$$g = \ddot{x}(t)$$

$$\int_0^t g\,d\tau = \int_0^t \ddot{x}(\tau)\,d\tau$$

$$g\,t = \dot{x}(t) - \dot{x}(0)$$

$$g\,t + \dot{x}(0) = \dot{x}(t)$$

$$\int_0^t g\,\tau + \dot{x}(0)\,d\tau = \int_0^t \dot{x}(\tau)\,d\tau$$

$$\frac{1}{2} g t^2 + \dot{x}(0)\, t = x(t) - x(0)$$

$$\frac{1}{2} g t^2 + \dot{x}(0)\, t + x(0) = x(t)$$

In this case, we were able to solve the ODE by simply integrating. However, what if there were more complicated forces on the ball? For example, what if there was drag that was proportional to velocity? How about if there was a spring force?

Talk about how the MBL which is really just a fancy mass/spring/damper system (the magnet force replaces the spring force)

2-Tank Hydraulic System

Do demo. This is an example of a coupled set of ODEs

Other ODEs

Talk about RCAM w/ Joystick and X-Plane

Roadmap of ODE Lectures

Roadmap

- Introduction to Ordinary Differential Equations.
- The Laplace Method
- Complex Numbers, Complex Variables, and Complex Functions
- The Laplace Transform
- Finding Roots
- Final Value Theorem
- Partial Fraction Expansion/Decomposition
- The Inverse Laplace Transform
- The Traditional Method
- Linear Homogeneous ODEs
- Linear Nonhomogeneous ODEs
- Standard 2nd Order ODEs
- State Space Representation of ODEs
- Solutions to State Space Systems
- The Matrix Exponential
- Working with ODEs Numerically
- Getting Started with Simulink
- ODEs & Dynamic Systems in Simulink
- Interacting with a Simulink Model from a Matlab Script
- Numerically Linearizing a Dynamic System

This entire playlist entitled ‘Ordinary Differential Equations’ is located at https://www.youtube.com/-playlist?list=PLxdnSsBqCrrHHvoFPxWq4l9D93jkCNIFN