Christopher Lum
lum@uw.edu

Lecture 03d
Standard 2nd Order ODEs: Natural Frequency and Damping Ratio

Lecture is on YouTube

The YouTube video entitled ‘Standard 2nd Order ODEs: Natural Frequency and Damping Ratio’ that covers this lecture is located at https://youtu.be/eJMf9CYHr6c.

Outline

-Mathematical Modeling of Simple Mechanical Systems
-2nd Order ODEs in Standard Form

Mathematical Modeling of Simple Mechanical Systems

We can start building mathematical models of systems. For the case of mechanical systems, the equations of motion typically are derived from Newton's laws.

Some definitions

Rigid Body: No internal deflections, the body does not deform, center of mass does not move relative to the body itself

Newton’s Laws: 1: Conservation of momentum
2: $\overline{F} = \frac{d}{dt}\big|_i \left[m\,\overline{V}\right]$ ($\overline{F} = m\,\overline{a}$ is a reasonable approximation)
3. action/reaction

Let us examine the 2nd law. If the mass is constant, we can write this as

$$\overline{F} = m\,\overline{a}$$

We need to be slightly more careful with rotation. Newton's law for rotational systems Newton's 2nd law is

$$\overline{T} = J\,\overline{\alpha}$$

where $\overline{T}$ = vector sum of all torques

J = inertia matrix

$\overline{\alpha}$ = angular acceleration vector

In the case of rotation, the inertia term is an entire matrix (as opposed to a scalar mass for the translational case). This accounts for the fact that rotation about a given axis can create rotation in another axis (talk about aircraft example).

Effects of Spring Elements

For translational systems, let's look at the effects of spring elements. Let us examine an example of a mass attached to a spring



Let us first draw a free body diagram of the system at rest



We can then apply Newton's second law for this system

$F = m\,a$ note: since system is at rest, $a = 0$

$m\,g - k\,\delta = 0$ note: positive direction is down

$m\,g = k\,\delta$ ***(Eq.A)***

We can now draw a free body diagram of the system at a positive deflection $y(t)$ (which is measured from the rest position)



Applying Newton's second law for this system yields

$F = m\, a$ note $a = \ddot{y}(t)$

$m\, g - k(\delta + y(t)) = m\, \ddot{y}(t)$

$m\, g - k\, \delta - k\, y(t) = m\, \ddot{y}(t)$ recall from Eq.A: $m\, g = k\, \delta$

$-k\, y(t) = m\, \ddot{y}(t)$

$m\, \ddot{y}(t) + k\, y(t) = 0$ ***(Eq.B)***

Note, it is interesting that the gravity does not play a part. This is because the spring is linear, we will see later that if the spring is non-linear, gravity does affect the dynamics.

This looks like a second order, homogeneous (unforced system). We can solve for $y(t)$ using the Laplace method.

$L\left[m\, \ddot{y}(t) + k\, y(t)\right] = L[0]$

$m\left(s^2\, Y(s) - s\, y(0) - \dot{y}(0)\right) + k\, Y(s) = 0$ note: $y(0) = y_0$, $\dot{y}(0) = \dot{y}_0$

$Y(s) = \frac{\dot{y}_0}{s^2 + k/m} + \frac{s\, y_0}{s^2 + k/m}$

We can rewrite this in a form which we can find in the Laplace transform table

$Y(s) = \sqrt{\frac{m}{k}}\ \dot{y}_0\, \frac{\sqrt{k/m}}{s^2 + \left(\sqrt{k/m}\right)^2} + y_0\, \frac{s}{s^2 + \left(\sqrt{k/m}\right)^2}$

Noting that $L\left[\sin\left(\sqrt{k/m}\ t\right)\right] = \frac{\sqrt{k/m}}{s^2 + \left(\sqrt{k/m}\right)^2}$

$L\left[\cos\left(\sqrt{k/m}\ t\right]\right. = \frac{s}{s^2 + \left(\sqrt{k/m}\right)^2}$

We can perform the inverse Laplace transform to obtain

$$y(t) = \sqrt{\frac{m}{k}}\ \dot{y}_0 \sin\left(\sqrt{k/m}\ t\right) + y_0 \cos\left(\sqrt{k/m}\ t\right)$$

```
y[t_] = √(m / k) ydot0 Sin[√(k / m) t] + y0 Cos[√(k / m) t]
```

$$\text{y0 Cos}\left[\sqrt{\frac{k}{m}}\ t\right] + \sqrt{\frac{m}{k}}\ \text{ydot0 Sin}\left[\sqrt{\frac{k}{m}}\ t\right]$$

We can verify it satisfies the initial conditions

```
y[0]
Simplify[D[y[t], t] /. {t → 0}, {m > 0, k > 0}]
```

```
y0
```

```
ydot0
```

We can now verify that it satisfies the original differential equation.

```
Simplify[m D[y[t], {t, 2}] + k y[t]] == 0
```

```
True
```

So the response of the system appears as

```
Plot[y[t] /. {m → 1, k → 1, ydot0 → 1, y0 → 2.2}, {t, 0, 15}]
```



If we look at this as

$$y(t) = a\cos(x) + b\sin(x)$$

where $a = y_0$

$b = \sqrt{m/k}\ \dot{y}_0$

$x = \sqrt{k/m}\ t$

```
a = y0;
b = √(m / k) ydot0;
x = √(k / m) t;
```

We can use the trig identity relating linear combinations of cos and sin functions.

$$a\cos(x) + b\sin(x) = A\cos(x - \varphi)$$

where $\varphi = \tan^{-1}(b/a)$ (note: be sure to use the 4 quadrant inverse tangent)

$A = \sqrt{a^2 + b^2}$

```
φ = ArcTan[a, b];
A = Sqrt[a^2 + b^2];
ycheck[t_] = A Cos[x - φ];

(*Check if this is the same*)
ycheck[t] == y[t] // Simplify
```

```
True
```

So we see they are identical and therefore, an alternate expression for $y(t)$ is

$$y(t) = A\cos(\omega t - \varphi)$$

where $A = \sqrt{{y_0}^2 + \frac{m\,{\dot{y}_0}^2}{k}}$

$\omega = \sqrt{k/m}$

$\varphi = \tan^{-1}\left(\sqrt{\frac{m}{k}}\ \dot{y}_0/y_0\right)$ (be sure to use 4-quadrant inverse tangent)

So we see the response is a sinusoidal function whose magnitude, frequency, period, and phase are all functions of $\dot{y}_0$, y_0, k, and m. For example, we can compute the period, T, of the sine wave. Recall this is the time required for the oscillation to repeat itself

$$\omega T = 2\pi$$

$$T = \frac{2\pi}{\sqrt{k/m}}$$

The frequency (in Hz) is simply (some conversion is necessary if you want units in radians/sec)

$$f = 1/T = \frac{\sqrt{k/m}}{2\pi}$$

This is the frequency of oscillation fo the system. In this case where there is no damping, you have see this frequency referred to as the undamped natural frequency. We will soon investigate how the frequency of oscillation changes if we introduce a damper.

$$\omega_n = 2\pi f = \sqrt{k/m} \qquad \text{(natural frequency)}$$

So, what does increasing k or m do to the system response? Have someone come to the board and draw the response and mark the appropriate points. What does this tell you about losses introduced through a spring?

Recall that previously, the EOM was $m\,\ddot{y}(t) + k\,y(t) = 0$, since $\omega_n = \sqrt{k/m}$ this often be rewritten in the form

$$\ddot{y}(t) + \omega_n^2\,y(t) = 0$$

2nd Order ODEs in Standard Form

Let us investigate how a mass/spring/damper system can be modeled using a 2nd order ODE.



If the mass is pulled down and let go, what would happen? This depends on how much damping and spring stiffness is in the system.

1. If *b* is small compared to *k*, a lot of vibration will occur (underdamped)
2. If *b* is large compared to *k*, the mass will go back to original position slowly, with no oscillation (overdamped)
3. If *b* and *k* are a good trade-off, mass will go back to original position quickly with little oscillations (critically damped)

Let us draw a free body diagram of system at a positive deflection $+x(t)$ and with a positive velocity $+\dot{x}(t)$



So equation of motion from Newton's second law is

$$\Sigma F = m\,a$$

$$-k\,x - b\,\dot{x} = m\,\ddot{x}$$

$$m\,\ddot{x} + b\,\dot{x} + k\,x = 0$$

$$\ddot{x} + \frac{b}{m}\dot{x} + \frac{k}{m}x = 0$$

$$\ddot{x} + 2\zeta\omega_n\dot{x} + \omega_n^2 x = 0 \qquad \textbf{\textit{(Eq.1)}}$$

where $\omega_n = \sqrt{k/m}$

$\zeta = \frac{b}{2\sqrt{k m}}$

Eq.1 is a 2nd order ODE in standard form.

We can apply the Laplace Transform to Eq.1 to obtain

$$\left(s^2 X(s) - s\,x(0) - \dot{x}(0)\right) + 2\zeta\omega_n(s\,X(s) - x(0)) + \omega_n^2 X(s) = 0$$

```
temp = Solve[(s^2 X - s x0 - xdot0) + 2 ξ ωn (s X - x0) + ωn^2 X == 0, X];
X[s_] = X /. temp[[1]]
Clear[temp]
```

$$\frac{s\,x0 + xdot0 + 2\,x0\,\xi\,\omega n}{s^2 + 2\,s\,\xi\,\omega n + \omega n^2}$$

So we have

$$X(s) = \frac{s\,x_0 + \dot{x}_0 + 2\,x_0\,\zeta\,\omega_n}{s^2 + 2\zeta\omega_n s + \omega_n^2}$$

Recall that at this point, in order to do the partial fraction expansion, we need to know something about the roots. The roots of the characteristic equation are given by

$$\frac{-b \pm \sqrt{b^2 - 4ac}}{2a} = \frac{-2\zeta\omega_n \pm \sqrt{(2\zeta\omega_n)^2 - 4\omega_n^2}}{2}$$

We are most interested in when the roots are real and distinct, real and repeated, or imaginary

1. $(2\zeta\omega_n)^2 > 4\omega_n^2 \Longleftrightarrow \zeta^2 > 1 \Rightarrow$ distinct real roots (overdamped)
2. $(2\zeta\omega_n)^2 = 4\omega_n^2 \Longleftrightarrow \zeta^2 = 1 \Rightarrow$ repeated real roots (critically damped)
3. $(2\zeta\omega_n)^2 < 4\omega_n^2 \Longleftrightarrow \zeta^2 < 1 \Rightarrow$ complex roots (under damped)

1. Overdamped

In this situation, we have $\zeta > 1$, so the poles are all real.

```
temp = Solve[s^2 + 2 ζ ωn s + ωn^2 == 0, s] // Simplify;
root1 = s /. temp[[1]]
root2 = s /. temp[[2]]
Clear[temp]
```

$$-\zeta\,\omega n - \sqrt{\left(-1+\zeta^2\right)\,\omega n^2}$$

$$-\zeta\,\omega n + \sqrt{\left(-1+\zeta^2\right)\,\omega n^2}$$

The partial fraction expansion is

$$\frac{s\,x_0+\dot{x}_0+2\,x_0\,\zeta\,\omega_n}{s^2+2\,\zeta\,\omega_n\,s+\omega_n{}^2} = \frac{a_1}{(s+r_1)} + \frac{a_2}{(s+r_2)}$$

where $r_1 = \zeta\,\omega_n + \sqrt{\left(\zeta^2-1\right)\omega_n{}^2}$

$$r_2 = \zeta\,\omega_n - \sqrt{\left(\zeta^2-1\right)\omega_n{}^2}$$

```
r1 = -root1;
r2 = -root2;
(s + r1) (s + r2) // Expand
```

$$s^2 + 2\,s\,\zeta\,\omega n + \omega n^2$$

We can now solve for the coefficients

$$\frac{s\,x_0+\dot{x}_0+2\,x_0\,\zeta\,\omega_n}{s^2+2\,\zeta\,\omega_n\,s+\omega_n{}^2} = \frac{a_1(s+r_2)+a_2(s+r_1)}{(s+r_1)\,(s+r_2)}$$

$$= \frac{a_1\,s+a_2\,s+a_1\,r_2+a_2\,r_1}{(s+r_1)\,(s+r_2)}$$

$$= \frac{(a_1+a_2)\,s+(a_1\,r_2+a_2\,r_1)}{(s+r_1)\,(s+r_2)}$$

```
Clear[a1, a2]
temp = Solve[{a1 + a2 == x0, a1 r2 + a2 r1 == xdot0 + 2 x0 ζ ωn}, {a1, a2}];
a1 = a1 /. temp[[1]] // FullSimplify
a2 = a2 /. temp[[1]] // FullSimplify
Clear[temp]
```

$$-\frac{\text{xdot0} + \text{x0}\,\zeta\,\omega n - \text{x0}\,\sqrt{\left(-1+\zeta^2\right)\,\omega n^2}}{2\,\sqrt{\left(-1+\zeta^2\right)\,\omega n^2}}$$

$$\frac{\text{xdot0} + \text{x0}\,\left(\zeta\,\omega n + \sqrt{\left(-1+\zeta^2\right)\,\omega n^2}\right)}{2\,\sqrt{\left(-1+\zeta^2\right)\,\omega n^2}}$$

So we have

$$a_1 = \frac{-\dot{x}_0 + x_0\left(-\zeta\,\omega_n + \sqrt{\left(\zeta^2-1\right)\omega_n^{\,2}}\right)}{2\sqrt{\left(\zeta^2-1\right)\omega_n^{\,2}}}$$

$$a_2 = \frac{\dot{x}_0 + x_0\left(\zeta\,\omega_n + \sqrt{\left(\zeta^2-1\right)\omega_n^{\,2}}\right)}{2\sqrt{\left(\zeta^2-1\right)\omega_n^{\,2}}}$$

So we have perform the inverse Laplace transform and we have

$$x(t) = a_1\, e^{r_1 t} + a_2\, e^{r_2 t}$$

And we once again note that since r_1 and r_2 are real, this system cannot oscillate.

```
x1[t_] = InverseLaplaceTransform[a1/(s + r1) + a2/(s + r2), s, t] // FullSimplify
```

$$e^{-t\,\zeta\,\omega n}\left(x0\,\mathrm{Cosh}\left[t\sqrt{-1+\zeta^2}\;\omega n\right] + \frac{(xdot0 + x0\,\zeta\,\omega n)\,\mathrm{Sinh}\left[t\sqrt{-1+\zeta^2}\;\omega n\right]}{\sqrt{-1+\zeta^2}\;\omega n}\right)$$

Once again, we can verify this satisfies the initial conditions and the original differential equation.

```
x1[0]
Simplify[D[x1[t], t] /. {t → 0}, {ζ > 1}]
Simplify[D[x1[t], {t, 2}] + 2 ζ ωn D[x1[t], t] + ωn^2 x1[t]]
Plot[x1[t] /. {ωn → 0.6, ζ → 1.2, x0 → 2, xdot0 → 0}, {t, 0, 10}]
```

x0

xdot0

0



2. Critically Damped

In this situation, we have $\zeta = 1$

So the equation for $X(s)$ becomes

$$X(s) = \frac{x_0\, s + \dot{x}_0 + 2\, x_0\, \omega_n}{s^2 + 2\, s\, \omega_n + \omega_n^{\,2}}$$

The partial fraction expansion is

$$\frac{x_0\, s + \dot{x}_0 + 2\, x_0\, \omega_n}{s^2 + 2\, s\, \omega_n + \omega_n^{\,2}} = \frac{a_1}{(s+\omega_n)} + \frac{a_2}{(s+\omega_n)^2}$$

$$= \frac{a_1 (s+\omega_n) + a_2}{(s+\omega_n)^2}$$

$$= \frac{a_1\, s + (a_1\, \omega_n + a_2)}{(s+\omega_n)^2}$$

```
Clear[a1, a2]
temp = Solve[{a1 == x0, a1 ωn + a2 == xdot0 + 2 x0 ωn}, {a1, a2}];
a1 = a1 /. temp[[1]]
a2 = a2 /. temp[[1]]
Clear[temp]
```

```
x0
```

```
xdot0 + x0 ωn
```

So we have

$$\frac{x_0\, s + \dot{x}_0 + 2\, x_0\, \omega_n}{s^2 + 2\, s\, \omega_n + \omega_n^{\,2}} = \frac{x_0}{(s+\omega_n)} + \frac{\dot{x}_0 + x_0\, \omega_n}{(s+\omega_n)^2}$$

```
a1/(s + ωn) + a2/(s + ωn)^2 // Together
```

$$\frac{s\ x0 + xdot0 + 2\ x0\ \omega n}{(s + \omega n)^2}$$

So the solution is

```
x2[t_] = InverseLaplaceTransform[a1/(s + ωn) + a2/(s + ωn)^2, s, t]
```

$$e^{-t\, \omega n}\ (x0 + t\ xdot0 + t\ x0\ \omega n)$$

```
Collect[Expand[x2[t]], Exp[-t ωn] t]
```

$$e^{-t\, \omega n}\ x0 + e^{-t\, \omega n}\ t\ (xdot0 + x0\ \omega n)$$

In a more easy to interpret form, we have

$$x(t) = x_0\, e^{-\omega_n\, t} + \left(\dot{x}_0 + x_0\, \omega_n\right) t\, e^{-\omega_n\, t}$$

So once again, we see that this system might initially increases or decrease due to the $t\, e^{-\omega_n\, t}$ term, but eventually it must go to zero (assuming $\omega_n \geq 0$) and it cannot oscillate.

We can verify this satisfies the initial conditions and the original differential equation.

```
x2[0]
D[x2[t], t] /. {t → 0}
Simplify[D[x2[t], {t, 2}] + 2 ζ ωn D[x2[t], t] + ωn^2 x2[t] /. {ζ → 1}]
Plot[x2[t] /. {x0 → 2, xdot0 → 3, ωn → 1}, {t, 0, 10}]
```

x0

xdot0

0



3. Underdamped

In this situation, we have $\zeta < 1$, so the poles are imaginary.

```
temp = Solve[s^2 + 2 ζ ωn s + ωn^2 == 0, s] // Simplify;
root1 = s /. temp[[1]]
root2 = s /. temp[[2]]
Clear[temp]
```

$-\zeta\,\omega n - \sqrt{\left(-1 + \zeta^2\right)\,\omega n^2}$

$-\zeta\,\omega n + \sqrt{\left(-1 + \zeta^2\right)\,\omega n^2}$

The partial fraction expansion is done by completing the squares of the denominator. We would like it to look like

$$s^2 + 2\,\zeta\,\omega_n\,s + \omega_n{}^2 = s^2 + d\,s + e = (s + \alpha)^2 + \omega^2$$

We can calculate α and ω using

$$\alpha = d/2$$
$$\omega = \sqrt{4\,e - d^2}\Big/2$$

```
Clear[α, ω, d, e]
d = 2 ζ ωn;
e = ωn^2;
α = d / 2
ω = Sqrt[4 e - d^2] / 2 // Simplify
```

$\zeta\,\omega n$

$\sqrt{-\left(-1+\zeta^2\right)\,\omega n^2}$

So we have

$$\alpha = \zeta\,\omega_n$$
$$\omega = \sqrt{\left(1-\zeta^2\right)\omega_n{}^2}$$
recall: in this case $\zeta < 1$, so term in square root is positive, so $\omega \in \mathbb{R}$

Let us verify this

```
(s + α)^2 + ω^2 == s^2 + 2 ζ ωn s + ωn^2 // Simplify
```

```
True
```

So we can write this as

$$\frac{s\,x_0+\dot{x}_0+2\,x_0\,\zeta\,\omega_n}{s^2+2\,\zeta\,\omega_n\,s+\omega_n{}^2} = \frac{x_0\,s+\dot{x}_0+2\,x_0\,\zeta\,\omega_n}{(s+\alpha)^2+\omega^2}$$

$$= \frac{x_0\,s+\dot{x}_0+2\,x_0\,\zeta\,\omega_n}{(s+\zeta\,\omega_n)^2+\omega^2}$$

$$= x_0\left(\frac{s+\frac{\dot{x}_0}{x_0}+2\,\zeta\,\omega_n}{(s+\zeta\,\omega_n)^2+\omega^2}\right)$$
note: add and subtract $\left(\frac{\dot{x}_0}{x_0}+\zeta\,\omega_n\right)$ to numerator

$$= x_0\left(\frac{s+\frac{\dot{x}_0}{x_0}+2\,\zeta\,\omega_n-\left(\frac{\dot{x}_0}{x_0}+\zeta\,\omega_n\right)+\left(\frac{\dot{x}_0}{x_0}+\zeta\,\omega_n\right)}{(s+\zeta\,\omega_n)^2+\omega^2}\right)$$

$$= x_0\left(\frac{s+\zeta\,\omega_n+\left(\frac{\dot{x}_0}{x_0}+\zeta\,\omega_n\right)}{(s+\zeta\,\omega_n)^2+\omega^2}\right)$$

$$= x_0\left(\frac{s+\zeta\,\omega_n}{(s+\zeta\,\omega_n)^2+\omega^2}+\frac{\left(\frac{\dot{x}_0}{x_0}+\zeta\,\omega_n\right)}{(s+\zeta\,\omega_n)^2+\omega^2}\right)$$

```
(x0 s + xdot0 + 2 x0 ξ ωn)/(s^2 + 2 ξ s ωn + ωn^2) == x0 ((s + ξ ωn)/((s + ξ ωn)^2 + ω^2) + (xdot0/x0 + ξ ωn)/((s + ξ ωn)^2 + ω^2)) // FullSimplify
```

```
True
```

We can now inverse Laplace transform this

$$x(t) = x_0\left(L^{-1}\left[\frac{s+\alpha}{(s+\alpha)^2+\omega^2}\right] + \frac{\beta}{\omega} L^{-1}\left[\frac{\omega}{(s+\alpha)^2+\omega^2}\right]\right) \quad \text{note: } \beta = \frac{\dot{x}_0}{x_0} + \zeta\,\omega_n$$

$$= x_0\left(e^{-\alpha t}\cos(\omega t) + \frac{\beta}{\omega} e^{-\alpha t}\sin(\omega t)\right)$$

$$= x_0\left(e^{-\alpha t}\cos(\omega t) + \frac{\frac{\dot{x}_0}{x_0}+\zeta\,\omega_n}{\omega} e^{-\alpha t}\sin(\omega t)\right)$$

$$= x_0\, e^{-\alpha t}\cos(\omega t) + \frac{\dot{x}_0+x_0\,\zeta\,\omega_n}{\omega} e^{-\alpha t}\sin(\omega t)$$

$$x(t) = a\cos(\omega t) + b\sin(\omega t)$$

where $a = x_0\, e^{-\alpha t}$

$b = \frac{\dot{x}_0+x_0\,\zeta\,\omega_n}{\omega} e^{-\alpha t}$

$\alpha = \zeta\,\omega_n$

$\omega = \sqrt{-(\zeta^2-1)\,\omega_n^{\,2}}$

```
a = x0 Exp[-α t];
b = (xdot0 + ξ ωn x0)/ω Exp[-α t];
x3[t_] = a Cos[ω t] + b Sin[ω t];
```

We can verify this satisfies the original differential equation of $\ddot{x}(t) + 2\,\zeta\,\omega_n\,\dot{x}(t) + \omega_n^{\,2} = 0$

```
D[x3[t], {t, 2}] + 2 ξ ωn D[x3[t], t] + ωn^2 x3[t] == 0 // Simplify
```

```
True
```

And we can verify that it satisfies the initial conditions.

```
x3[0]
Simplify[D[x3[t], t] /. {t → 0}, {ξ < 1}]
```

```
x0
```

```
xdot0
```

Let us choose some values and plot this.

```
ωnplot = 1.6;
ζplot = 0.22;
x0plot = 1.25;
xdot0plot = 0;
Plot[x3[t] /. {ωn → ωnplot, ζ → ζplot, x0 → x0plot, xdot0 → xdot0plot},
 {t, 0, 10}, PlotRange → All]
```



A perhaps more useful form of the equation is

$$x(t) = x_0\, e^{-\zeta\,\omega_n\, t} \cos\left(\sqrt{\left(1-\zeta^2\right)\omega_n{}^2}\; t\right) + \frac{\dot{x}_0 + x_0\,\zeta\,\omega_n}{\sqrt{\left(1-\zeta^2\right)\omega_n{}^2}}\, e^{-\zeta\,\omega_n\, t} \sin\left(\sqrt{\left(1-\zeta^2\right)\omega_n{}^2}\; t\right)$$

$$= x_0\, e^{-\zeta\,\omega_n\, t} \cos\left(\omega_n\,\sqrt{\left(1-\zeta^2\right)}\; t\right) + \frac{\dot{x}_0 + x_0\,\zeta\,\omega_n}{\omega_n\,\sqrt{\left(1-\zeta^2\right)}}\, e^{-\zeta\,\omega_n\, t} \sin\left(\omega_n\,\sqrt{\left(1-\zeta^2\right)}\; t\right)$$

$$x(t) = e^{-\zeta\,\omega_n\, t}\left(x_0\, \cos\left(\omega_n\,\sqrt{\left(1-\zeta^2\right)}\; t\right) + \frac{\dot{x}_0 + x_0\,\zeta\,\omega_n}{\omega_n\,\sqrt{\left(1-\zeta^2\right)}} \sin\left(\omega_n\,\sqrt{\left(1-\zeta^2\right)}\; t\right)\right)$$

Once again, if we look at $x(t)$ as

$$x(t) = e^{-\zeta\,\omega_n\, t}(a\cos(z) + b\sin(z))$$

where $a = x_0$

$$b = \frac{\dot{x}_0 + x_0\,\zeta\,\omega_n}{\omega_n\,\sqrt{\left(1-\zeta^2\right)}}$$

$$z = \omega_n\,\sqrt{\left(1-\zeta^2\right)}\; t$$

```
a = x0;
b = (xdot0 + x0 ξ ωn) / (ωn Sqrt[(1 - ξ^2)]);
z = ωn Sqrt[(1 - ξ^2)] t;
Simplify[x3[t] == Exp[-ξ ωn t] (a Cos[z] + b Sin[z]), {ωn > 0}]
```

True

We can use the trig identity relating linear combinations of cos and sin functions to write the term (a cos(z)+b sin(z)) as a single sinusoid.

$$a\cos(z) + b\sin(z) = A\cos(z - \phi)$$

where $\phi = \tan^{-1}(b/a)$ (note: be sure to use the 4 quadrant inverse tangent)

$A = \sqrt{a^2 + b^2}$

```
Print["ϕ"]
ϕ = ArcTan[a, b]

Print["A"]
A = Sqrt[a^2 + b^2]

Print["Solution"]
x3check[t_] = Simplify[Exp[-ξ ωn t] (A Cos[z - ϕ]), {ωn > 0, ξ > 0, ξ < 1}]
```

ϕ

$$\text{ArcTan}\left[\text{x0}, \frac{\text{xdot0} + \text{x0}\,\xi\,\omega\text{n}}{\sqrt{1-\xi^2}\;\omega\text{n}}\right]$$

A

$$\sqrt{\text{x0}^2 + \frac{(\text{xdot0} + \text{x0}\,\xi\,\omega\text{n})^2}{(1-\xi^2)\;\omega\text{n}^2}}$$

Solution

$$e^{-t\,\xi\,\omega\text{n}}\sqrt{\text{x0}^2 - \frac{(\text{xdot0} + \text{x0}\,\xi\,\omega\text{n})^2}{(-1+\xi^2)\;\omega\text{n}^2}}\;\text{Cos}\left[t\sqrt{1-\xi^2}\;\omega\text{n} - \text{ArcTan}\left[\text{x0}\sqrt{1-\xi^2}\;\omega\text{n}, \text{xdot0} + \text{x0}\,\xi\,\omega\text{n}\right]\right]$$

Let us verify this trig identity is correct by checking that it satisfies the initial conditions and the original equation

```
Simplify[x3check[0], {ωn > 0, ξ > 0, ξ < 1}]
Simplify[D[x3check[t], t] /. {t → 0}, {ωn > 0, ξ > 0, ξ < 1}]
Simplify[D[x3check[t], {t, 2}] + 2 ξ ωn D[x3check[t], t] + ωn^2 x3check[t]]
```

```
x0
```

```
xdot0
```

```
0
```

We can plot the two expressions using some numeric values to verify that they are the same.

```
Plot[{x3check[t] - x3[t]} /. {ωn → 23.2, ξ → 0.33, xdot0 → -23.3, x0 → 2.2}, {t, 0, 15}]
```



So we see that we can write the solution as

$$x(t) = A\, e^{-\zeta\, \omega_n\, t} \cos(\omega_d\, t - \phi) \qquad \textbf{\textit{(Eq.C.1)}}$$

where $\quad A = \sqrt{{x_0}^2 + \left(\frac{(\dot{x}_0 + x_0\, \zeta\, \omega_n)^2}{\left(1-\zeta^2\right) {\omega_n}^2}\right)}$

$$\omega_d = \omega_n\, \sqrt{1 - \zeta^2}$$

$\phi = \text{ArcTan}(x,\ y) = \text{ArcTan}\left(x_0,\ \frac{\dot{x}_0 + x_0\, \zeta\, \omega_n}{\omega_n\, \sqrt{1-\zeta^2}}\right)$ (remember to use the 4 quadrant inverse tangent)

So from this we see that the frequency of oscillation for the underdamped system is governed by the quantity

$$\omega = \omega_d = \omega_n\, \sqrt{\left(1 - \zeta^2\right)}$$

Often this is referred to as the damped natural frequency. It is important to reiterate that an under-damped system oscillates at a frequency of ω_d, not ω_n.

We can make some interesting observations of the response of the system. For example, if we look at the expression $x(t) = A\, e^{-\zeta\, \omega_n\, t} \cos\left(\omega_n\, \sqrt{1 - \zeta^2}\ t - \phi\right)$, we notice that the cos wave repeats when

$$\omega_n \sqrt{1-\zeta^2}\; t = 2\pi n \qquad\qquad n = 0, 1, 2, \ldots$$

$$t_{\text{repeat}}(n) = \frac{2\pi}{\omega_n \sqrt{1-\zeta^2}}\, n \qquad\qquad n = 0, 1, 2, \ldots$$

```
trepeat[n_] = 2 π / (ωn Sqrt[1 - ξ^2]) n;
```

Let us examine the ratio of the signal at two times

$$\eta = \frac{x(t_{\text{repeat}}(n+1))}{x(t_{\text{repeat}}(n))} = \frac{x\left(\frac{2\pi}{\omega_n \sqrt{1-\zeta^2}}(n+1)\right)}{x\left(\frac{2\pi}{\omega_n \sqrt{1-\zeta^2}}\, n\right)}$$

```
η = Simplify[x3[trepeat[n + 1]] / x3[trepeat[n]], {ωn > 0}]
```

$$e^{-\frac{2\pi\xi}{\sqrt{1-\xi^2}}}$$

So we see that the ratio of decrement is given by

$$\eta = e^{-2\pi\zeta/\sqrt{1-\zeta^2}}$$

We can even solve for ζ if we measure the ratio η

```
temp = Solve[ηmeasured == η, ξ] // Simplify
ξsolution = ξ /. temp[[3]]
```

$$\left\{\left\{\xi \to -\frac{\text{Log}\left[-\sqrt{\eta\text{measured}}\right]}{\sqrt{\pi^2 + \text{Log}\left[-\sqrt{\eta\text{measured}}\right]^2}}\right\}, \left\{\xi \to \frac{\text{Log}\left[-\sqrt{\eta\text{measured}}\right]}{\sqrt{\pi^2 + \text{Log}\left[-\sqrt{\eta\text{measured}}\right]^2}}\right\}, \left\{\xi \to -\frac{\text{Log}[\eta\text{measured}]}{\sqrt{4\pi^2 + \text{Log}[\eta\text{measured}]^2}}\right\}, \left\{\xi \to \frac{\text{Log}[\eta\text{measured}]}{\sqrt{4\pi^2 + \text{Log}[\eta\text{measured}]^2}}\right\}\right\}$$

$$-\frac{\text{Log}[\eta\text{measured}]}{\sqrt{4\pi^2 + \text{Log}[\eta\text{measured}]^2}}$$

So for example, if the measured ratio of $\eta = 0.7528$, we know this means the system has a damping ratio of $\zeta = 0.045$

```
ξsolution /. {ηmeasured → 0.7528}
```

0.0451469

So the interesting thing to notice is that if we look at the ratio of the signal at two crucial times, the decrease is only a function of ζ (it is independent of ω_n)

```
Plot[η, {ζ, 0, 1}, AxesLabel → {"ζ", "η"}]
```



We can arrive at this same result if we look at when the function touches the exponential decay curve. Using the form of the equation where $x(t) = A\, e^{-\zeta\, \omega_n\, t} \cos\left(\omega_n \sqrt{1-\zeta^2}\; t - \phi\right)$, we see that the function will equal $A\, e^{-\zeta\, \omega_n\, t}$ (touches the exponential decay curve) when the $\cos\left(\omega_n \sqrt{1-\zeta^2}\; t - \phi\right) = 1$. This occurs when

$$\omega_n \sqrt{1-\zeta^2}\; t - \phi = 2\,\pi\, n \qquad n = 0, 1, 2, \ldots$$

$$t_{\text{touch}}(n) = \frac{2\,\pi\, n + \phi}{\omega_n \sqrt{1-\zeta^2}} \qquad n = 0, 1, 2, \ldots \text{ (times when function touches decay envelope)}$$

```
ttouch[n_] = (2 π n + ϕ) / (ωn Sqrt[1 - ζ^2])
```

$$\frac{2\, n\, \pi + \text{ArcTan}\left[\text{x0}, \frac{\text{xdot0} + \text{x0}\, \zeta\, \omega\text{n}}{\sqrt{1-\zeta^2}\; \omega\text{n}}\right]}{\sqrt{1-\zeta^2}\; \omega\text{n}}$$

At these specific points, the function simply becomes

$$x(t_{\text{touch}}(n)) = A\, e^{-\zeta\, \omega_n\, t} \text{ evaluated at } t = t_{\text{touch}} \qquad n = 0, 1, 2, \ldots$$

We can look at the ratio of the $A\, e^{-\zeta\, \omega_n\, t}$ term at two successive points

$$\frac{x(t_{\text{touch}}(n+1))}{x(t_{\text{touch}}(n))} = \frac{A\, e^{-\zeta\, \omega_n\, t} \text{ evalulated at } t = t_{\text{touch}}(n+1)}{A\, e^{-\zeta\, \omega_n\, t} \text{ evalulated at } t = t_{\text{touch}}(n)}$$

$$= \frac{e^{-\zeta\, \omega_n\, t_{\text{touch}}(n+1)} \sqrt{x_0{}^2 - \frac{(\dot{x}_0 + x_0\, \zeta\, \omega_n)^2}{(\zeta^2-1)\, \omega_n{}^2}}}{e^{-\zeta\, \omega_n\, t_{\text{touch}}(n)} \sqrt{x_0{}^2 - \frac{(\dot{x}_0 + x_0\, \zeta\, \omega_n)^2}{(\zeta^2-1)\, \omega_n{}^2}}}$$

$$= \frac{e^{-\zeta\omega_n \frac{2\pi(n+1)+\phi}{\omega_n\sqrt{1-\zeta^2}}}}{e^{-\zeta\omega_n \frac{2\pi n+\phi}{\omega_n\sqrt{1-\zeta^2}}}}$$

$$= e^{-\zeta\omega_n \frac{2\pi(n+1)+\phi}{\omega_n\sqrt{1-\zeta^2}} + \zeta\omega_n \frac{2\pi n+\phi}{\omega_n\sqrt{1-\zeta^2}}}$$

$$= e^{-\zeta\omega_n \frac{2\pi n+2\pi+\phi}{\omega_n\sqrt{1-\zeta^2}} + \zeta\omega_n \frac{2\pi n+\phi}{\omega_n\sqrt{1-\zeta^2}}}$$

$$= e^{-\zeta\omega_n \frac{2\pi n+\phi}{\omega_n\sqrt{1-\zeta^2}} - \zeta\omega_n \frac{2\pi}{\omega_n\sqrt{1-\zeta^2}} + \zeta\omega_n \frac{2\pi n+\phi}{\omega_n\sqrt{1-\zeta^2}}}$$

$$= e^{-\frac{2\pi\zeta\omega_n}{\omega_n\sqrt{1-\zeta^2}}}$$

$$\frac{x(t_{\max}(n+1))}{x(t_{\max}(n))} = e^{-\frac{2\pi\zeta}{1-\zeta^2}}$$

This is the same result as last time

```
η == Simplify[x3check[ttouch[n + 1]], {ωn > 0, ζ > 0, ζ < 1}] / Simplify[x3check[ttouch[n]], {ωn > 0, ζ > 0, ζ < 1}] // Simplify
```

```
True
```

Note that the decrement ratio is valid for any two points on the curve as long as they are chosen a time $T = \frac{2\pi}{\omega_n\sqrt{1-\zeta^2}}$ apart. The easiest thing to do would be to choose two maxima or two minima since they are the most obvious and have the most change. Therefore, ζ can be calculated once η is known

```
Solve[η == ηmeasured, ζ]
```

Solve::ifun : Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information. ≫

$$\left\{\left\{\zeta \to -\frac{\text{Log}\left[-\sqrt{\eta\text{measured}}\right]}{\sqrt{\pi^2 + \text{Log}\left[-\sqrt{\eta\text{measured}}\right]^2}}\right\}, \left\{\zeta \to \frac{\text{Log}\left[-\sqrt{\eta\text{measured}}\right]}{\sqrt{\pi^2 + \text{Log}\left[-\sqrt{\eta\text{measured}}\right]^2}}\right\}, \left\{\zeta \to -\frac{\text{Log}[\eta\text{measured}]}{\sqrt{4\pi^2 + \text{Log}[\eta\text{measured}]^2}}\right\}, \left\{\zeta \to \frac{\text{Log}[\eta\text{measured}]}{\sqrt{4\pi^2 + \text{Log}[\eta\text{measured}]^2}}\right\}\right\}$$

So we have (be careful to know the difference between Log and Ln with different computer applications. For example, Mathematica uses Log to denote the natural log)

$$\zeta = \frac{-\ln(\eta_{\text{measured}})}{\sqrt{4\pi^2 + \ln(\eta_{\text{measured}})^2}} \qquad \textbf{\textit{(Eq.C.1)}}$$

```
Plot[-Log[ηmeasured] / Sqrt[4 π^2 + Log[ηmeasured]^2], {ηmeasured, 0, 1}]
```



Furthermore, the period of oscillation, T, can be measured and from this we can calculate the natural frequency

$$T = \frac{2\pi}{\omega_n \sqrt{1-\zeta^2}} \qquad \textbf{(Eq.C.2)}$$

$$\omega_n = \frac{2\pi}{T \sqrt{1-\zeta^2}} \qquad \textbf{(Eq.C.3)}$$

For discussions of the velocity and acceleration see notes in AA447 lecture 15

Relationship to Root Locations

Recall that in the case of an underdamped system, the poles are given by

$$\text{roots} = \frac{-2\zeta\omega_n \pm \sqrt{(2\zeta\omega_n)^2 - 4\omega_n^2}}{2}$$

$$= \frac{-2\zeta\omega_n \pm \sqrt{4\zeta^2\omega_n^2 - 4\omega_n^2}}{2}$$

$$= \frac{-2\zeta\omega_n \pm \sqrt{4\omega_n^2(\zeta^2-1)}}{2}$$

$$= \frac{-2\zeta\omega_n \pm 2\omega_n\sqrt{\zeta^2-1}}{2}$$

$$= -\zeta\omega_n \pm \omega_n\sqrt{\zeta^2-1}$$

$$= -\zeta\omega_n \pm \omega_n\sqrt{1-\zeta^2}\, i \qquad \text{(recall } \zeta < 1 \text{ so } 1-\zeta^2 > 0 \text{ so } \sqrt{1-\zeta^2} \in \mathbb{R}$$

$$\text{roots} = -\sigma \pm \omega_d\, i$$

where $\sigma = \zeta\, \omega_n$

$\omega_d = \omega_n \sqrt{1 - \zeta^2}$

Damping Ratio

Consider pole locations along lines with angle θ. This is shown below



We can calculate the damping ratio for any pole along this line. For any pole along this line, we have the relationship

$$\tan(\theta) = \frac{\omega_d}{\sigma}$$

$$= \frac{\omega_n \sqrt{1-\zeta^2}}{\zeta\, \omega_n}$$

$$= \frac{\sqrt{1-\zeta^2}}{\zeta}$$

$$\theta = \tan^{-1}\left(\frac{\sqrt{1-\zeta^2}}{\zeta}\right) \qquad \textbf{\textit{(Eq.1)}}$$

This states that any pole along any line intersecting the origin has the same damping ratio. For example, any pole with damping of $\zeta = 1\big/\sqrt{2} \approx 0.707$ (considered optimal in many cases) must be along the line

$$\theta = \tan^{-1}\left(\frac{\sqrt{1-\left(1/\sqrt{2}\right)^2}}{1/\sqrt{2}}\right) = 45^o$$

In other words, if you want a system to have good damping of $\zeta = 1/\sqrt{2}$, the roots must be on the 45^o line.

Natural Frequency

We can also investigate the natural frequency. Consider pole locations along a circle centered at the origin with radius r



From geometry, we see that

$$r = \left(\omega_d{}^2 + \sigma^2\right)^{1/2}$$

$$= \left(\left(\omega_n \sqrt{1-\zeta^2}\right)^2 + (\zeta\,\omega_n)^2\right)^{1/2}$$

```
Simplify[((ωn Sqrt[1 - ξ^2])^2 + (ξ ωn)^2)^(1/2), ωn > 0]
```

ωn

So we see that

$$r = \omega_n$$

In other words, the radial distance from the origin dictates the natural frequency of the system.

Damped Natural Frequency

Keep in mind that the damped natural frequency, ω_d, is the frequency that the system will oscillate at. If we recall that $\omega_d = \omega_n \sqrt{1-\zeta^2}$ we see that this is simply the imaginary component of the pole.

Therefore the distance from the x-axis dictates the damped natural frequency of the system.

2nd Order System Example: Pendulum

Let us examine the classical pendulum example



We can write Newton's second law about the pivot point

$$T = J\,\ddot{\theta}(t)$$

$-m\,g\,L\sin(\theta(t)) - b\,\dot{\theta}(t) = J\,\ddot{\theta}(t)$ note: assuming string is mass-less and mass is a point mass, $J = m\,L^2$

$$0 = m\,L^2\,\ddot{\theta}(t) + b\,\dot{\theta}(t) + m\,g\,L\sin(\theta(t))$$

$$\ddot{\theta}(t) + \frac{b}{m\,L^2}\,\dot{\theta}(t) + \frac{g}{L}\sin(\theta(t)) = 0$$

Note that this is a second order, non-linear, time invariant system. Therefore, we cannot solve with the Laplace transform because of the non-linear sin term. However, if θ is small, we can use the small angle approximation of $\sin(\theta(t)) \approx \theta(t)$

$$\ddot{\theta}(t) + \frac{b}{m\,L^2}\,\dot{\theta}(t) + \frac{g}{L}\,\theta(t) = 0$$

```
Clear[b, m, L, g]
temp = Solve[{2 ζ ωn == b/(m L^2), ωn^2 == g / L}, {ζ, ωn}]

(*we want the second solution*)
ζSolution = ζ /. temp[[2]]
ωnSolution = ωn /. temp[[2]]
```

$$\left\{\left\{\zeta \to -\frac{b}{2\sqrt{g}\,L^{3/2}\,m},\ \omega n \to -\frac{\sqrt{g}}{\sqrt{L}}\right\},\ \left\{\zeta \to \frac{b}{2\sqrt{g}\,L^{3/2}\,m},\ \omega n \to \frac{\sqrt{g}}{\sqrt{L}}\right\}\right\}$$

$$\frac{b}{2\sqrt{g}\,L^{3/2}\,m}$$

$$\frac{\sqrt{g}}{\sqrt{L}}$$

So we can rewrite this in standard second order form as

$$\ddot{\theta}(t) + 2\,\zeta\,\omega_n\,\dot{\theta}(t) + \omega_n^{\,2}\,\theta(t) = 0$$

where $\omega_n = \sqrt{g/L}$

$\zeta = \frac{b}{2\,m\,L^{3/2}\,\sqrt{g}}$

At this point, we can experimentally determine the damping ratio by swinging the pendulum (using a small angle, a large angle will not have the appropriate period as we will show later).

$$\eta = e^{-2\,\pi\,\zeta/\sqrt{1-\zeta^2}}$$

or solving for ζ

$$\zeta = \frac{-\ln(\eta_{\text{measured}})}{\sqrt{4\,\pi^2 + \ln(\eta_{\text{measured}})^2}}$$

For this specific pendulum model, we can even go one step further and experimentally solve for a model specific parameter, namely b. Recall that the damping ratio was given by

$$\zeta = \frac{b}{2\,m\,L^{3/2}\,\sqrt{g}}$$

So solving for b yields

$$b = 2\,\zeta\,m\,L^{3/2}\,\sqrt{g}$$

So we can experimentally find the damping coefficient from this

1. Swing pendulum and measure decrement ratio, η
2. Calculate $\zeta = \frac{-\ln(\eta_{\text{measured}})}{\sqrt{4\pi^2+\ln(\eta_{\text{measured}})^2}}$
3. Calculate b using $b = 2\,\zeta\, m\, L^{3/2}\,\sqrt{g}$

What about if we had measured θ from vertical?

We can write Newton' s second law about the pivot point

$$T = J\,\ddot{\theta}(t)$$

$m\,g\,L\,\sin(\theta(t)) - b\,\dot{\theta}(t) = J\,\ddot{\theta}(t)$ note: assuming string is mass-less and mass is a point mass, $J = m\,L^2$

$$0 = m\,L^2\,\ddot{\theta}(t) + b\,\dot{\theta}(t) - m\,g\,L\,\sin(\theta(t))$$

$$\ddot{\theta}(t) + \frac{b}{m\,L^2}\,\dot{\theta}(t) - \frac{g}{L}\,\sin(\theta(t)) = 0$$

Once again, if θ is small, we can use the small angle approximation of $\sin(\theta(t)) \approx \theta(t)$

$$\ddot{\theta}(t) + \frac{b}{m\,L^2}\,\dot{\theta}(t) - \frac{g}{L}\,\theta(t) = 0$$

Notice that there is a sign is difference. We will investigate the ramifications of this in later discussions on stability.

<Go over Simulink model of this system.>

Nonlinear system

$$\ddot{\theta}(t) = -\frac{b}{m\,L^2}\,\dot{\theta}(t) - \frac{g}{L}\,\sin(\theta(t))$$

Linear system (down position)

$$\ddot{\theta}(t) = -\frac{b}{m\,L^2}\,\dot{\theta}(t) - \frac{g}{L}\,\theta(t) = 0$$

Let us run through an example with constants of

$m = 1$
$b = 0.1$
$L = 0.5$
$g = 9.81$

```
ζSolution /. {m → 1, b → 0.1, L → 0.5, g → 9.81}
ωnSolution /. {m → 1, b → 0.1, L → 0.5, g → 9.81}
```

```
0.0451524
```

```
4.42945
```

So we see the system is underdamped and should oscillate.

The appropriate Simulink model for both models is shown below (build models in class to give students more experience with Simulink)



We can simulate the system with different initial conditions to investigate the response and how the linear and non-linear model differ. For example, with initial conditions of

$$\theta(0) = 10\,\frac{\pi}{180}$$
$$\dot{\theta}(0) = 0$$

The two simulations are nearly identical as seen on the plot below.



However, if we increase the initial condition to $\theta(0) = 100\,\frac{\pi}{180}$, $\dot{\theta}(0) = 0$, they begin to deviate as shown below



This becomes even more extreme as we increase to $\theta(0) = 190\,\pi/180$, $\dot{\theta}(0) = 0$



We see that the nonlinear model correctly predicts that the pendulum will fall in the opposite direction before oscillating about $\theta = 2\pi$ whereas the linear model predicts unreasonable behavior. It may help us visualize what the system is doing if we animate the scenario using Matlab (animation in Matlab will be covered in a later lecture).