Christopher Lum
lum@uw.edu

Lecture 03b
Nonhomogeneous Linear Ordinary Differential Equations

The YouTube video entitled 'Nonhomogeneous Linear Ordinary Differential Equations' that covers this lecture is located at https://youtu.be/t98ILS2YdrU.

Outline

-Nonhomogeneous Linear Ordinary Differential Equations
-Method of Underdetermined Coefficients

Nonhomogeneous Linear Ordinary Differential Equations

We now investigate what happens when we have a particular solution (ie a forcing function) present in the ordinary differential equation. This system is represented as

$$a\,\ddot{x}(t) + b\,\dot{x}(t) + c\,x(t) = g(t)$$

with $x(t_0) = x_0$
$\dot{x}(t_0) = \dot{x}_0$

To make notation simpler, let us define

$$Q\,[x(t)] \triangleq a\,\ddot{x}(t) + b\,\dot{x}(t) + c\,x(t)$$

In this case, we can write the system as

$$Q\,[x(t)] = g(t) \qquad \textbf{\textit{(Eq.2.1)}}$$

with $x(t_0) = x_0$
$\dot{x}(t_0) = \dot{x}_0$

Note that in Mathematica, we can define this operator using the 'set delay' operator. We will cover this in a later lecture (likely lecture 6).

This is referred to as the non-homogeneous equation. This has a corresponding homogeneous equation of

$$a\,\ddot{x}(t) + b\,\dot{x}(t) + c\,x(t) = 0 \qquad \textbf{(Eq.2.2)}$$

Theorem: If $x_1(t)$ and $x_2(t)$ are solutions to the non-homogeneous equation, then their difference $x_1(t) - x_2(t)$ is a solution to the corresponding homogeneous equation. In addition, if $x_{1,\text{homogeneous}}(t)$ and $x_{2,\text{homogeneous}}(t)$ are a fundamental set of solutions of the homogeneous equation, then

$$x_1(t) - x_2(t) = c_1\,x_{1,\text{homogeneous}}(t) + c_2\,x_{2,\text{homogeneous}}(t) \qquad \textbf{(Eq.2.3)}$$

Proof: If $x_1(t)$ and $x_2(t)$ are solutions to the non-homogeneous equation, then

$$Q[x_1(t)] = Q[x_2(t)] = g(t) \qquad \textbf{(Eq.2.3a)}$$

So their difference is zero

$$Q[x_1(t)] - Q[x_2(t)] = g(t) - g(t) = 0 \qquad \textbf{(Eq.2.4)}$$

Let us examine the LHS of the above equation

$$Q[x_1(t)] - Q[x_2(t)] = a\,\ddot{x}_1(t) + b\,\dot{x}_1(t) + c\,x_1(t) - \left(a\,\ddot{x}_2(t) + b\,\dot{x}_2(t) + c\,x_2(t)\right)$$

$$= a\left(\frac{d^2}{dt^2}[x_1(t)] - \frac{d^2}{dt^2}[x_2(t)]\right) + b\left(\frac{d}{dt}[x_1(t)] - \frac{d}{dt}[x_2(t)]\right) + c(x_1(t) - x_2(t))$$

$$= a\left(\frac{d^2}{dt^2}[x_1(t) - x_2(t)]\right) + b\left(\frac{d}{dt}[x_1(t) - x_2(t)]\right) + c(x_1(t) - x_2(t))$$

$$Q[x_1(t)] - Q[x_2(t)] = Q[x_1(t) - x_2(t)] \qquad \textbf{(Eq.2.5)}$$

Combining Eq .2.4 and Eq.2.5 yields

$$Q[x_1(t) - x_2(t)] = 0$$

This states that the difference of the total solution is a solution to the homogeneous part of the equation (ie $a\,\ddot{x}(t) + b\,\dot{x}(t) + c\,x(t) = 0$).

We showed previously that a combination of the homogeneous solutions are also a solution to the homogeneous equation. Writing this using our Q operator, we have

$$Q\left[c_1\,x_{1,\text{homogeneous}}(t) + c_2\,x_{2,\text{homogeneous}}(t)\right] = 0$$

Therefore

$$Q[x_1(t) - x_2(t)] = Q\left[c_1\, x_{1,\text{homogeneous}}(t) + c_2\, x_{2,\text{homogeneous}}(t)\right] = 0 \qquad \textbf{\textit{(Eq.2.6)}}$$

Therefore implying that

$$x_1(t) - x_2(t) = c_1\, x_{1,\text{homogeneous}}(t) + c_2\, x_{2,\text{homogeneous}}(t)$$

We are now in the position to look at the total solution of the system

Theorem: The general (total) solution of the non-homogeneous equation can be written in the form

$$x(t) = \phi(t) = c_1\, x_{1,\text{homogeneous}}(t) + c_2\, x_{2,\text{homogeneous}}(t) + x_{\text{particular}}(t) \qquad \textbf{\textit{(Eq.2.7)}}$$

where $x_{1,\text{homogeneous}}(t)$ and $x_{2,\text{homogeneous}}(t)$ are fundamental set of solutions of the corresponding homogeneous equation

c_1 and c_2 are arbitrary constants

$x_{\text{particular}}(t)$ is some specific solution of the non-homogeneous equation

Proof: We can use the previous theorem of Eq.2.3 to show this. Recall this stated that

$$x_1(t) - x_2(t) = c_1\, x_{1,\text{homogeneous}}(t) + c_2\, x_{2,\text{homogeneous}}(t)$$

where $x_1(t)$ and $x_2(t)$ are solutions to the non-homogeneous equation of Eq.2.1. We can then call $x_1(t) = x(t) = \phi(t)$ as the general solution and then $x_2(t) = x_{\text{particular}}(t)$ as a particular solution. Eq.2.3 becomes

$$x_1(t) - x_2(t) = c_1\, x_{1,\text{homogeneous}}(t) + c_2\, x_{2,\text{homogeneous}}(t) \qquad \text{let:} \quad x_1(t) = x(t)$$

$$x_2(t) = x_{\text{particular}}(t)$$

$$x(t) - x_{\text{particular}}(t) = c_1\, x_{1,\text{homogeneous}}(t) + c_2\, x_{2,\text{homogeneous}}(t)$$

$$x(t) = c_1\, x_{1,\text{homogeneous}}(t) + c_2\, x_{2,\text{homogeneous}}(t) + x_{\text{particular}}(t)$$

Therefore, the process becomes

1. Find the general solution $c_1\, x_{1,\text{homogeneous}}(t) + c_2\, x_{2,\text{homogeneous}}(t)$ of the corresponding homogeneous equation. (AKA the complementary, homogeneous, autonomous)
2. Find some single solution $x_{\text{particular}}(t)$ of the non-homogeneous equation (AKA the particular solution)
3. Add the two function together as in Eq.2.7.

Note that we can gain some insight into the solutions of linear ordinary differential equations from this process. Namely we see that the solution to the overall system will have a portion of it that is related to the nature of the system itself (AKA the homogeneous portion) as well as a portion that is related to the input/forcing function (the particular solution).

Method of Underdetermined Coefficients

In this scenario, we assume a solution form with unspecified coefficients then try to choose coefficients which will satisfy the equation. The major limitation to this method is that we require knowledge of the what the solution will look like ahead of time. This is generally restricted to problems with the constant coefficients of the homogeneous equation and a relatively small class of non-homogeneous functions.

Example : Non - Homogeneous ODE with Exponential Forcing Function

$$\ddot{x}(t) + 3\,\dot{x}(t) + 2\,x(t) = 3\,e^{2\,t}$$

where $x(0) = 3$
$\dot{x}(0) = -2$

```
g[t_] = 3 Exp[2 t];
```

We already know how to find the homogeneous solution to this. We start by finding the solution to the characteristic equation.

```
temp = Solve[r^2 + 3 r + 2 == 0, r];
r1 = r /. temp[[1]]
r2 = r /. temp[[2]]
```

```
-2
```

```
-1
```

So the homogeneous solution is

$$x_{\text{homogeneous}}(t) = x_1(t) + x_2(t) = c_1\,e^{r_1\,t} + c_2\,e^{r_2\,t}$$

```
Clear[c1, c2]
xhomogeneous[t_] = c1 Exp[r1 t] + c2 Exp[r2 t];
```

So the homogeneous solution is

$$x_{\text{homogeneous}}(t) = c_1\,e^{-2\,t} + c_2\,e^{-t}$$

We can verify this is the homogeneous solution

```
D[xhomogeneous[t], {t, 2}] + 3 D[xhomogeneous[t], t] + 2 xhomogeneous[t] == 0 // FullSimplify
```

```
True
```

Note that we do not solve for the constants c_1 and c_2 yet

We can now find the particular solution. In this situation, $g(t) = 3\,e^{2\,t}$. We know for this class of functions, we can assume a particular solution of the form

$$x_{\text{particular}}(t) = A\,e^{2\,t}$$

```
xparticular[t_] = A Exp[2 t];
```

Recall that in order for $x_{\text{particular}}(t)$ to be a valid solution, it must satisfy $Q\left[x_{\text{particular}}(t)\right] = g(t)$ (Eq.2.3a)

$$g(t) = Q\left[x_{\text{particular}}(t)\right]$$

$$= Q\left[A\,e^{2\,t}\right]$$

$$= \frac{d^2}{d\,t^2}\left[A\,e^{2\,t}\right] + 3\,\frac{d}{d\,t}\left[A\,e^{2\,t}\right] + 2\left(A\,e^{2\,t}\right)$$

$$= 4\,A\,e^{2\,t} + 3\times\left(2\,A\,e^{2\,t}\right) + 2\left(A\,e^{2\,t}\right)$$

$$= 4\,A\,e^{2\,t} + 6\,A\,e^{2\,t} + 2\,A\,e^{2\,t}$$

$$3\,e^{2\,t} = 12\,A\,e^{2\,t}$$

```
D[xparticular[t], {t, 2}] + 3 D[xparticular[t], t] + 2 xparticular[t] // FullSimplify
```

$12\,A\,e^{2\,t}$

Solving for the coefficients yields

$$A = 1/4$$

```
A = 1 / 4;
```

So the particular solution is

$$x_{\text{particular}}(t) = \frac{1}{4}\,e^{2\,t}$$

So the total solution is

$$x(t) = \phi(t) = c_1\,x_{1,\text{homogeneous}}(t) + c_2\,x_{2,\text{homogeneous}}(t) + x_{\text{particular}}(t)$$

$$= c_1\,e^{-2\,t} + c_2\,e^{-t} + \frac{1}{4}\,e^{2\,t}$$

```
x[t_] = xhomogeneous[t] + xparticular[t]
```

$c1\,e^{-2\,t} + c2\,e^{-t} + \frac{e^{2\,t}}{4}$

We can verify this works ($Q[x(t)] = g(t)$)

```
D[x[t], {t, 2}] + 3 D[x[t], t] + 2 x[t] == g[t] // FullSimplify
```

```
True
```

We can now use the initial conditions to solve for the coefficients. First starting with the position initial condition

$$x(0) = 3$$

$$c_1\, e^{-2\times(0)} + c_2\, e^{-(0)} + \frac{1}{4}\, e^{2\times(0)} = 3$$

$$c_1 + c_2 + \frac{1}{4} = 3$$

$$c_1 + c_2 = 2.75$$

Now applying the velocity initial condition

$$\dot{x}(0) = -2$$

$$\frac{d}{d\,t}[x(t)]\,|_{\,t=0} = -2$$

$$\frac{d}{d\,t}\left[c_1\, e^{-2\,t} + c_2\, e^{-t} + \frac{1}{4}\, e^{2\,t}\right]|_{\,t=0} = -2$$

$$\left(-2\, c_1\, e^{-2\,t} - c_2\, e^{-t} + \frac{1}{2}\, e^{2\,t}\right)|_{\,t=0} = -2$$

$$-2\, c_1\, e^{-2\times(0)} - c_2\, e^{-(0)} + \frac{1}{2}\, e^{2\times(0)} = -2$$

$$-2\, c_1 - c_2 + \frac{1}{2} = -2$$

$$-2\, c_1 - c_2 = -2.5$$

Solving these two equations simultaneously yields

$$c_1 = -\frac{1}{4}$$
$$c_2 = 3$$

```
dx[t_] = D[x[t], t];
temp = Solve[{x[0] == 3, dx[0] == -2}, {c1, c2}];
c1 = c1 /. temp[[1]]
c2 = c2 /. temp[[1]]
```

$-\frac{1}{4}$

3

So the total function is

$$x(t) = \phi(t) = -\tfrac{1}{4}\, e^{-2\,t} + 3\, e^{-t} + \tfrac{1}{4}\, e^{2\,t}$$

We can verify this satisfies the initial conditions

```
x[0] == 3
(D[x[t], t] /. {t → 0}) == -2
```

True

True

We can again verify this satisfies the non-homogeneous equation of $\ddot{x}(t) - 3\,\dot{x}(t) - 4\,x(t) = 3\,e^{2\,t}$

```
D[x[t], {t, 2}] + 3 D[x[t], t] + 2 x[t] == g[t] // Simplify
```

True

```
p1 = Plot[{x[t], g[t]}, {t, 0, 1}]
```



Note that the response due to the homogeneous solution decays to zero and we are left with only the particular solution. Also note that $x_{\text{particular}}(t)$ does not necessarily converge to $g(t)$. This makes sense if we consider the system as an input/output system with $g(t)$ as the input and $x(t)$ as the output. The system likely does not have zero steady state error in response to an arbitrary input $g(t)$.

Compare with Laplace Technique

The major disadvantage of the previous approach is that it required prior knowledge of the form for the particular solution. Let us look at solving this using the Laplace method and compare approaches.

Recall the original ODE is

$$\ddot{x}(t) + 3\,\dot{x}(t) + 2\,x(t) = 3\,e^{2\,t}$$

Taking the Laplace transform of both sides we obtain

$$L\left[\ddot{x}(t) + 3\,\dot{x}(t) + 2\,x(t)\right] = L\left[3\,e^{2\,t}\right]$$

$$L\left[\ddot{x}(t)\right] + 3\,L\left[\dot{x}(t)\right] + 2\,L[\,x(t)] = 3\,L\left[\,e^{2\,t}\right]$$

$$\left(s^2\,X(s) - s\,x(0) - \dot{x}(0)\right) + 3\,(s\,X(s) - x(0)) + 2\,X(s) = \frac{3}{s-2} \qquad \text{note: } x(0) = 3,\, \dot{x}(0) = -2$$

$$\left(s^2\,X(s) - 3\,s + 2\right) + 3\,(s\,X(s) - 3) + 2\,X(s) = \frac{3}{s-2}$$

Solving for $X(s)$

```
Solve[(s^2 X - 3 s + 2) + 3 (s X - 3) + 2 X == 3/(s - 2), X]
```

$$\left\{\left\{X \to \frac{-11 + s + 3\,s^2}{(-2+s) \times \left(2 + 3\,s + s^2\right)}\right\}\right\}$$

$$X(s) = \frac{3\,s^2 + s - 11}{(s-2)\left(s^2 + 3\,s + 2\right)}$$

```
Solve[(s - 2) (s^2 + 3 s + 2) == 0, s]
```

$\{\{s \to -2\}, \{s \to -1\}, \{s \to 2\}\}$

We know the roots of the denominator are at +2, -1, and -2, so the partial fraction expansion should be

$$\frac{3\,s^2 + s - 11}{(s-2)\left(s^2 + 3\,s + 2\right)} = \frac{a_1}{(s-2)} + \frac{a_2}{(s+1)} + \frac{a_3}{(s+2)}$$

```
polynomial = Together[a1/(s - 2) + a2/(s + 1) + a3/(s + 2)]
num[s_] = Collect[Numerator[polynomial], s];
den[s_] = Denominator[polynomial];
```

$$\frac{2\,a1 - 4\,a2 - 2\,a3 + 3\,a1\,s - a3\,s + a1\,s^2 + a2\,s^2 + a3\,s^2}{(-2+s) \times (1+s) \times (2+s)}$$

```
Clear[a1, a2, a3]
temp = Solve[
   {Coefficient[num[s], s^2] == 3, Coefficient[num[s], s] == 1, num[0] == -11}, {a1, a2, a3}];
a1 = a1 /. temp[[1]]
a2 = a2 /. temp[[1]]
a3 = a3 /. temp[[1]]
```

$\frac{1}{4}$

3

$-\frac{1}{4}$

So the expansion is

$$X(s) = \frac{3\,s^2+s-11}{(s-2)\left(s^2+3\,s+2\right)} = \frac{1/4}{(s-2)} + \frac{3}{(s+1)} + \frac{-1/4}{(s+2)}$$

```
(3 s^2 + s - 11)/((s - 2) (s^2 + 3 s + 2)) == num[s]/den[s] // Simplify
```

True

Performing the inverse Laplace transform yields

$$x(t) = L^{-1}\left[\frac{1/4}{(s-2)} + \frac{3}{(s+1)} + \frac{-1/4}{(s+2)}\right]$$

$$= \frac{1}{4}\,L^{-1}\left[\frac{1}{(s-2)}\right] + 3\,L^{-1}\left[\frac{1}{(s+1)}\right] - \frac{1}{4}\,L^{-1}\left[\frac{1}{(s+2)}\right]$$

$$x(t) = \frac{1}{4}\,e^{2\,t} + 3\,e^{-t} - \frac{1}{4}\,e^{-2\,t}$$

So we arrive at the same solution

```
InverseLaplaceTransform[num[s]/den[s], s, t] // Expand
```

$-\frac{1}{4}\,e^{-2\,t} + 3\,e^{-t} + \frac{e^{2\,t}}{4}$

```
Clear[a1, a2, a3, temp, den, num, c2, c1, dx, x, A, xparticular, xhomogeneous, r2, r1, g]
```

Example : Non-Homogeneous ODE with Quadratic Forcing Function

$$\ddot{x}(t) + 3\,\dot{x}(t) + 2\,x(t) = 4 - t + 3\,t^2$$

where $x(0) = 3$
$\dot{x}(0) = -2$

```
g[t_] = 4 - t + 3 t^2;
```

So the homogeneous solution is the same as previous

$$x_{\text{homogeneous}}(t) = x_1(t) + x_2(t) = c_1\, e^{r_1 t} + c_2\, e^{r_2 t}$$

```
r1 = -2;
r2 = -1;
xhomogeneous[t_] = c1 Exp[r1 t] + c2 Exp[r2 t];
```

Note that we do <u>not</u> solve for the constants c_1 and c_2 yet

We can now find the particular solution. In this situation, $g(t) = 4 - t + 3\,t^2$ (a polynomial). We know for this class of functions, we can assume a particular solution of the form

$$x_{\text{particular}}(t) = A_0 + A_1\, t + A_2\, t^2$$

```
xparticular[t_] = A0 + A1 t + A2 t^2;
```

Recall that in order for $x_{\text{particular}}(t)$ to be a valid solution, it must satisfy $Q\left[x_{\text{particular}}(t)\right] = g(t)$ (Eq.2.3a)

```
equation[t_] = Collect[
  Expand[D[xparticular[t], {t, 2}] + 3 D[xparticular[t], t] + 2 xparticular[t]], {t, t^2}]
```

```
2 A0 + 3 A1 + 2 A2 + (2 A1 + 6 A2) t + 2 A2 t^2
```

Solving for the coefficients yields

$$\begin{aligned}
&2\,\text{A0} + 3\,\text{A1} + 2\,\text{A2} = 4 && \text{(constant)}\\
&2\,\text{A1} + 6\,\text{A2} = -1 && \text{(coefficient of } t\text{)}\\
&2\,\text{A2} = 3 && \text{(coefficient of } t^2\text{)}
\end{aligned}$$

```
Clear[A0, A1, A2]
temp = Solve[{equation[0] == 4, Coefficient[equation[t], t] == -1,
    Coefficient[equation[t], t^2] == 3}, {A0, A1, A2}];
A0 = A0 /. temp[[1]]
A1 = A1 /. temp[[1]]
A2 = A2 /. temp[[1]]
```

8

-5

$\frac{3}{2}$

So the particular solution is

$$x_{\text{particular}}(t) = 8 - 5\,t + \frac{3}{2}\,t^2$$

So the total solution is

$$x(t) = \phi(t) = c_1\, x_{1,\text{homogeneous}}(t) + c_2\, x_{2,\text{homogeneous}}(t) + x_{\text{particular}}(t)$$

$$= c_1\, e^{-2t} + c_2\, e^{-t} + 8 - 5\,t + \frac{3}{2}\,t^2$$

```
x[t_] = xhomogeneous[t] + xparticular[t]
```

$8 + c1\, e^{-2\,t} + c2\, e^{-t} - 5\,t + \frac{3\,t^2}{2}$

We can verify this works ($Q[x(t)] = g(t)$)

```
D[x[t], {t, 2}] + 3 D[x[t], t] + 2 x[t] == g[t] // FullSimplify
```

```
True
```

We can now use the initial conditions to solve for the coefficients. First starting with the position initial condition

```
x[0]
```

$8 + c1 + c2$

Now applying the velocity initial condition

```
dx[t_] = D[x[t], t];
dx[0]
```

$-5 - 2\,c1 - c2$

Solving these two equations simultaneously yields

$$c_1 = 2$$
$$c_2 = -7$$

```
Clear[c1, c2]
temp = Solve[{x[0] == 3, dx[0] == -2}, {c1, c2}];
c1 = c1 /. temp[[1]]
c2 = c2 /. temp[[1]]
```

2

-7

So the total function is

$$x(t) = \phi(t) = 2\,e^{-2t} - 7\,e^{-t} + 8 - 5\,t + \frac{3}{2}\,t^2$$

We can verify this satisfies the initial conditions

```
x[0] == 3
(D[x[t], t] /. {t → 0}) == -2
```

```
True
```

```
True
```

We can again verify this satisfies the non-homogeneous equation of $\ddot{x}(t) - 3\,\dot{x}(t) - 4\,x(t) = g(t)$

```
D[x[t], {t, 2}] + 3 D[x[t], t] + 2 x[t] == g[t] // Simplify
```

```
True
```

```
p1 = Plot[{x[t], g[t]}, {t, 0, 10}]
```



Compare with Laplace Technique

Again we can use the Laplace technique

The major disadvantage of the previous approach is that it required prior knowledge of the form for the particular solution. Let us look at solving this using the Laplace method and compare approaches. Recall the original ODE is

$$\ddot{x}(t) + 3\,\dot{x}(t) + 2\,x(t) = 4 - t + 3\,t^2$$

Taking the Laplace transform of both sides we obtain

$$L\left[\ddot{x}(t) + 3\,\dot{x}(t) + 2\,x(t)\right] = L\left[4 - \;t + 3\,t^2\right]$$

$$L\left[\ddot{x}(t)\right] + 3\,L\left[\dot{x}(t)\right] + 2\,L[\,x(t)] = 4\,L[1] - L[\,t] + 3\,L\left[t^2\right]$$

$$\left(s^2\,X(s) - s\,x(0) - \dot{x}(0)\right) + 3\,(s\,X(s) - x(0)) + 2\,X(s) = \frac{4}{s} - \frac{1}{s^2} + 3\,\frac{2}{s^3} \qquad \text{note: } x(0) = 3,\, \dot{x}(0) = -2$$

$$\left(s^2\,X(s) - 3\,s + 2\right) + 3\,(s\,X(s) - 3) + 2\,X(s) = \frac{4}{s} - \frac{1}{s^2} + 3\,\frac{2}{s^3}$$

Solving for $X(s)$

```
Clear[X]
temp = Solve[(s^2 X - 3 s + 2) + 3 (s X - 3) + 2 X == 4/s - 1/s^2 + 3 2/s^3, X];
X[s_] = X /. temp[[1]]
```

$$\frac{6 - s + 4 s^2 + 7 s^3 + 3 s^4}{s^3 \left(2 + 3 s + s^2\right)}$$

We could perform standard partial fraction expansion but for simplicity, we simply use Mathematica

```
InverseLaplaceTransform[X[s], s, t] // Expand
```

$$8 + 2 e^{-2 t} - 7 e^{-t} - 5 t + \frac{3 t^2}{2}$$

Which again is the same solution we obtained previously

Summary

<u>Standard/Traditional Method for Non-Homogeneous ODEs</u>

Given equation of form

$$Q[x(t)] \triangleq a\, \ddot{x}(t) + b\, \dot{x}(t) + c\, x(t) = g(t)$$

1. Solve characteristic equation to obtain roots r_1 and r_2

$$a\, r^2 + b\, r + c = 0$$

2. Assuming roots are distinct and real, form homogeneous solution (if not, will need to assume a different form of $x_{\text{homogeneous}}(t)$)

$$x_{\text{homogeneous}}(t) = x_1(t) + x_2(t) = c_1\, e^{r_1 t} + c_2\, e^{r_2 t}$$

3. Based on $g(t)$ assume a particular solution, $x_{\text{particular}}(t)$, with undetermined constants.

4. Use $Q\left[x_{\text{particular}}(t)\right] = g(t)$ to solve for constants in $x_{\text{particular}}(t)$

5. Form total solution $x(t) = \phi(t) = c_1\, x_{1,\text{homogeneous}}(t) + c_2\, x_{2,\text{homogeneous}}(t) + x_{\text{particular}}(t)$

6. Use initial conditions $x(0) = x_0$ and $\dot{x}(0) = \dot{x}_0$ to solve for c_1 and c_2

<u>Laplace Method</u>

Given equation of form

$$a\, \ddot{x}(t) + b\, \dot{x}(t) + c\, x(t) = g(t)$$

1. Laplace transform both sides (applying initial conditions for derivative terms)

2. Solve for $X(s)$ using algebra

3. Use partial fraction expansion to get $X(s)$ into a desirable form

4. Inverse Laplace transform $X(s)$ to obtain $x(t)$