Christopher Lum
lum@uw.edu

Lecture 07d
PDEs: Heat Equation

The YouTube videos that cover this lecture are located at:

-'Heat Transfer Demonstration' at https://youtu.be/FsLFZT44l48.

-'Derivation of the Heat Equation' at https://youtu.be/ixsRJPlO_rc.

-'Solving the 1D Heat Equation' at https://youtu.be/I3jiMhVGmcg.

NOTE: If we are running short on time during lecture07, this can be moved to lecture08 as Hw07 does not require this information.

Modeling: Heat Flow from a Body in Space. Heat Equation

Let us investigate the next big PDE, the **heat equation**, which governs the temperature u in a body in space.

See https://youtu.be/FsLFZT44l48 for a demonstration of heat transfer.

We make some simplifying assumptions in order to model heat transfer in a system.

1. The specific heat, $\sigma\left(\frac{J}{\text{kg}\,K}\right)$, and the *density*, $\rho\left(\frac{\text{kg}}{m^3}\right)$, and the thermal conductivity, $K\left(\frac{J}{K\,s\,m}\right)$, of the material of the body are constant. This is the case for a homogeneous material and non-extreme temperatures.

2. No heat is produced nor disappears in the body.

3. Experiments show that, in a body, heat flows in the direction of decreasing temperature, and the rate of flow is proportional to the gradient of the temperature; that is, the velocity $\overline{v}\left(\frac{J}{s}\right)$ of the heat flow in the body is of the form

$$\overline{v} = -K \text{ grad } u \qquad \textbf{\textit{(Eq.1)}}$$

where $u(x, y, z, t)$ is the temperature at a point (x, y, z) at time t (units of K)

K = thermal conductivity

We can draw a diagram of our system



Let T be a region in the body bounded by a surface S with outer unit normal vector $\overline{n}$ such that the Divergence Theorem applies (T is a closed bounded region in space whose boundary is a piecewise smooth orientable surface S). Then $\overline{v} \cdot \overline{n}$ is the component of $\overline{v}$ in the direction of $\overline{n}$. Furthermore, we assume that $\overline{n}$ is oriented "outwards" hence $|\overline{v} \cdot \overline{n} \Delta A|$ is the amount of heat leaving T (if $\overline{v} \cdot \overline{n} > 0$ at some point P) or entering T (if $\overline{v} \cdot \overline{n} < 0$ at P) per unit time at some point P of S through small portion ΔS of area ΔA. Hence the total amount of heat that flows across S from T (units of J/s) is given by the surface integral

$$\text{total heat flow across } S = \iint_S \overline{v} \cdot \overline{n} \, dA$$

Substituting in Eq.1 for $\overline{v}$ yields

$$= -K \iint_S (\text{grad } u) \cdot \overline{n} \, dA$$

Recall the Divergence Theorem of Gauss from section 10.7 which allowed us to convert between a surface integral and volume integral over the region T stated that $\iint_S \overline{F} \cdot \overline{n} \, dA = \iiint_T \text{div } \overline{F} \, dV$. Applying this to our scenario yields

$$= -K \iiint_T \text{div} (\text{grad } u) \, dx \, dy \, dz$$

Recall from Section 9.8, Eq.3, that the divergence of the gradient of a function is equal to the Laplacian of the function. In this case, div(grad u) = $\nabla^2 u$, so we obtain

$$\text{total heat flow across } S = -K \iiint_T \nabla^2 u \, dx \, dy \, dz$$

We now consider the total amount of heat in T which we denote as H

$$H = \iiint_T \sigma \rho u \, dx \, dy \, dz \qquad \text{(total heat in } T\text{)}$$

where σ =specific heat $\left(\frac{J}{\text{kg } K}\right)$

$\rho = \text{density}\left(\frac{\text{kg}}{m^3}\right)$

$u = \text{temperature}\ (K)$

The time rate of decrease of total heat is given by (note minus sign to signify heat decrease)

$$-\frac{\partial H}{\partial t} = -\iiint_T \sigma\,\rho\,\frac{\partial u}{\partial t}\,dx\,dy\,dz \quad \text{(rate of heat decrease in } T\text{)}$$

The total amount of heat leaving must be equal to the amount of heat flow across S due to our assumption that no heat is produced nor disappears in the body. Therefore

rate of heat decrease in T = total heat flow across S

$$-\iiint_T \sigma\,\rho\,\frac{\partial u}{\partial t}\,dx\,dy\,dz = -K\iiint_T \nabla^2 u\,dx\,dy\,dz$$

$$\iiint_T \sigma\,\rho\,\frac{\partial u}{\partial t}\,dx\,dy\,dz - K\iiint_T \nabla^2 u\,dx\,dy\,dz = 0 \qquad \text{note: } \sigma \text{ and } \rho \text{ are constant and not } 0$$

$$\iiint_T \frac{\partial u}{\partial t}\,dx\,dy\,dz - \frac{K}{\sigma\,\rho}\iiint_T \nabla^2 u\,dx\,dy\,dz = 0$$

$$\iiint_T \left(\frac{\partial u}{\partial t} - \frac{K}{\sigma\,\rho}\nabla^2 u\right) dx\,dy\,dz = 0$$

Since this must hold everywhere for any, arbitrary region T in the body, the integrand must be 0 everywhere

$$\frac{\partial u}{\partial t} - \frac{K}{\sigma\,\rho}\nabla^2 u = 0$$

$$\frac{\partial u}{\partial t} - c^2\,\nabla^2 u = 0 \qquad \textit{\textbf{(Eq.3)}}$$

where $c^2 = \frac{K}{\sigma\,\rho} = \text{thermal diffusivity}\left(\frac{m^2}{s}\right)$

$u = u(x,\ y,\ z,\ t) = \text{temperature at position } (x,\ y,\ z) \text{ at time } t\ (K)$

$K = \text{thermal conductivity}\left(\frac{J}{K\,s\,m}\right)$

$\sigma = \text{specific heat}\left(\frac{J}{\text{kg}\,K}\right)$

$\rho = \text{density}\left(\frac{\text{kg}}{m^3}\right)$

$\nabla^2 u = \frac{\partial^2 u}{\partial x^2} + \frac{\partial^2 u}{\partial y^2} + \frac{\partial^2 u}{\partial z^2} = \text{Laplacian of } u$

This is the **heat equation**, the fundamental PDE modeling heat flow. This is also called the **diffusion equation** because it also models chemical diffusion processes of one substance or gas into another.

As can be seen, the thermal diffusivity, c^2, plays a huge role in the PDE. This effectively characterizes the rate of transfer of heat through a material.

Heat Equation: Solution by Fourier Series. Steady Two-Dimensional Heat Problems. Dirichlet Problem.

One Dimensional Heat Equation

Before considering the multiple spatial dimensions, we first consider the heat equation in one spatial dimension. Note that the 1D refers to the single spatial variable. There is still a temporal independent variable, t, so there are two independent variables, thereby creating a PDE rather than an ODE.

We consider a long, thin metal bar of constant cross section and homogeneous material which is oriented along the x-axis and is perfectly insulated laterally, so that heat flows in the x-direction only. In this situation, the Laplacian reduces to $\nabla^2 u = u_{xx} = \partial^2 u / \partial x^2$ and the heat equation becomes the one-dimensional heat equation

$$\frac{\partial u}{\partial t} = c^2 \frac{\partial^2 u}{\partial x^2} \qquad \textbf{(Eq.1)}$$

We can solve Eq.1 for some types of boundary and initial conditions.

<u>Fixed End Temperatures</u>

We can investigate the scenario where the ends of the bar ($x = 0$ and $x = L$) are kept at temperature of 0 (for example both ends are attached to blocks of ice).

$$u(0, t) = u(L, t) = 0 \quad \forall\, t \qquad \textbf{(Eq.2)}$$

We assume that the initial temperature profile of the bar at $t = 0$ is given by $f(x)$ so we have an initial condition of

$$u(x, 0) = f(x) \qquad \textbf{(Eq.3)}$$

where $\quad f(x) = 0$ initial temperature distribution of bar (but we require that $f(0) = f(L) = 0$ to satisfy boundary condition of Eq.2)

We will solve this problem in much the same way we did with the 1D wave equation. We will employ separation of variables followed by the use of Fourier series.

Step 1. Apply method of separating variables (aka product method) to obtain two ordinary differential equations.

Step 2. Determine solutions of those two equations that satisfy the boundary conditions

Step 3. Use Fourier series to compose those solutions to get a solution to the heat equation that also satisfies the initial conditions.

Step 1. Two ODEs from the Heat Equation

Once again, we assume that the solution to the heat equation can be written as the product of two functions, one which is solely a function of x and the other that is solely a function of t

$$u(x,\ t) = F(x)\,G(t)$$

Following the same procedure as before (substituting into Eq.1 and manipulating the equations), we obtain

$$\frac{\dot{G}}{c^2\,G} = \frac{F''}{F} \qquad \textbf{(Eq.4)}$$

Once again we notice that the left side depends only on t and the right side only on x, so that both sides must equal a constant k. You can show that for $k \geq 0$, the only solution $u = F\,G$ satisfying Eq.2 (boundary conditions) is $u = 0$ (trivial solution). Therefore, we consider $k = -p^2 < 0$ and have

$$\frac{\dot{G}}{c^2\,G} = \frac{F''}{F} = -p^2$$

We immediately obtain two ODEs

$$F''(x) + p^2\,F(x) = 0 \qquad \textbf{(Eq.5)}$$

$$\dot{G}(t) + c^2\,p^2\,G(t) = 0 \qquad \textbf{(Eq.6)}$$

Step 2. Satisfying the Boundary Conditions

We now would like to determine solutions for $F(x)$ and $G(t)$ such that $u(x,\ t) = F(x)\,G(t)$ satisfies the boundary conditions previously described in Eq.2. In other words

$$u(0,\ t) = F(0)\,G(t) = 0\,\forall\,t$$

$$u(L,\ t) = F(L)\,G(t) = 0\,\forall\,t$$

Solving the First ODE (Eq.5)

The first ODE is again an undamped oscillator so the general solution is

$$F(x) = A\cos(p\,x) + B\sin(p\,x) \qquad \textbf{(Eq.7)}$$

```
In[1]:= Fgeneral[x_] = A Cos[p x] + B Sin[p x];
```

We can now apply the boundary conditions to find A and B

<u>Apply BC#1: $F(0) \equiv 0$</u>

$$F(0) = A\cos(0) + B\sin(0)$$

$$0 = A$$

<u>Apply BC#2: $F(L) \equiv 0$</u>

$$F(L) = A\cos(p\,L) + B\sin(p\,L) \qquad \text{note: } A = 0$$

$$0 = B\sin(p\,L)$$

We notice that $B = 0$ yields the trivial solution of $F(x) = 0$, so we need to find another solution such that $0 = B\sin(p\,L)$ is satisfied but we do not obtain a trivial solution. We see that this is only possible if $\sin(p\,L) = 0$. We notice that any p such that $p\,L = n\,\pi$ for $n \in Z$ (integer) will satisfy this equation. Therefore, we see that

$$p\,L = n\,\pi$$

$$p = \frac{n\,\pi}{L} \qquad n \in Z \qquad \textbf{\textit{(Eq.9)}}$$

Therefore, if Eq.9 is satisfied, then B can be arbitrary and we obtain

$$F_n(x) = B\sin\left(\frac{n\,\pi}{L}\,x\right) \quad n \in Z \qquad \textbf{\textit{(Eq.10)}}$$

We can verify that this satisfies the original ordinary differential equation and the boundary conditions. Note that these are not the eigenfunctions of the entire problem as $u = F\,G$ is the solution to the entire problem.

```
In[2]:= Fn[x_] = Fgeneral[x] /. {A → 0}
Out[2]= B Sin[p x]
```

In[3]:=
```
Print["Satisfies PDE?"]
D[Fn[x], {x, 2}] + p^2 Fn[x] == 0

Print["Satisfies boundary conditions?"]
Fn[0] == 0
Simplify[Fn[L] /. {p → n π / L}, n ∈ Integers] == 0
```

```
Satisfies PDE?
```

Out[4]= True

```
Satisfies boundary conditions?
```

Out[6]= True

Out[7]= True

So we see that we satisfy all the requirements. Notice that Eq.5 and the boundary conditions for the heat equation is identical to the 1D wave equation we analyzed previously. Therefore, as expected, all of the results and analysis so far have been the same.

Solving the Second ODE (Eq.6)

We now turn our attention to Eq.6, which is repeated here for convenience

$$\dot{G}(t) - c^2 k\, G(t) = 0 \qquad \textbf{(Eq.6)}$$

Recall that the separation constant k must be negative and we substituted $k = -p^2$. Furthermore, from the previous section, we determined that p must take specific values, namely $p = \frac{n\pi}{L}$ with $n \in Z$ (Eq.9). So valid values of k are

$$k = -\left(\frac{n\pi}{L}\right)^2 \qquad n \in Z$$

Therefore, we can rewrite Eq.6 as

$$\dot{G}(t) - c^2\left(-\left(\frac{n\pi}{L}\right)^2\right) G(t) = 0$$

$$\dot{G}(t) + \left(\frac{c\,n\pi}{L}\right)^2 G(t) = 0$$

$$\dot{G}(t) + \lambda_n^{\,2}\, G(t) = 0$$

where $\lambda_n = c\,n\,\pi / L$

Once again, we recognize this as a first order ordinary differential equation with general solution of

$$G_n(t) = B_n\, e^{-\lambda_n^{\,2} t} \qquad n \in Z$$

We will temporarily defer finding the coefficient B_n

So the solution is given by $u_n(x, t) = F_n(x)\, G_n(t)$

$$u_n(x,\,t) = B_n\, e^{-\lambda_n{}^2 t} \sin\!\left(\frac{n\,\pi}{L}\, x\right) \qquad\qquad n \in Z \qquad \textbf{\textit{(Eq.8)}}$$

where $\quad \lambda_n = c\, n\, \pi / L$

In[8]:=
```
λn[n_] = c n π / L;
un[x_, t_, n_] = Bn Exp[-λn[n]^2 t] Fn[x] /. {p → n π / L, B → 1}

(*verify we satisfy original PDE*)
Print["Satisfies original PDE?"]
D[un[x, t, n], {t, 1}] == c^2 D[un[x, t, n], {x, 2}] // Simplify

(*verify that we satisfy the boundary condition*)
Print["Satisfies boundary conditions?"]
un[0, t, n] == 0
Simplify[un[L, t, n], n ∈ Integers] == 0
```

Out[9]= $\text{Bn}\, e^{-\frac{c^2 n^2 \pi^2 t}{L^2}}\, \text{Sin}\left[\frac{n\,\pi\,x}{L}\right]$

```
Satisfies original PDE?
```

Out[11]= True

```
Satisfies boundary conditions?
```

Out[13]= True

Out[14]= True

The family of functions described by Eq.8 are the **eigenfunctions** values $\lambda_n = c\, n\, \pi / L$ are the **eigenvalues.**

Step 3. Solution of the Entire Problem. Fourier Series

In a similar fashion, we now consider the total solution to be a Fourier series of the eigenfunctions

$$u(x,\,t) = \sum_{n=1}^{\infty} u_n(x,\,t) = \sum_{n=1}^{\infty} B_n\, e^{-\lambda_n{}^2 t} \sin\!\left(\frac{n\,\pi}{L}\, x\right) \qquad \textbf{\textit{(Eq.9)}}$$

where $\quad \lambda_n = c\, n\, \pi / L$

Satisfying Initial Condition (Temperature Distribution)

We now would like to satisfy the initial condition of $u(x,\,0) = f(x)$. Eq.9 becomes

$$u(x,\,0) = \sum_{n=1}^{\infty} B_n \sin\!\left(\frac{n\,\pi}{L}\, x\right) = f(x)$$

Therefore, we need to choose the coefficients B_n such that $u(x,\,0) = f(x)$. Recall from Section 11.2 Eq.5**, if we extend a function as an odd function, the half range expansion is given as

$$f(x) = \sum_{n=1}^{\infty} b_n \sin\!\left(\frac{n\,\pi}{L}\, x\right)$$

where $\quad b_n = \frac{2}{L}\int_0^L f(x)\sin\left(\frac{n\pi}{L}x\right)dx \qquad n = 1, 2 \ldots$

So we see that we need to choose coefficients B_n using

$$B_n = \frac{2}{L}\int_0^L f(x)\sin\left(\frac{n\pi}{L}x\right)dx \qquad n = 1, 2 \ldots \qquad \textbf{\textit{(Eq.10)}}$$

Example 1: Sinusoidal Initial Temperature

Find the temperature $u(x, t)$ in a laterally insulated copper bar 80 cm long if the initial temperature is $100\sin(\pi x/80)$ (degrees Celsius) and the ends are kept at $0^o\,C$. How long will it take for the maximum temperature in the bar to drop to $50^o\,C$. Some relevant physical data for copper is shown below.

$\rho = 8.92\,g/\text{cm}^3$

$\sigma = \text{specific heat } = 0.092\,\text{cal}/g\,^o C$

$K = \text{thermal conductivity } = 0.95\,\text{cal}/(\text{cm sec}\,^o C)$

We can now find the solution $u(x, t)$ for this situation

In[15]:=
```
f[x_] = 100 Sin[π x / 80];

(*Visualize the function*)
Plot[f[x], {x, 0, 80}, AxesLabel → {"x", "f(x)"}]
```

Out[16]=



Recall that the general solution is given as

$$u(x, t) = \sum_{n=1}^{\infty} u_n(x, t) = \sum_{n=1}^{\infty} B_n e^{-\lambda_n^2 t}\sin\left(\frac{n\pi}{L}x\right) \qquad \textbf{\textit{(Eq.9)}}$$

where $\quad \lambda_n = c\,n\,\pi/L$

$$B_n = \frac{2}{L}\int_0^L f(x)\sin\left(\frac{n\pi}{L}x\right)dx$$

We can calculate the B_n coefficients

In[17]:=
```
Bn[n_] = 2/L (Integrate[f[x] Sin[n π/L x], {x, 0, L}]) /. {L → 80} // Simplify
```

Out[17]= $\frac{200\,\text{Sin}[n\,\pi]}{\pi - n^2\,\pi}$

So we have

$$B_n = \frac{200 \sin(n\,\pi)}{\pi - n^2\,\pi}$$

Note that we cannot evaluate this for $n = 1$ because we obtain a 0/0 indeterminate expression. Instead, we can write out Eq.9 at $t = 0$ (which must equal the initial condition)

$$u(x,\, 0) = f(x) = \sum_{n=1}^{\infty} B_n\, e^{-\lambda_n^{\,2} (0)} \sin\left(\frac{n\,\pi}{L}\, x\right)$$

$$f(x) = \sum_{n=1}^{\infty} B_n \sin\left(\frac{n\,\pi}{L}\, x\right) \qquad \text{recall: } f(x) = 100 \sin\left(\frac{1\,\pi}{L}\, x\right)$$

$$100 \sin\left(\frac{1\,\pi}{L}\, x\right) = B_1 \sin\left(\frac{1\,\pi}{L}\, x\right) + B_2 \sin\left(\frac{2\,\pi}{L}\, x\right) + B_3 \sin\left(\frac{3\,\pi}{L}\, x\right) + \ldots$$

So from inspection, we see that we must have $B_1 = 100$ and $B_2 = B_3 =. .. = 0$

So the solution takes the form of

$$u(x,\, t) = B_1\, e^{-\lambda_1^{\,2}\, t} \sin\left(\frac{\pi}{L}\, x\right)$$

where $B_1 = 100$

$\lambda_1 = c\,\pi / L$

$c = \left(\frac{K}{\sigma\,\rho}\right)^{1/2}$

```
In[18]:= u[x_, t_] = 100 Exp[-λn[1]^2 t] Sin[π/L x];

         (*verify we satisfy original PDE*)
         Print["Satisfies original PDE?"]
         D[u[x, t], {t, 1}] == c^2 D[u[x, t], {x, 2}] // Simplify

         (*verify that we satisfy the boundary condition*)
         Print["Satisfies boundary conditions?"]
         u[0, t] == 0
         Simplify[u[L, t], n ∈ Integers] == 0

         (*verify that we satisfy the initial condition*)
         Print["Satisfies initial conditions?"]
         u[x, 0] == 100 Sin[π/L x]

         Satisfies original PDE?

Out[20]= True

         Satisfies boundary conditions?

Out[22]= True

Out[23]= True

         Satisfies initial conditions?

Out[25]= True
```

We can visualize this example

In[26]:=
```
(*Define constants given by problem*)
ρGiven = 8.92;
σGiven = 0.092;
KGiven = 0.95;
LGiven = 80;

(*Compute intermediate variables*)
cGiven = (KGiven/(σGiven ρGiven))^(1/2);

(*Define parameters for the animation*)
tMax = 1500;
uMin = 0;
uMax = 100;

(*Animate the scenario*)
Manipulate[
 Plot[u[x, t] /. {c → cGiven, L → LGiven}, {x, 0, LGiven},
  PlotRange → {{0, LGiven}, {uMin, uMax}},
  PlotLabel → "u(x,t)",
  AxesLabel → {"x", "u(x,t)"}],

 {t, 0, tMax}
]
```

Out[34]=



We see that the maximum temperature of the bar is halfway along the length of the bar. If we isolate this spatial variable, we can identify how this location's temperature varies with time

$$u_{\text{middle}}(t) = u\left(\frac{L}{2}, t\right)$$

$$= B_1 e^{-\lambda_1^2 t} \sin\left(\frac{\pi}{L} x\right) |_{x=L/2} \qquad \text{recall: } B_1 = 100, \lambda_1 = c\,\pi/L$$

$$= 100\, e^{-\left(\frac{c\pi}{L}\right)^2 t} \sin\left(\frac{\pi}{2}\right)$$

$$u_{\text{middle}}(t) = 100\, e^{-\left(\frac{c\pi}{L}\right)^2 t}$$

In[35]:=
```
uMiddle[t_] = u[L / 2, t]
```

Out[35]= $100\, e^{-\frac{c^2 \pi^2 t}{L^2}}$

We can plot this

In[36]:=
```
Plot[uMiddle[t] /. {c → cGiven, L → LGiven}, {t, 0, 1000},
 PlotRange → All, AxesLabel → {"t (sec)", "umiddle(t)"}]
```

Out[36]=



So we see that we simply need to solve for the time required to reach 50^o

In[37]:=
```
Solve[(uMiddle[t] /. {c → cGiven, L → LGiven}) == 50, t]
```

Solve: Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.

Out[37]= {{t → 388.271}}

So we see that it takes approximately 388 seconds ≈ 6.5 min for the bar to cool to 50 degrees.

Example 2: Speed of Decay

We can repeat the problem with a different initial condition of

$$f_2(x) = u(x, 0) = 100 \sin\left(\frac{3\pi}{80} x\right)$$

We can now find the solution $u(x, t)$ for this situation

In[38]:=
```
f2[x_] = 100 Sin[3 π x / 80];

(*Visualize the function*)
Plot[f2[x], {x, 0, 80}, AxesLabel → {"x", "f(x)"}]
```

Out[39]=



The solution is virtually the same. Computing the coefficients B_n yields

In[40]:=
```
Bn2[n_] = 2/L (Integrate[f2[x] Sin[n π/L x], {x, 0, L}]) /. {L → 80} // Simplify
```

Out[40]= $\frac{600 \operatorname{Sin}[n \pi]}{9 \pi - n^2 \pi}$

So we have

$$B_n = \frac{600 \sin(n \pi)}{9 \pi - n^2 \pi}$$

Note that we cannot evaluate this for $n = 3$ because we obtain a 0/0 indeterminate expression. Instead, we can write out Eq.9 at $t = 0$ (which must equal the initial condition)

$$u(x, 0) = f(x) = \sum_{n=1}^{\infty} B_n e^{-\lambda_n^2 (0)} \sin\left(\frac{n \pi}{L} x\right)$$

$$f(x) = \sum_{n=1}^{\infty} B_n \sin\left(\frac{n \pi}{L} x\right) \qquad \text{recall: } f(x) = 100 \sin\left(\frac{3 \pi}{80} x\right), L = 80$$

$$100 \sin\left(\frac{3 \pi}{80} x\right) = B_1 \sin\left(\frac{1 \pi}{L} x\right) + B_2 \sin\left(\frac{2 \pi}{L} x\right) + B_3 \sin\left(\frac{3 \pi}{L} x\right) + \ldots$$

So from inspection, we see that we must have $B_1 = B_2 = 0$, $B_3 = 100$, and $B_4 = B_5 = ... = 0$

So the solution takes the form of

$$u(x, t) = B_3 e^{-\lambda_3^2 t} \sin\left(\frac{3 \pi}{L} x\right)$$

where $\quad B_3 = 100$

$\lambda_3 = 3 c \pi / L$

$$c = \left(\frac{K}{\sigma \rho}\right)^{1/2}$$

In[41]:=
```
u2[x_, t_] = 100 Exp[-λn[3]^2 t] Sin[3 π/L x];

(*verify we satisfy original PDE*)
Print["Satisfies original PDE?"]
D[u2[x, t], {t, 1}] == c^2 D[u2[x, t], {x, 2}] // Simplify

(*verify that we satisfy the boundary condition*)
Print["Satisfies boundary conditions?"]
u2[0, t] == 0
Simplify[u2[L, t], n ∈ Integers] == 0

(*verify that we satisfy the initial condition*)
Print["Satisfies initial conditions?"]
(u2[x, 0] /. {L → LGiven}) == f2[x]
```

```
Satisfies original PDE?
```

Out[43]= True

```
Satisfies boundary conditions?
```

Out[45]= True

Out[46]= True

```
Satisfies initial conditions?
```

Out[48]= True

We can visualize this example

In[49]:=
```
(*Animate the scenario*)
Manipulate[
 Plot[u2[x, t] /. {c → cGiven, L → LGiven}, {x, 0, LGiven},
  PlotRange → {{0, LGiven}, {-uMax, uMax}},
  PlotLabel → "u(x,t)",
  AxesLabel → {"x", "u(x,t)"}],

 {t, 0, tMax / 2}
]
```

Out[49]=



We see that the maximum temperature of the bar is halfway along the length of the bar (also at locations of 1/4 and 3/4 along the bar). If we isolate this spatial variable, we can identify how this location's temperature varies with time

$$u_{\text{middle}}(t) = u\left(\frac{L}{2}, t\right)$$

$$= B_1 e^{-\lambda_1^2 t} \sin\left(\frac{\pi}{L} x\right) \Big|_{x=L/2} \qquad \text{recall: } B_1 = 100, \lambda_1 = c\,\pi / L$$

$$= 100\, e^{-\left(\frac{c\pi}{L}\right)^2 t} \sin\left(\frac{\pi}{2}\right)$$

$$u_{\text{middle}}(t) = 100\, e^{-\left(\frac{c\pi}{L}\right)^2 t}$$

In[50]:=
```
uMiddle2[t_] = u2[L / 2, t]
```

Out[50]= $-100\, e^{-\frac{9 c^2 \pi^2 t}{L^2}}$

We can plot this

In[51]:=
```
Plot[uMiddle2[t] /. {c → cGiven, L → LGiven},
 {t, 0, 1000}, PlotRange → All, AxesLabel → {"t (sec)", "umiddle(t)"}]
```

Out[51]=



So we see that we simply need to solve for the time required to reach -50^o

In[52]:=
```
Solve[(uMiddle2[t] /. {c → cGiven, L → LGiven}) == -50, t]
```

Solve: Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.

Out[52]= {{t → 43.1412}}

So we see that it takes approximately 43 seconds for the bar to cool to 50 degrees. This is much faster due to the fact that the thermal gradients are much more drastic which promote faster heat transfer.