Christopher Lum
lum@uw.edu

Lecture 02h
Deriving Percent Overshoot, Settling Time, and Other Performance Metrics of a 2nd Order Dynamic System

Lecture is on YouTube

The YouTube video entitled ‘Deriving Percent Overshoot, Settling Time, and Other Performance Metrics of a 2nd Order Dynamic System’ that covers this lecture is located at https://youtu.be/QWCLthgJEbc.

Outline

-Performance of a 2nd Order System
-Underdamped System
  -Time of First Peak (T_p)
  -Amplitude of First Peak (M_{P_t})
  -Percent Overshoot (PO)
  -Settling Time (T_s)
  -Rise Time (T_r)
  -Summary
  -Example: Simple System

Introduction

The goal of this lecture is to develop and understanding between the natural frequency and damping ratio of a system and the resulting performance.

Performance of a 2nd Order System

Let us study the response of a 2nd order system. Let us choose a system of the form

$$\ddot{y}(t) + 2\,\zeta\,\omega_n\,\dot{y}(t) + \omega_n^{\,2}\,y(t) = \omega_n^{\,2}\,r(t) \qquad \textbf{(Eq.1)}$$

Recall that the solution to this equation relied on the poles of the characteristic equation. This could be further parameterized in terms of the damping ratio, ζ

1. $(2\,\zeta\,\omega_n)^2 > 4\,\omega_n^2 \Longleftrightarrow \zeta^2 > 1 \Rightarrow$ distinct real roots (overdamped)
2. $(2\,\zeta\,\omega_n)^2 = 4\,\omega_n^2 \Longleftrightarrow \zeta^2 = 1 \Rightarrow$ repeated real roots (critically damped)
3. $(2\,\zeta\,\omega_n)^2 < 4\,\omega_n^2 \Longleftrightarrow \zeta^2 < 1 \Rightarrow$ complex roots (under damped)

We can calculate the transfer function of output to input

$$s^2\,Y(s) + 2\,\zeta\,\omega_n\,s\,Y(s) + \omega_n^2\,Y(s) = \omega_n^2\,R(s)$$

$$G(s) = \frac{Y(s)}{R(s)} = \frac{\omega_n^2}{s^2 + 2\,\zeta\,\omega_n\,s + \omega_n^2} \qquad \textbf{\textit{(Eq.2)}}$$

where $\omega_n > 0$ is referred to as the natural frequency of the system
$\zeta > 0$ is referred to as the damping ratio of the system

```
In[1]:= G[s_] = ωn^2/(s^2 + 2 ξ ωn s + ωn^2);
```

We can find the DC gain of the system, which is really the steady state response of the system with respect to a step input.

$$\text{DC gain } = G(0) = \frac{\omega_n^2}{\omega_n^2} = 1$$

So we see that the steady state error of this system will be

$$e_{ss} = (1 - G(0))\,A$$

$$= (1 - 1)\,A$$

$$e_{ss} = 0$$

Let us verify this. We can compute the response of the system to a step function ($R(s) = A/s$)

```
In[13]:= yStep[t_] = FullSimplify[InverseLaplaceTransform[G[s] A/s, s, t], {ξ > 0, ωn > 0}] // Expand;
yStep[t] /. {Sqrt[-1 + ξ^2] → α}
```

$$\text{Out[14]= } A - \frac{1}{2}\,A\,e^{-t\,(\alpha+\xi)\,\omega n} - \frac{1}{2}\,A\,e^{2\,t\,\alpha\,\omega n - t\,(\alpha+\xi)\,\omega n} + \frac{A\,e^{-t\,(\alpha+\xi)\,\omega n}\,\xi}{2\,\sqrt{-1+\xi^2}} - \frac{A\,e^{2\,t\,\alpha\,\omega n - t\,(\alpha+\xi)\,\omega n}\,\xi}{2\,\sqrt{-1+\xi^2}}$$

$$y(t) = A - \frac{1}{2}\,A\,e^{-(\alpha+\zeta)\,\omega_n\,t} - \frac{1}{2}\,A\,e^{2\,t\,\alpha\,\omega_n - (\alpha-\zeta)\,\omega_n\,t} + \frac{A\,\zeta}{2\,\alpha}\,e^{-(\alpha+\zeta)\,\omega_n\,t} - \frac{A\,\zeta}{2\,\alpha}\,e^{2\,t\,\alpha\,\omega_n - (\alpha+\zeta)\,\omega_n\,t}$$

$$= A - \frac{1}{2}\,A\,e^{-(\alpha+\zeta)\,\omega_n\,t} - \frac{1}{2}\,A\,e^{(\alpha-\zeta)\,\omega_n\,t} + \frac{A\,\zeta}{2\,\alpha}\,e^{-(\alpha+\zeta)\,\omega_n\,t} - \frac{A\,\zeta}{2\,\alpha}\,e^{(\alpha-\zeta)\,\omega_n\,t}$$

$$= A - \frac{1}{2}\,A\,e^{-(\alpha+\zeta)\,\omega_n\,t} - \frac{1}{2}\,A\,e^{-(\zeta-\alpha)\,\omega_n\,t} + \frac{A\,\zeta}{2\,\alpha}\,e^{-(\alpha+\zeta)\,\omega_n\,t} - \frac{A\,\zeta}{2\,\alpha}\,e^{-(\zeta-\alpha)\,\omega_n\,t}$$

$$= A - \frac{1}{2} A\, e^{-(\alpha+\zeta)\,\omega_n\, t} + \frac{A\,\zeta}{2\,\alpha}\, e^{-(\alpha+\zeta)\,\omega_n\, t} - \frac{1}{2} A\, e^{-(\zeta-\alpha)\,\omega_n\, t} - \frac{A\,\zeta}{2\,\alpha}\, e^{-(\zeta-\alpha)\,\omega_n\, t}$$

$$= A + \left(-\frac{1}{2} A + \frac{A\,\zeta}{2\,\alpha}\right) e^{-(\alpha+\zeta)\,\omega_n\, t} + \left(-\frac{1}{2} A - \frac{A\,\zeta}{2\,\alpha}\right) e^{-(\zeta-\alpha)\,\omega_n\, t}$$

$$y(t) = A + \left(\frac{A\,\zeta}{2\,\alpha} - \frac{1}{2} A\right) e^{-(\alpha+\zeta)\,\omega_n\, t} - \left(\frac{1}{2} A + \frac{A\,\zeta}{2\,\alpha}\right) e^{-(\zeta-\alpha)\,\omega_n\, t} \qquad \textbf{\textit{(Eq.3)}}$$

where $\quad \alpha = \sqrt{\zeta^2 - 1}$

In[18]:=
```
yStepCheck[t_] =
  A + (A ζ/(2 α) - 1/2 A) Exp[-(α + ζ) ωn t] - (1/2 A + A ζ/(2 α)) Exp[-(ζ - α) ωn t] /. {α → Sqrt[-1 + ζ^2]}
yStepCheck[t] == yStep[t] // Simplify
```

Out[18]= $A + e^{t\left(-\zeta-\sqrt{-1+\zeta^2}\right)\omega n}\left(-\frac{A}{2} + \frac{A\,\zeta}{2\sqrt{-1+\zeta^2}}\right) - e^{t\left(-\zeta+\sqrt{-1+\zeta^2}\right)\omega n}\left(\frac{A}{2} + \frac{A\,\zeta}{2\sqrt{-1+\zeta^2}}\right)$

Out[19]= `True`

Let us ensure that the term $\alpha + \zeta > 0$

$$\alpha + \zeta > 0 \qquad \text{recall: } \alpha = \sqrt{\zeta^2 - 1}$$

$$\sqrt{\zeta^2 - 1} + \zeta > 0$$

We see that for any $\zeta > 0$, the term $\sqrt{\zeta^2 - 1}$ will always be imaginary or greater than or equal to 0 so $\alpha + \zeta > 0$.

$$\zeta = 0 \qquad \sqrt{\zeta^2 - 1} = i$$
$$\zeta = 1 \qquad \sqrt{\zeta^2 - 1} = 0$$
$$\zeta > 1 \qquad \sqrt{\zeta^2 - 1} > 1$$

Let us ensure that the term $\zeta - \alpha > 0$

$$\zeta - \alpha > 0 \qquad \text{recall: } \alpha = \sqrt{\zeta^2 - 1}$$

$$\zeta - \sqrt{\zeta^2 - 1} > 0$$

Once again, we see that for any $\zeta > 0$, the term $\sqrt{\zeta^2 - 1}$ will always be less than ζ, so $\zeta - \alpha > 0$.

Using these two results we see that

$$y_{ss}(t) = A$$

This confirms our result that the DC gain of the system is 1. We could also arrive at this conclusion using the final value theorem.

Note that we can rewrite the response $y(t)$ as

$$y(t) = A + \left(\frac{A\zeta}{2\sqrt{\zeta^2-1}} - \frac{1}{2}A\right)e^{-\left(\sqrt{\zeta^2-1}+\zeta\right)\omega_n t} - \left(\frac{1}{2}A + \frac{A\zeta}{2\sqrt{\zeta^2-1}}\right)e^{-\left(\zeta-\sqrt{\zeta^2-1}\right)\omega_n t}$$

$$= A + a\,e^{-\left(\sqrt{\zeta^2-1}+\zeta\right)\omega_n t} - b\,e^{-\left(\zeta-\sqrt{\zeta^2-1}\right)\omega_n t}$$

$$= A + a\,e^{\left(-\sqrt{\zeta^2-1}\,\omega_n t-\zeta\,\omega_n t\right)} - b\,e^{\left(-\zeta\,\omega_n t+\sqrt{\zeta^2-1}\,\omega_n t\right)}$$

$$= A + a\,e^{-\zeta\,\omega_n t}\,e^{-\sqrt{\zeta^2-1}\,\omega_n t} - b\,e^{-\zeta\,\omega_n t}\,e^{\sqrt{\zeta^2-1}\,\omega_n t} \qquad \text{let: } \sigma = \zeta\,\omega_n \text{ and } \omega = \sqrt{\zeta^2-1}\,\omega_n$$

$$= A + a\,e^{-\sigma t}\,e^{-\omega t} - b\,e^{-\sigma t}\,e^{\omega t}$$

$$y(t) = A + e^{-\sigma t}\left(a\,e^{-\omega t} - b\,e^{\omega t}\right) \qquad \textbf{\textit{(Eq.4)}}$$

where
$$a = \left(\frac{A\zeta}{2\sqrt{\zeta^2-1}} - \frac{1}{2}A\right)$$
$$b = \left(\frac{1}{2}A + \frac{A\zeta}{2\sqrt{\zeta^2-1}}\right)$$
$$\sigma = \zeta\,\omega_n$$
$$\omega = \sqrt{\zeta^2-1}\,\omega_n$$

In[20]:=
```
yStepCheck2[t_] = A + Exp[-σ t] (a Exp[-ω t] - b Exp[ω t])
(yStepCheck2[t] /. {a → A ξ/(2 Sqrt[-1 + ξ^2]) - 1/2 A, b → 1/2 A + A ξ/(2 Sqrt[-1 + ξ^2]), σ → ξ ωn, ω → Sqrt[ξ^2 - 1] ωn}) ==
  yStep[t] // FullSimplify
```

Out[20]= $A + e^{-t\,\sigma}\left(a\,e^{-t\,\omega} - b\,e^{t\,\omega}\right)$

Out[21]= True

A final form of the equation that may be useful is if we write this in terms of cos and sin

$$y(t) = A + e^{-\sigma t}\left(a\,e^{-\omega t} - b\,e^{\omega t}\right)$$

$$= A + e^{-\sigma t}\left(a\,e^{-\sqrt{\zeta^2-1}\,\omega_n t} - b\,e^{\sqrt{\zeta^2-1}\,\omega_n t}\right)$$

$$= A + e^{-\sigma t}\left(a\, e^{-i \sqrt{1-\zeta^2}\, \omega_n t} - b\; e^{i \sqrt{1-\zeta^2}\, \omega_n t}\right)$$

Note that we can alternately write a and b as

$$a = \left(\frac{A\,\zeta}{2\,\sqrt{\zeta^2-1}} - \frac{1}{2}\,A\right) = \left(\frac{A\,\zeta}{2\,\sqrt{1-\zeta^2}\;i} - \frac{1}{2}\,A\right)$$

$$b = \left(\frac{1}{2}\,A + \frac{A\,\zeta}{2\,\sqrt{\zeta^2-1}}\right) = \left(\frac{1}{2}\,A + \frac{A\,\zeta}{2\,\sqrt{1-\zeta^2}\;i}\right)$$

We can now rewrite the exponents as trigonometric functions using the relationship that $e^{i\,x} = \cos(x) + i \sin(x)$.

In[22]:=
```
yStepTrig[t_] =
  A + Exp[-σ t] (a ExpToTrig[Exp[-I Sqrt[1 - ζ^2] ωn t]] - b ExpToTrig[Exp[I Sqrt[1 - ζ^2] ωn t]]) /.
    {a → (A ζ)/(2 Sqrt[1 - ζ^2] I) - 1/2 A, b → 1/2 A + (A ζ)/(2 Sqrt[1 - ζ^2] I), σ → ζ ωn} // Expand // Simplify
```

Out[22]= $A - A\, e^{-t\,\zeta\,\omega n} \operatorname{Cos}\left[t\,\sqrt{1-\zeta^2}\;\omega n\right] - \frac{A\, e^{-t\,\zeta\,\omega n}\,\zeta \operatorname{Sin}\left[t\,\sqrt{1-\zeta^2}\;\omega n\right]}{\sqrt{1-\zeta^2}}$

So we have

$$y(t) = A - A\,e^{-\zeta\,\omega_n t}\left(\cos\left(\sqrt{1-\zeta^2}\;\omega_n t\right) + \frac{\zeta}{\sqrt{1-\zeta^2}} \sin\left(\sqrt{1-\zeta^2}\;\omega_n t\right)\right)$$

We can verify that this is a solution to the original differential equation and has no initial conditions.

In[23]:=
```
Print["Satisfies original ODE"]
D[yStepTrig[t], {t, 2}] + 2 ζ ωn D[yStepTrig[t], t] + ωn^2 yStepTrig[t] == ωn^2 A // Simplify

Print["Initial conditions"]
yStepTrig[0]
D[yStepTrig[t], t] /. {t → 0} // Simplify
```

```
Satisfies original ODE
```

Out[24]= True

```
Initial conditions
```

Out[26]= 0

Out[27]= 0

We can make substitutions of $\sigma = \zeta\,\omega_n$ and $\omega_d = \sqrt{1-\zeta^2}\;\omega_n$

In[28]:=
```
yStepTrigSimple[t_] = yStepTrig[t] /. {ξ ωn → σ, Sqrt[1 - ξ^2] ωn → ωd}
```

Out[28]= $A - A e^{-t\sigma} \text{Cos}[t\,\omega d] - \frac{A e^{-t\sigma} \xi\, \text{Sin}[t\,\omega d]}{\sqrt{1-\xi^2}}$

So the trajectory of the system in response to a step of magnitude *A* can also be written as

$$y(t) = A - A e^{-\sigma t}\left(\cos(\omega_d t) + \frac{\zeta}{\sqrt{1-\zeta^2}} \sin(\omega_d t)\right) \qquad \textbf{\textit{(Eq.5)}}$$

where $\sigma = \zeta\,\omega_n$

$\omega_d = \sqrt{1-\zeta^2}\;\omega_n$

Underdamped System

Most of the interesting behavior occurs when the system is underdamped. If $\zeta \in (0, 1)$, we can analyze the behavior. Recall the trajectory of the system in response to a step input of magnitude *A* was given by Eq.5

$$y(t) = A - A e^{-\sigma t}\left(\cos(\omega_d t) + \frac{\zeta}{\sqrt{1-\zeta^2}} \sin(\omega_d t)\right)$$

where $\sigma = \zeta\,\omega_n$

$\omega_d = \sqrt{1-\zeta^2}\;\omega_n$

Since $\zeta \in (0, 1)$, $\sqrt{1-\zeta^2} > 0$ and all the terms of real. Therefore the system is just oscillatory with frequency $\omega_d = \sqrt{1-\zeta^2}\;\omega_n$.

In[29]:=
```
Agiven = 2.3;
ωngiven = 1.5;
ξgiven = 0.2;

tFinal = 14;

Legended[

 Show[
  (*Plot ystep(t)  (the unintuitive solution)*)
  Plot[yStep[t] /. {A → Agiven, ξ → ξgiven, ωn → ωngiven}, {t, 0, tFinal},
   PlotRange → All,
   PlotStyle → Red],

  (*Plot ystep,trig(t)  (the formatted, helpful solution to show it is the same as ystep)*)
  Plot[yStepTrig[t] /. {A → Agiven, ξ → ξgiven, ωn → ωngiven}, {t, 0, tFinal},
   PlotRange → All,
   PlotStyle → Green],

  Plot[Agiven, {t, 0, tFinal},
   PlotStyle → Blue]
 ],

 (*Add legend information*)
 SwatchLegend[{Red, Green, Blue}, {"yStep", "yStepTrig", "A"}]
]
```

Out[33]=



Time of First Peak (T_p)

The time to first peak, T_p, is the time it takes the system to reach the first peak of oscillation

T_p = time elapsed before signal reaches first peak



We can calculate the time to the first peak by taking the derivative and solving for when it is zero.

In[34]:=
```
yStepTrigDot[t_] = D[yStepTrig[t], t] // Simplify
```

Out[34]= $\dfrac{A\, e^{-t\,\zeta\,\omega n}\,\omega n\,\mathrm{Sin}\left[t\,\sqrt{1-\zeta^2}\,\omega n\right]}{\sqrt{1-\zeta^2}}$

So we have

$$\dot{y}(t) = \frac{A\,\omega_n}{\sqrt{1-\zeta^2}}\, e^{-\zeta\,\omega_n\,t} \sin\left(\sqrt{1-\zeta^2}\;\omega_n\,t\right) \qquad \textbf{(Eq.6)}$$

Solving for when the derivative is 0,

$$\frac{A\,\omega_n}{\sqrt{1-\zeta^2}}\, e^{-\zeta\,\omega_n\,t} \sin\left(\sqrt{1-\zeta^2}\;\omega_n\,t\right) = 0 \qquad \text{note: } \frac{A\,\omega_n}{\sqrt{1-\zeta^2}}\, e^{-\zeta\,\omega_n\,t} \neq 0$$

$$\sin\left(\sqrt{1-\zeta^2}\;\omega_n\,t\right) = 0$$

This occurs when

$$\sqrt{1-\zeta^2}\;\omega_n\,t = k\,\pi \qquad k = 0,\ 1,\ 2,\ \ldots$$

Since we want the first peak, we choose $k = 1$ (since $k = 0$ corresponds to when the system is just starting)

$$T_p = \frac{\pi}{\sqrt{1-\zeta^2}\ \omega_n} \qquad \textbf{\textit{(Eq.7)}}$$

In[35]:=
```
Tp = π / (Sqrt[1 - ξ^2] ωn);
```

In[36]:=
```
yStepTrigDot[Tp]
```

Out[36]= 0

The interesting thing to notice is that the time to the first peak is not a function of the magnitude of the step.

Amplitude of First Peak (M_{P_t})

The amplitude at the first peak is simply the function evaluated at T_p

$$M_{P_t} = y(T_p)$$

In[37]:=
```
Mpt = yStepTrig[Tp] // Simplify
```

Out[37]= $A\left(1 + e^{-\frac{\pi \zeta}{\sqrt{1-\zeta^2}}}\right)$

So we have

$$M_{P_t} = A\left(1 + e^{-\frac{\pi \zeta}{\sqrt{1-\zeta^2}}}\right) \qquad \textbf{\textit{(Eq.8)}}$$

The interesting thing to notice is that the amplitude at the first peak not a function of ω_n. Note that this is only valid for the system we are considering, namely a 2nd order system with a DC gain of 1.

Percent Overshoot (PO)

The percent overshoot measures how much the system overshoots the steady state value.

$$\text{PO} = \left(\frac{M_{P_t} - y_{ss}}{y_{ss}}\right) \times 100$$

Recall that this system has a DC gain of 1, so $y_{ss} = A$

In[38]:=
```
PO = (Mpt - A)/A * 100 // Simplify
```

Out[38]= $100\, e^{-\frac{\pi \zeta}{\sqrt{1-\zeta^2}}}$

So we have

$$\text{PO} = 100\, e^{-\frac{\pi \zeta}{\sqrt{1-\zeta^2}}} \qquad \textbf{\textit{(Eq.9)}}$$

The interesting thing to notice is that the percent overshoot is only a function of the damping ratio.

Also note that it is not a function of the magnitude of the step.

Settling Time (T_s)

We can define the settling time as the time the response takes to remain within a certain percentage of the steady state value.

T_s = time elapsed before signal remains within the specified percentage of the final value



Recall that the time evolution of the system is given by

$$y(t) = A - A\, e^{-\sigma t}\left(\cos(\omega_d\, t) + \frac{\zeta}{\sqrt{1-\zeta^2}}\sin(\omega_d\, t)\right)$$

The deviation between the function and the steady state value can be defined as

$$g(t) = y(t) - y_{ss}(t)$$

In this situation, $y_{ss}(t) = A$

$$g(t) = A - A\, e^{-\sigma t}\left(\cos(\omega_d\, t) + \frac{\zeta}{\sqrt{1-\zeta^2}}\sin(\omega_d\, t)\right) - A$$

$$g(t) = -A\, e^{-\sigma t}\left(\cos(\omega_d\, t) + \frac{\zeta}{\sqrt{1-\zeta^2}}\sin(\omega_d\, t)\right)$$

We would like to know the time when the system trajectory stays within δ percent of the final value. In other words, we want to know when $|g(t)| < \delta\, y_{ss}$. In this case, $\delta \in (0, 1]$ is referred to as the settling fraction and denotes the fraction of settling time to within steady state (ie 0.05 ⇒ settle to within 5% of y_{ss}).

The decay envelope of this deviation function is given by

$$d(t) = \pm A\, e^{-\sigma t}$$

In[39]:=
```
d[t_] = -A Exp[-ξ ωn t];
Plot[{d[t] /. {A → Agiven, ξ → ξgiven, ωn → ωngiven},
  -d[t] /. {A → Agiven, ξ → ξgiven, ωn → ωngiven}}, {t, 0, 12},
 AxesLabel → {"t", "d(t)  (decay envelop of deviation function)"}]
```

Out[40]=



We know that the deviation will stay within this decay envelope. Therefore, a conservative estimate of the settling time (ie the actual settling time may be less) is when the decay envelop equals the point $\delta\, y_{ss}$. Let us denote this conservative estimate as $\tilde{T}_s$

$$|d(\tilde{T}_s)| = \delta\, y_{ss}$$

$$|d(\tilde{T}_s)| = \delta\, A$$

$$A\, e^{-\zeta\, \omega_n\, \tilde{T}_s} = \delta\, A$$

$$e^{-\zeta\, \omega_n\, \tilde{T}_s} = \delta$$

In[44]:=
```
Clear[TsTilde]
temp = Solve[Exp[-ξ ωn TsTilde] == δ, TsTilde];
TsTilde = TsTilde /. temp[[1]]
```

Out[46]=

$$\frac{2\, i\, \pi\, c_1 + \text{Log}\left[\frac{1}{\delta}\right]}{\xi\, \omega n} \quad \text{if } c_1 \in \mathbb{Z}$$

Note that if $\delta > 0$, $\ln\left(\frac{1}{\delta}\right) = -\ln(\delta)$

In[51]:=
```
Simplify[Log[1/δ], {δ > 0}]
```

Out[51]= $-\text{Log}[\delta]$

So the approximate settling time is given by

$$\tilde{T}_s = \frac{-\ln(\delta)}{\zeta\,\omega_n} \qquad \textbf{(Eq.10)}$$

where $\delta \in (0,\, 1]$

The interesting thing to notice is that settling time is not a function of the magnitude of the step.

Rise Time (T_r)

The rise time is defined as the time it takes for the signal to reach a certain percentage of its steady state value. This the time before the signal first reaches this value. Note that the signal is not required to remain within the bounds (as is the case with settling time). Sometimes this is defined as the time it takes for the signal to go from a certain start percentage to a certain end percentage of the final value

T_r =time elapsed between signal reaching specified start percentage of final value and signal reaching specified end percentage of final value

For example, in the figure below, the rise time is calculated as the time elapsed between when the signal reaches 10% and 90%



Recall that the time evolution of the system is given by

$$y(t) = A - A\,e^{-\sigma t}\left(\cos(\omega_d\,t) + \frac{\zeta}{\sqrt{1-\zeta^2}}\sin(\omega_d\,t)\right)$$

Since the steady state value is $y_{ss}(t) = A$, we the rise time (assuming starting from 0% of final value) is given by

$$y(T_r) = \delta\,A \qquad\qquad \delta \in (0,\,1)$$

$$A - A\,e^{-\sigma\,T_r}\left(\cos(\omega_d\,T_r) + \frac{\zeta}{\sqrt{1-\zeta^2}}\sin(\omega_d\,T_r)\right) = \delta\,A$$

$$1 - e^{-\sigma\,T_r}\left(\cos(\omega_d\,T_r) + \frac{\zeta}{\sqrt{1-\zeta^2}}\sin(\omega_d\,T_r)\right) = \delta$$

At this point, it appears the equation is difficult to solve directly. It is important to note that the solution does not depend on the magnitude of the step, so the rise time is not a function of *A*.

The rise time can be numerically calculated or obtained via a linear approximation.

Summary

$$T_p = \frac{\pi}{\sqrt{1-\zeta^2}\;\omega_n} \qquad\qquad \text{(time to first peak)}$$

$M_{P_t} = A\left(1 + e^{-\frac{\pi\zeta}{\sqrt{1-\zeta^2}}}\right)$ (magnitude at first peak)

$\text{PO} = 100\, e^{-\frac{\pi\zeta}{\sqrt{1-\zeta^2}}}$ (percent overshoot)

$\tilde{T}_s = \frac{-\ln(\delta)}{\zeta\,\omega_n}$ (approximate settling time)

T_r = <numerical calculation> (rise time)

This is sometimes referred to as **time domain analysis** because it deals with analyzing the performance of the system in the time domain.

Example: Simple System

Consider an example system with

$A = 2.3$
$\omega_n = 1.5$
$\zeta = 0.2$

In[52]:=
```
(*Input constants*)
Agiven = 2.3;
ωngiven = 1.5;
ξgiven = 0.2;

tFinal = 14;

(*compute performance metrics*)
Print["Tp"]
Tp = π/(Sqrt[1 - ξ^2] ωn) /. {A → Agiven, ξ → ξgiven, ωn → ωngiven}

Print["Mpt"]
Mpt = A (1 + Exp[-(π ξ)/Sqrt[1 - ξ^2]]) /. {A → Agiven, ξ → ξgiven, ωn → ωngiven}

Print["PO"]
PO = 100 Exp[-(π ξ)/Sqrt[1 - ξ^2]] /. {A → Agiven, ξ → ξgiven, ωn → ωngiven}

Print["T̃s"]
TsTilde = -Log[δ]/(ξ ωn) /. {A → Agiven, ξ → ξgiven, ωn → ωngiven, δ → 0.05}

(*Plot the output*)
Plot[yStep[t] /. {A → Agiven, ξ → ξgiven, ωn → ωngiven}, {t, 0, tFinal},
 PlotRange → All]
```

$$T_p = \frac{\pi}{\sqrt{1-\xi^2}\,\omega n}$$

$$Mpt = A\left(1 + \mathrm{Exp}\left[-\frac{\pi\,\xi}{\sqrt{1-\xi^2}}\right]\right)$$

$$PO = 100\,\mathrm{Exp}\left[-\frac{\pi\,\xi}{\sqrt{1-\xi^2}}\right]$$

$$TsTilde = \frac{-\mathrm{Log}[\delta]}{\xi\,\omega n}$$

Tp

Out[57]= 2.13758

Mpt

Out[59]= 3.51123

PO

Out[61]= 52.6621

T̃s

Out[63]= 9.98577

Out[64]=