Christopher Lum
lum@uw.edu

Lecture06b
The Divergence Theorem

Lecture is on YouTube

The YouTube video entitled ‘The Divergence Theorem’ that covers this lecture is located at https://youtu.be/y5rABxjF_o8.

Introduction

The Divergence Theorem (AKA Gauss’s Theorem or Ostrogradsky’s Theorem) relates the flux over a closed surface to the divergence in the volume.

https://en.wikipedia.org/wiki/Divergence_theorem

Prerequisites:
-Video entitled ‘Triple Integrals (AKA Volume Integrals)’ located at https://youtu.be/jd-0thQnddY.
-Video entitled ‘The Laplace Operator, Divergence, and Curl’ at https://youtu.be/KOlVHPShCOk).
-Video entitled ‘Surface Integrals of Scalar and Vector Fields/Functions’ at https://youtu.be/34Xfij-7gcI

Review of Triple Integrals

Recall that we discussed triple integrals in our video entitled ‘Triple Integrals (AKA Volume Integrals)’ located at https://youtu.be/jd-0thQnddY.

A **triple integral** is an integral of a function $f(x, y, z)$ taken over a closed bounded, three-dimensional region T in space. We subdivide T into boxes with planes parallel to the coordinate planes. Then we consider those boxes of the subdivision that lie entirely inside T, and number them from 1 to n. In each sub box we choose an arbitrary point (x_k, y_k, z_k) in box k. The volume of the box k we denote by ΔV_k. We now form the sum

$$J_n = \sum_{k=1}^{n} f(x_k, y_k, z_k)\, \Delta V_k$$

Which is imagined as the small volume weighted by the function at that point.

As these boxes get smaller and smaller towards 0, we have the triple integral

$$\iiint_T f(x, y, z)\, dx\, dy\, dz = \iiint_T f(x, y, z)\, dV$$

Note that many scenarios may lend themselves to evaluating the triple integral in cylindrical or spherical coordinates.

For example, converting a triple integral into spherical coordinates yields

$$\iiint_T f(x, y, z)\, dV = \iiint_D f(r \cos(\theta) \sin(\phi), r \sin(\theta) \sin(\phi), r \cos(\phi))\, r^2 \sin(\phi)\, dr\, d\theta\, d\phi$$

Example: Triple Integral of a Sphere

The volume of a sphere with radius R can be calculated by integrating the constant function 1 over the sphere

$$V = \iiint_T f(x, y, z)\, dx\, dy\, dz$$

It is easier to change to spherical coordinates

$$= \iiint_D f(r \cos(\theta) \sin(\phi), r \sin(\theta) \sin(\phi), r \cos(\phi))\, r^2 \sin(\phi)\, dr\, d\theta\, d\phi$$

$$= \iiint_D r^2 \sin(\phi)\, dr\, d\theta\, d\phi$$

Note that the sphere is described by $r \in [0, R]$, $\theta \in [0, 2\pi]$, $\phi \in [0, \pi]$

In[1]:=
```
polarAngleMax = 2 π;  (*We call this ϕ*)
azimuthAngleMax = π;  (*We call this θ*)

SphericalPlot3D[r /. {r → 1},
 {polarAngle, 0, polarAngleMax}, {azimuthAngle, 0, azimuthAngleMax},
 AxesLabel → {"x", "y", "z"}]
Clear[polarAngleMax, azimuthAngleMax, R]
```

Out[3]=



so we have

$$= \int_0^{\pi}\int_0^{2\pi}\int_0^{R} r^2 \sin(\phi)\, dr\, d\theta\, d\phi$$

$$= \int_0^{\pi}\int_0^{2\pi} \frac{R^3 \sin(\phi)}{3}\, d\theta\, d\phi$$

$$= \int_0^{\pi} \frac{2\pi R^3 \sin(\phi)}{3}\, d\phi$$

$$V = \frac{4}{3}\pi R^3$$

which is our familiar volume of a sphere formula.

```
In[5]:= Integrate[
  Integrate[
   Integrate[r^2 Sin[ϕ],
    {r, 0, R}],
   {θ, 0, 2 π}],
  {ϕ, 0, π}
 ]
```

Out[5]= $\frac{4 \pi R^3}{3}$

We will investigate another example of evaluating a triple integral in Cartesian coordinates in the following section.

Divergence Theorem of Gauss

Triple integrals can be transformed into surface integrals over the boundary surface of a region in space (and vice versa). This is useful because often, one type of integral is easier to evaluate than the other. This transformation is done via the **divergence theorem**, which involves the divergence of a vector function $\overline{F} = [F_1, F_2, F_3]$. Recall that the divergence of a vector field is given by (see our video entitled 'The Laplace Operator, Divergence, and Curl' at https://youtu.be/KOlVHPShCOk).

$$\text{div}\,\overline{F} = \frac{\partial F_1}{\partial x} + \frac{\partial F_2}{\partial y} + \frac{\partial F_3}{\partial z}$$

Theorem 1: Divergence Theorem of Gauss

Let T be a closed bounded region in space whose boundary is a piecewise, smooth, orientable surface S. Let $\overline{F}(x, y, z)$ be a vector function that is continuous and has continuous first partial derivatives in some domain containing T. Then

$$\iiint_T \text{div}\,\overline{F}\, dV = \iint_S \overline{F} \cdot \overline{n}\, dA \qquad \textbf{\textit{(Eq.2)}}$$

In a loose sense, the left side effectively measures how much of $\overline{F}$ is generated within the region T and the right side measures how much of $\overline{F}$ crosses the surface S.

Further note that the normal vector $\overline{n}$ is oriented outwards from the body T.

Example 1

Example 1: Evaluation of a Surface Integral by the Divergence Theorem (Simple Example)

In this example, we use the Divergence Theorem to avoid calculating a surface integral and instead calculate a volume integral.

Evaluate

$$I = \iint_S \overline{F} \cdot \overline{n}\, dA$$

where $\overline{F} = 2x\,\hat{i} + y^2\,\hat{j} + z^2\,\hat{k} = \begin{pmatrix} 2x \\ y^2 \\ z^2 \end{pmatrix}$

S = unit sphere defined by $x^2 + y^2 + z^2 = 1$

In[6]:=
```
F[x_, y_, z_] = {2 x, y^2, z^2};
```

We can visualize the scenario

In[7]:=
```
min = -1.25;
max = 1.25;

Show[
 (*Plot the surface*)
 ContourPlot3D[x^2 + y^2 + z^2 == 1, {x, min, max}, {y, min, max}, {z, min, max}],

 (*Plot the vector field*)
 VectorPlot3D[F[x, y, z], {x, min, max}, {y, min, max}, {z, min, max}],

 (*Plot options*)
 AxesLabel → {"x", "y", "z"}
]
```

Out[9]=



The direct evaluation of the surface integral is quite difficult. But we can simplify the problem by applying the divergence theorem to relate the surface integral to a triple integral

$$\iint_S \overline{F} \cdot \overline{n} \, dA = \iiint_T \operatorname{div} \overline{F} \, dV$$

$$= \iiint_T \frac{\partial}{\partial x}[2x] + \frac{\partial}{\partial y}\left[y^2\right] + \frac{\partial}{\partial z}\left[z^2\right] dV$$

$$= \iiint_T 2 + 2y + 2z \, dV$$

$$= 2\iiint_T dV + 2\iiint_T y \, dV + 2\iiint_T z \, dV$$

We can examine each of these. The first term we recognize as simply the volume of the sphere (since we integrate a constant value of 1 over the volume of the sphere). With the second term, we realize that the function y is positive in the left hemisphere of the sphere and negative in the other hemisphere with equal magnitude. Therefore, this term evaluates to 0. The same is true for the third term which looks at integrating the function z over the sphere. Therefore, only the first term contributes as we see that the total answer is twice the volume of a sphere

$$\iint_S \overline{F} \cdot \overline{n} \, dS = 2\iiint_T dV = \frac{8\pi}{3}$$

In[10]:=
```
Clear[max, min, F]
```

Example 2

Example 2: Evaluation of a Surface Integral by the Divergence Theorem (More Complex Example)

In this example, we use the Divergence Theorem to avoid calculating a surface integral and instead calculate a volume integral.

Evaluate

$$I = \iint_S x^3 \, dy \, dz + x^2 y \, dz \, dx + x^2 z \, dx \, dy$$

where S is the close surface consisting of
- -The cylinder $x^2 + y^2 = a^2$ with $z \in [0, b]$ (surface S_1) (sides of the can)
- -The circular disk $x^2 + y^2 \leq a^2$ with $z = 0$ (surface S_2) (bottom of the can)
- -The circular disk $x^2 + y^2 \leq a^2$ with $z = b$ (surface S_3) (top of the can)

Note that we can use Mathematica's 'RegionPlot3D' command to visualize the surfaces S_2 and S_3. We do this by approximating these surfaces as volumes with finite thicknesses as shown below.

In[11]:=
```
Manipulate[
 Show[
  (*Plot S1 (side of the can)*)
  ContourPlot3D[x^2 + y^2 == a^2, {x, -a, a},
   {y, -a, a}, {z, 0, b}, AxesLabel → {"x", "y", "z"}],

  (*Plot S2 and S3 (bottom and top of the can)*)
  RegionPlot3D[x^2 + y^2 ≤ a^2, {x, -a, a}, {y, -a, a}, {z, 0, 0.001}],
  RegionPlot3D[x^2 + y^2 ≤ a^2, {x, -a, a}, {y, -a, a}, {z, b, b + 0.001}]
 ],
 {a, 1, 3},
 {b, 1, 5}
]
```

Out[11]=



In[12]:=
```
Clear[a, b]
```

From Eq.5 from our video entitled ‘Surface Integrals’ at https://youtu.be/34Xfij-7gcI , we see that we can write the surface integral as

$$\iint_S (F_1\, dy\, dz + F_2\, dz\, dx + F_3\, dx\, dy) = \iint_S \overline{F} \cdot \overline{n}\, dA$$

where $\overline{F} = < x^3, x^2 y, x^2 z >$

```
In[13]:= F[x_, y_, z_] = {x^3, x^2 y, x^2 z};
```

The divergence theorem states that we can write this surface integral as a volume integral

$$\iint_S \overline{F} \cdot \overline{n}\, dA = \iiint_T \operatorname{div} \overline{F}\, dV$$

We can now evaluate the triple integral as it may be easier than the surface integral. We first compute the divergence of $\overline{F}$

```
In[14]:= D[F[x, y, z][[1]], x] + D[F[x, y, z][[2]], y] + D[F[x, y, z][[3]], z]
Out[14]= 5 x^2
```

So we have

$$\iiint_T \operatorname{div} \overline{F}\, dV = \iiint_T 5\, x^2\, dV$$

The text shows how to evaluate this triple integral by switching to polar coordinates. For practice, we will use an alternative approach and first treat the ends as type I regions and then perform 3 successive integrations



So we see that in the XY-plane we have

$$R = \{(x, y) \mid -a \le x \le a,\ g(x) \le y \le h(x)\} \qquad \text{(Type I)}$$

where $g(x) = -\sqrt{a^2 - x^2}$
$h(x) = \sqrt{a^2 - x^2}$

So the triple integral becomes

$$\iiint_T 5\, x^2\, dV = \int_0^b \int_{-a}^{a} \int_{-\sqrt{a^2-x^2}}^{\sqrt{a^2-x^2}} 5\, x^2\, dy\, dx\, dz$$

In[15]:=
```
t1 = Integrate[5 x^2, {y, -Sqrt[a^2 - x^2], Sqrt[a^2 - x^2]}]
```

Out[15]= $10\,x^2\,\sqrt{a^2 - x^2}$

So we have

$$= \int_0^b \int_{-a}^{a} 10\,x^2\,\sqrt{a^2 - x^2}\,dx\,dz$$

In[16]:=
```
t2 = Simplify[Integrate[t1, {x, -a, a}], a > 0]
```

Out[16]= $\frac{5\,a^4\,\pi}{4}$

So we have

$$= \int_0^b \frac{5}{4}\,a^4\,\pi\,dz$$

In[17]:=
```
t3 = Integrate[t2, {z, 0, b}]
```

Out[17]= $\frac{5}{4}\,a^4\,b\,\pi$

So we obtain

$$\iiint_T 5\,x^2\,dV = \frac{5}{4}\,a^4\,b\,\pi$$

which is the same result if we had converted to polar coordination to perform the integration.

In[18]:=
```
Clear[t3, t2, t1, F]
```

Example 3

Example: Adding Surfaces to Create a Closed Surface

Consider the vector field

$$\overline{F} = <3\,x\,e^z, -3\,y\,e^z, x^2>$$

In[19]:=
```
F[x_, y_, z_] = {3 x Exp[z], -3 y Exp[z], x^2};
```

Consider the top half of the unit circle defined by

$$S_1 = \left\{<x, y, z> \,\middle|\, x^2 + y^2 + z^2 = 1,\ z \geq 0\right\}$$

In[20]:=
```
ContourPlot3D[x^2 + y^2 + z^2 == 1, {x, -1, 1}, {y, -1, 1}, {z, 0, 1},
 AxesLabel → {"x", "y", "z"}, PlotRange → {{-1, 1}, {-1, 1}, {-1, 1}}]
```

Out[20]=



The surface is oriented upward (so the normal points approximately in the positive z direction). Note that we say oriented "upward" instead of "outward" because this is an open surface so the concept of inward/outward is not applicable. This is an open surface because we do not include the "bottom lid".

We would like to compute $\iint_{S_1} \overline{F} \cdot \overline{n}\, dA$.

Direct calculation of this is somewhat cumbersome because $\overline{F}$ is somewhat complex. Furthermore, since the surface S_1 does not enclose a closed volume, we cannot apply the Divergence Theorem and must compute the cumbersome surface integral. However, notice that if the surface had included the "bottom lid", then it would be closed and we could use the Divergence Theorem.

To facilitate this, we consider adding a surface S_2 and now considered the alternative problem of computing

$$\iint_S \overline{F} \cdot \overline{n}\, dA$$

where $S = S_1 \cup S_2$

$S_2 = \left\{ < x, y, z > \mid x^2 + y^2 \leq 1, z = 0 \right\}$

In this case, we see that $S = S_1 \cup S_2$ does indeed enclose a closed volume so we can apply the Divergence Theorem.

$$\iint_S \overline{F} \cdot \overline{n}\, dA = \iiint_T \text{div}\, \overline{F}\, dV$$

Writing this out

$$\iint_{S_1 \cup S_2} \overline{F} \cdot \overline{n}\, dA = \iiint_T \text{div}\, \overline{F}\, dV$$

$$\iint_{S_1} \overline{F} \cdot \overline{n}\, dA + \iint_{S_2} \overline{F} \cdot \overline{n}\, dA = \iiint_T \text{div}\, \overline{F}\, dV$$

$$\iint_{S_1} \overline{F} \cdot \overline{n}\, dA = \iiint_T \text{div}\, \overline{F}\, dV - \iint_{S_2} \overline{F} \cdot \overline{n}\, dA$$

At this point, you might ask if we have really saved any work because we still need to compute $\iint_{S_2} \overline{F} \cdot \overline{n}\, dA$ with a fairly complex $\overline{F}$. However, the surface S_2 is significantly simpler because $z = 0$ on this surface. We can parameterize this surface easily using polar coordinates

$$S_2 = \overline{r}_2(u, v) = < u \cos(v),\, u \sin(v),\, 0 >$$

$u \in [0, 1]$
$v = [0, 2\pi]$

```
In[21]:= r2[u_, v_] = {u Cos[v], u Sin[v], 0};
```

So the surface integral over S_2 is given as

$$\iint_{S_2} \overline{F} \cdot \overline{n}\, dA = \iint_{R_2} \overline{F}(\overline{r}_2(u, v) \cdot \overline{N}_2(u, v)\, du\, dv$$

where $\quad \overline{N}_2 = -(\overline{r}_{2,u} \times \overline{r}_{2,v})$

Here, make special note of the minus sign that is used to ensure that the surface normal is in same "outward" orientation as in the triple integral over the surface.

```
In[22]:= N2[u_, v_] = -Cross[D[r2[u, v], u], D[r2[u, v], v]] // Simplify
Out[22]= {0, 0, -u}
```

We can draw this scenario to illustrate the surface S_2 and the surface normal $\overline{N}_2$ (note that the pdf version of the notes may not display the graphics properly but the following source code is correct and will generate the appropriate graphic when entered into Mathematica).

In[23]:=
```
Manipulate[
 (*What is the equivalent xyz point*)
 r2Given = r2[uGiven, vGiven];

 (*What is the surface normal at this point?*)
 N2Given = N2[uGiven, vGiven];

 (*Plot the scenario*)
 Show[
  ParametricPlot3D[r2[u, v], {u, 0, 1}, {v, 0, 2 π}],

  (*Plot the surface normal*)
  Graphics3D[
   {
    Magenta, Arrow[{{r2Given[[1]], r2Given[[2]], r2Given[[3]]},
      {r2Given[[1]] + N2Given[[1]], r2Given[[2]] + N2Given[[2]], r2Given[[3]] + N2Given[[3]]}}]
   }
  ],

  (*Add plot information*)
  AxesLabel → {"x", "y", "z"},
  PlotLabel → {"Surface S2 Surface Normal N2"},
  PlotRange → {{-1, 1}, {-1, 1}, {-1, 1}}
 ],

 (*Manipulate u and v*)
 {uGiven, 0, 1},
 {vGiven, 0, 2 π}

]
```

Out[23]=



We can now evaluate the vector field along this surface

$$\overline{F}(\overline{r}_2(u, v) = <3\,x\,e^z, -3\,y\,e^z, x^2> \qquad \text{recall: } x = u\cos(v),\ y = u\sin(v),\ z = 0$$

In[24]:=
```
F[u Cos[v], u Sin[v], 0] // Simplify
```

Out[24]=
```
{3 u Cos[v], -3 u Sin[v], u^2 Cos[v]^2}
```

So we have

$$\int\int_{S_2} \overline{F}\cdot\overline{n}\,dA = \int\int_{R_2} \overline{F}(\overline{r}_2(u, v)\cdot\overline{N}_2(u, v)\,du\,dv$$

$$= \int\int_{R_2} <3\,u\cos(v), -3\,u\sin(v), u^2\cos^2(v)> \cdot <0, 0, -u> du\,dv$$

$$= \int_0^{2\pi}\int_0^1 -u^3\cos^2(v)\,du\,dv$$

$$= \int_0^{2\pi} \frac{-1}{4}\cos^2(v)\,dv$$

$$\iint_{S_2} \overline{F} \cdot \overline{n}\, dA = -\frac{\pi}{4}$$

In[25]:=
```
t1 = F[u Cos[v], u Sin[v], 0].N2[u, v]
t2 = Integrate[t1, {u, 0, 1}]
t3 = Integrate[t2, {v, 0, 2 π}]
```

Out[25]= $-u^3 \text{Cos}[v]^2$

Out[26]= $-\frac{1}{4} \text{Cos}[v]^2$

Out[27]= $-\frac{\pi}{4}$

We can now turn our attention to the term $\iiint_T \text{div}\, \overline{F}\, dV$. We first compute the divergence, $\text{div}\, \overline{F} = \frac{\partial F_1}{\partial x} + \frac{\partial F_2}{\partial y} + \frac{\partial F_3}{\partial z}$

In[28]:=
```
Div[F[x, y, z], {x, y, z}]
```

Out[28]= 0

So we see that

$$\iiint_T \text{div}\, \overline{F}\, dV = 0$$

So substituting back into our original expression

$$\iint_{S_1} \overline{F} \cdot \overline{n}\, dA = \iiint_T \text{div}\, \overline{F}\, dV - \iint_{S_2} \overline{F} \cdot \overline{n}\, dA$$

$$= 0 - \left(\frac{-\pi}{4}\right)$$

$$\iint_{S_1} \overline{F} \cdot \overline{n}\, dA = \frac{\pi}{4}$$

Brute Force Approach: Direct Calculation of Surface Integral

We can verify this by direct calculation of the original surface integral

$$S_1 = \left\{ <x, y, z> \,\middle|\, x^2 + y^2 + z^2 = 1,\ z \geq 0 \right\}$$

We can parametrize this surface using spherical coordinates

$$S_1 = \overline{r}_1(u, v) = \ <\cos(u) \sin(v),\ \sin(u) \sin(v),\ \cos(v)>$$

$u \in [0, 2\pi]$ (typically referred to as θ in other literature)

$v = [0, \pi/2]$ (typically referred to as ϕ in other literature)

In[29]:=
```
r1[u_, v_] = {Cos[u] Sin[v], Sin[u] Sin[v], Cos[v]};
```

So the surface integral over S_1 is given as

$$\iint_{S_1} \overline{F} \cdot \overline{n}\, dA = \iint_{R_1} \overline{F}\big(\overline{r}_1(u, v) \cdot \overline{N}_1(u, v)\, du\, dv$$

where $\overline{N}_1 = -\left(\overline{r}_{1,u} \times \overline{r}_{1,v}\right)$

Here, make special note of the minus sign that is used to ensure that the surface normal is in same "outward" orientation as in the triple integral over the surface.

In[30]:=
```
N1[u_, v_] = -Cross[D[r1[u, v], u], D[r1[u, v], v]] // Simplify
```

Out[30]=
```
{Cos[u] Sin[v]^2, Sin[u] Sin[v]^2, Cos[v] Sin[v]}
```

We can draw this scenario to illustrate the surface S_1 and the normal vector $\overline{N}_1$ (note that the pdf version of the notes may not display the graphics properly but the following source code is correct and will generate the appropriate graphic when entered into Mathematica).

In[31]:=
```
Manipulate[
 (*What is the equivalent xyz point*)
 r1Given = r1[uGiven, vGiven];

 (*What is the surface normal at this point?*)
 N1Given = N1[uGiven, vGiven];

 (*Plot the scenario*)
 Show[
  ParametricPlot3D[r1[u, v], {u, 0, 2 π}, {v, 0, π / 2}],

  (*Plot the surface normal*)
  Graphics3D[
   {
    Magenta, Arrow[{{r1Given[[1]], r1Given[[2]], r1Given[[3]]},
      {r1Given[[1]] + N1Given[[1]], r1Given[[2]] + N1Given[[2]], r1Given[[3]] + N1Given[[3]]}}]
   }
  ],

  (*Add plot information*)
  AxesLabel → {"x", "y", "z"},
  PlotLabel → {"Surface S1 Surface Normal N1"},
  PlotRange → {{-2, 2}, {-2, 2}, {-2, 2}}
 ],

 (*Manipulate u and v*)
 {uGiven, 0, 2 π},
 {vGiven, 0, π / 2}
]
```

Out[31]=



We can now evaluate the vector field along this surface

$$\overline{F}(\overline{r}(u, v)) = <3\,x\,e^z, -3\,y\,e^z, x^2> \qquad \text{recall: } x = \cos(u)\sin(v),\ y = \sin(u)\sin(v),\ \text{and } z = \cos(v)$$

$$S_1 = \overline{r}_1(u, v) = <\cos(u)\sin(v), \sin(u)\sin(v), \cos(v)>$$

In[32]:=
```
F[Cos[u] Sin[v], Sin[u] Sin[v], Cos[v]] // Simplify
```

Out[32]= $\{3\, e^{\mathrm{Cos}[v]}\, \mathrm{Cos}[u]\, \mathrm{Sin}[v], -3\, e^{\mathrm{Cos}[v]}\, \mathrm{Sin}[u]\, \mathrm{Sin}[v], \mathrm{Cos}[u]^2\, \mathrm{Sin}[v]^2\}$

In[33]:=
```
N1[u, v]
```

Out[33]= $\{\mathrm{Cos}[u]\, \mathrm{Sin}[v]^2, \mathrm{Sin}[u]\, \mathrm{Sin}[v]^2, \mathrm{Cos}[v]\, \mathrm{Sin}[v]\}$

So we have

$$\iint_{S_1} \overline{F}\cdot\overline{n}\, dA = \iint_{R_1} \overline{F}(\overline{r}_1(u, v)\cdot\overline{N}_1(u, v)\, du\, dv$$

$$= \iint_{R_1} <3\, e^{\cos(v)}\cos(u)\sin(v), -3\, e^{\cos(v)}\sin(u)\sin(v),$$
$$\cos^2(u)\sin^2(v)> \cdot <\cos(u)\sin^2(v), \sin(u)\sin^2(v), \cos(v)\sin(v)>du\, dv$$

$$= \int_0^{\pi/2} \int_0^{2\pi} 3\, e^{\cos(v)} \cos^2(u) \sin^3(v) + \cos^2(u) \cos(v) \sin^3(v) - 3\, e^{\cos(v)} \sin^2(u) \sin^3(v)\, du\, dv$$

$$= \int_0^{\pi/2} \pi \cos(v) \sin^3(v)\, dv$$

$$\iint_{S_1} \overline{F} \cdot \overline{n}\, dA = \frac{\pi}{4}$$

In[34]:=
```
t1 = F[Cos[u] Sin[v], Sin[u] Sin[v], Cos[v]].N1[u, v]
t2 = Integrate[t1, {u, 0, 2 π}]
t3 = Integrate[t2, {v, 0, π / 2}]
```

Out[34]= $3\, e^{\text{Cos}[v]}\, \text{Cos}[u]^2\, \text{Sin}[v]^3 + \text{Cos}[u]^2\, \text{Cos}[v]\, \text{Sin}[v]^3 - 3\, e^{\text{Cos}[v]}\, \text{Sin}[u]^2\, \text{Sin}[v]^3$

Out[35]= $\pi\, \text{Cos}[v]\, \text{Sin}[v]^3$

Out[36]= $\frac{\pi}{4}$

This is the same as we obtained previously, albeit with a lot more complicated work involved.

Theorem: Creating a Closed Surface

Let S_1 be a surface (not closed) and S_2 be another surface such that $S_1 \cup S_2$ is a closed surface. Let $S_1 \cup S_2$ be oriented outward. Let $\overline{F}$ be a well behaved vector field on $S_1 \cup S_2$. Let T be the region enclosed by $S_1 \cup S_2$. Then

$$\iint_{S_1} \overline{F} \cdot d\overline{s} = \iiint_T \text{div}\, \overline{F}\, dV - \iint_{S_2} \overline{F} \cdot d\overline{s}$$

Once again, this is useful if $\iint_{S_1} \overline{F} \cdot d\overline{s}$ is difficult to calculate but $\iiint_T \text{div}\, \overline{F}\, dV$ and $\iint_{S_2} \overline{F} \cdot d\overline{s}$ are easier to calculate.

Discussion of the Divergence Theorem

The Divergence Theorem can be useful if $\overline{F}$ is somewhat complicated. This is because div $\overline{F}$ involves taking derivatives of $\overline{F}$ and therefore, div $\overline{F}$ is usually simpler than $\overline{F}$. Therefore, if tasked with computing $\iint_S \overline{F} \cdot \overline{n}\, dA$, it may be useful to apply the divergence theorem $\iint_S \overline{F} \cdot \overline{n}\, dA = \iiint_T \text{div}\, \overline{F}\, dV$ in hopes that the integrand becomes simpler. Of course, the trade off is that instead of a surface integral, we now must compute a triple integral.

We can look at some examples where the Divergence Theorem might be useful to help understand some engineering phenomena.

Example 1: Fluid Flow. Physical Interpretation of the Divergence

Let S be the boundary surface of a region T is space, and let $\overline{n}$ be the outer unit normal vector of S. Then $\overline{v} \cdot \overline{n}$ is the normal component of $\overline{v}$ in the direction of $\overline{n}$. If $\overline{v} \cdot \overline{n} > 0$, this implies that the fluid flow is roughly aligned with the outward normal, so fluid is flowing out of the volume. Similarly, if $\overline{v} \cdot \overline{n} < 0$, then the velocity is roughly against the outward normal, so fluid is entering the volume.

Hence the total mass of fluid that leaves T by flowing across S is given by

$$\iint_S \overline{v} \cdot \overline{n}\, dA$$

Dividing by the volume V of T gives the average flow out of T

$$\frac{1}{V} \iint_S \overline{v} \cdot \overline{n}\, dA \qquad \textbf{(Eq.1)}$$

If we have steady flow and we assume the fluid to be incompressible, conservation of mass states that the flow across S should be 0. If it is not 0, then there must be a source or sink inside of T

Example 2: Physical Meaning of the Divergence of Velocity

We now have the machinery in place to show that if we have a vector field describing the velocity of a fluid flow, $\overline{v}$, then div $\overline{v} = \nabla \cdot \overline{v}$ (divergence of $\overline{v}$) is physically the time rate of change of the volume of a moving fluid element, per unit volume. Recall that we examined this earlier in Lecture04 when discussing the divergence operator.

The following discussion is from 'Anderson - Fundamentals of Aerodynamics 4th Ed, section 2.3.4'.

Consider a control volume moving with the fluid. This control volume is always made up of the same fluid particles as it moves with the flow as shown below



Finite control volume moving with the fluid such that the same fluid particles are always in the same control volume

Hence its mass is fixed. However, its volume T and control surface S are changing with time as it moves to different regions of the flow where different values of ρ exist. That is, this moving control volume of fixed mass is constantly increasing or decreasing its volume and is changing its shape, depending on the characteristics of the flow. This control volume is shown below



Consider an infinitesimal element of the surface dS moving at the local velocity $\overline{v}$. From the figure, we can see that the change in volume of the control volume ΔT due to just the movement of dS over a time increment Δt is equal to the volume of the long, thin cylinder with base area dS and altitude $(\overline{v}\,\Delta t)\cdot\overline{n}$

$$\Delta T = [(\overline{v}\,\Delta t)\cdot\overline{n}]\,dS = \overline{v}\,\Delta t\cdot d\overline{S} \qquad \textbf{\textit{(Eq.A.1)}}$$

Over the time increment Δt, the total change in volume of the whole control volume is equal to the summation of Eq.A.1 over the total control surface. In the limit as $dS \to 0$, the sum becomes the surface integral

$$\text{total volume change} = \iint_S [(\overline{v}\,\Delta t)\cdot\overline{n}]\,dS$$

Dividing by time yields the rate of change which we denote as DT/Dt (the 'D' notation denotes a substantial derivative which denotes how this changes as we move along with the element as opposed to looking at a fixed point in the fluid, for further discussion on this see Anderson section 2.9).

$$\frac{DT}{Dt} = \iint_S \overline{v}\cdot\overline{n}\,dS$$

We can now apply the Divergence Theorem $\left(\iint_S \overline{F}\cdot\overline{n}\,dA = \iiint_T \operatorname{div}\overline{F}\,dV\right)$

$$\frac{DT}{Dt} = \iiint_T \operatorname{div}\overline{v}\,dT$$

Now, let us imagine that the moving control volume is shrunk to a very small volume δT, essentially becoming an infinitesimal moving fluid element. Then this can be written as

$$\frac{D}{Dt}[\delta T] = \iiint_{\delta T} \operatorname{div}\overline{v}\,d\delta T$$

If we assume that δT is small enough such that $\operatorname{div}\overline{v}$ is essentially constant throughout the volume δT, then it can be brought outside the integral (we now have an approximation as opposed to an equality)

$$\frac{D}{Dt}[\delta T] = \operatorname{div}\overline{v}\iiint_{\delta T} d\delta T$$

$$= \operatorname{div}\overline{v}\,\delta T$$

$$\text{div}\,\overline{V} = \frac{1}{\delta T}\frac{D}{Dt}[\delta\text{T}] \qquad \textbf{\textit{(Eq.A.2)}}$$

Examining Eq.A.2 we see that it states that div $\overline{V}$ is physically the time rate of change of the volume of a moving fluid element, per unit volume.