Christopher Lum
lum@uw.edu

Lecture05f
Surface Integrals of Scalar and Vector Fields/Functions

The YouTube video entitled ‘Surface Integrals of Scalar and Vector Fields/Functions’ that covers this lecture is located at https://youtu.be/34Xfij-7gcI.

Introduction

Line integrals integrated functions over curves in space. With surface integrals, we integrate functions over surfaces in space. To do so, we need to find a parametric representation for the surfaces in space. This was described in the YouTube video entitled ‘Parameterizing Surfaces and Computing Surface Normal Vectors’ at https://youtu.be/a3_c4c9PYNg

We define a surface integral in a similar fashion to that of a line integral in the sense that we can either integrate a scalar or vector function over the surface.

Surface Integrals of Scalar Functions

For a given scalar function f, we define the surface integral over a surface S as

$$\text{surface integral } = \iint_S f \, ds$$

where $\quad f$ = scalar field/function

If the surface S is parameterized as we discussed previously ($S = \bar{r}(u, v)$ and u, v varies over a region R in the uv plane), then we can write this as

$$\iint_S f \, ds = \iint_R f(\bar{r}(u, v)) \cdot \left| \bar{N}(u, v) \right| du \, dv \qquad \textbf{\textit{(Eq.1)}}$$

where $\quad \bar{N} = \bar{r}_u \times \bar{r}_v = \left| \bar{N} \right| \bar{n}$ (surface normal vector, <u>not</u> normalized)

Example: Surface Area

Consider the hemisphere example from our previous video (see ‘Parameterizing Surfaces and Computing Surface Normal Vectors‘ at https://youtu.be/a3_c4c9PYNg)

$$\bar{r}(u, v) = \begin{pmatrix} a\sin(u)\cos(v) \\ a\sin(u)\sin(v) \\ a\cos(u) \end{pmatrix}$$

$u \in [0, \pi/2]$
$v \in [0, 2\pi]$

In[1]:=
```
r[u_, v_] = {a Sin[u] Cos[v], a Sin[u] Sin[v], a Cos[u]};

a = 5;
ParametricPlot3D[r[u, v], {u, 0, π / 2}, {v, 0, 2 π},
 AxesLabel → {"x", "y", "z"}]
Clear[a]
```

Out[3]=


If we consider the function $f = 1$ then the surface integral of $\iint_S f\, ds$ actually gives the surface area.

```
In[5]:= ru[u_, v_] = D[r[u, v], u]
        rv[u_, v_] = D[r[u, v], v]
        NVector[u_, v_] = Cross[ru[u, v], rv[u, v]] // Simplify
        MagNVector[u_, v_] =
         Simplify[((NVector[u, v][[1]])^2 + (NVector[u, v][[2]])^2 + (NVector[u, v][[3]])^2)^(1/2), {a > 0}]
```

Out[5]= $\{a \cos[u] \cos[v], a \cos[u] \sin[v], -a \sin[u]\}$

Out[6]= $\{-a \sin[u] \sin[v], a \cos[v] \sin[u], 0\}$

Out[7]= $\{a^2 \cos[v] \sin[u]^2, a^2 \sin[u]^2 \sin[v], a^2 \cos[u] \sin[u]\}$

Out[8]= $a^2 \sqrt{\sin[u]^2}$

So we have

$$\text{surface area} = \iint_S f\, ds$$

$$= \iint_R f(\bar{r}(u, v) \cdot |\bar{N}(u, v)|\, du\, dv$$

$$= \int_0^{2\pi} \int_0^{\pi/2} 1 \cdot a^2 (\sin(u)^2)^{1/2}\, du\, dv$$

```
In[9]:= term1 = Integrate[1 * MagNVector[u, v], {u, 0, π / 2}]
        Integrate[term1, {v, 0, 2 π}]
```

Out[9]= a^2

Out[10]= $2 a^2 \pi$

This agrees with the well known formula that states the surface area of a sphere is $4 \pi a^2$. In this case, we only have half a sphere so we only have half the surface area. Note that this gives the surface area of one side of the surface, it does not give the “total wetted area”.

Example: Mass of Sheet

Recall the line integral example where we computed the total mass of a wire with varying density using a line integral. Let us extend this idea to a surface. In this case, instead of a wire with varying linear density, let’s consider a piece of non-homogeneous sheet metal where the mass varies over the surface. Consider the specific density in units of kg/m^2 of

$$\rho(u, v) = 3u + v$$

And a radius of

$$a = 5$$

```
In[11]:= (*Define density function*)
         ρ[u_, v_] = 3 u + v;
```

```
(*Chose a*)
a = 5;

(*Manipulate multiple Show plots simultaneously.  These need
 to be placed inside a Grid in order to manipulate simulataneously*)
Manipulate[
 Grid[
  {{"Plot3D of ρ(u,v)", "Density Plot of ρ(u,v)", "ParametricPlot3D of r(u,v)"}, {Show[
      (*Plot ρ using Plot3D*)
      Plot3D[ρ[u, v], {u, 0, π / 2}, {v, 0, 2 π},
       AxesLabel → {"u", "v", "ρ(u,v)"},
       PlotRange → All],

      (*Plot the point of interest*)
      Graphics3D[
       {
        AbsolutePointSize[15], Green, Point[{uGiven, vGiven, ρ[uGiven, vGiven]}]
       }
      ]
     ], Show[
      (*Plot ρ using DensityPlot*)
      DensityPlot[ρ[u, v], {u, 0, π / 2}, {v, 0, 2 π},
       AxesLabel → {"u", "v"},
       PlotLegends → Automatic,
       FrameLabel → {"u", "v"},
       PlotRange → All],

      (*Plot the point of interest*)
      Graphics[
       {
        AbsolutePointSize[15], Green, Point[{uGiven, vGiven}]
       }
      ]
     ],
     Show[
      (*Plot surface using ParametricPlot3D*)
      ParametricPlot3D[r[u, v], {u, 0, π / 2}, {v, 0, 2 π},
       AxesLabel → {"x", "y", "z"}],

      (*Plot the point of interest*)
      Graphics3D[
       {
        AbsolutePointSize[15], Green,
        Point[{r[uGiven, vGiven]〚1〛, r[uGiven, vGiven]〚2〛, r[uGiven, vGiven]〚3〛}]
       }
      ]
```

```
    ]}}],
 {uGiven, 0, π / 2},
 {vGiven, 0, 2 π}
]
```

Out[13]=



So we have

$$\text{total mass} = \int\int_S f \, ds$$

$$= \int\int_R f(\bar{r}(u, v)) \cdot \left| \bar{N}(u, v) \right| du \, dv$$

$$= \int_0^{2\pi} \int_0^{\pi/2} \rho(u, v) \cdot a^2 \left(\sin(u)^2\right)^{1/2} du \, dv$$

$$= \int_0^{2\pi} \int_0^{\pi/2} (3u + v) \cdot 5^2 \left(\sin(u)^2\right)^{1/2} du \, dv$$

In[14]:=
```
integrand = ρ[u, v] * MagNVector[u, v]
term1 = Integrate[integrand, {u, 0, π / 2}]
totalMass = Integrate[term1, {v, 0, 2 π}]
totalMass // N
```

Out[14]= $25 \times (3u + v) \sqrt{\text{Sin}[u]^2}$

Out[15]= $25 \times (3 + v)$

Out[16]= $150 \pi + 50 \pi^2$

Out[17]= 964.719

```
In[18]:= (*Clear[totalMass,term1,integrand,a,ρ,MagNVector,NVector,rv,ru,r]*)
```

Surface Integrals of Vector Functions

For a given vector function $\overline{F}$, we define the **surface integral** over a surface S as

$$\text{surface integral} = \int\int_S \overline{F} \cdot d\overline{s} = \int\int_S \overline{F} \cdot \overline{n}\, dA$$

where $\overline{F}$ =vector field/function
$\overline{n}$ = unit normal vector
dA =elemental area

If the surface S is parameterized as we discussed previously ($S = \overline{r}(u, v)$ and u, v varies over a region R in the uv plane), then we can write this as

$$\int\int_S \overline{F} \cdot \overline{n}\, dA = \int\int_R \overline{F}(\overline{r}(u, v) \cdot \overline{N}(u, v)\, du\, dv \qquad \textbf{(Eq.3)}$$

where $\overline{N} = \overline{r}_u \times \overline{r}_v = |\overline{N}|\, \overline{n}$ (surface normal vector, <u>not</u> normalized)

It may be somewhat unexpected that Eq.3 uses $\overline{N}$ instead of $\overline{n}$. However, if we are careful and recall how u and v are defined, we can show this is correct. Recall that $\overline{N} = \overline{r}_u \times \overline{r}_v$. From the definition of the cross product, we know that $|\overline{N}| = |\overline{r}_u \times \overline{r}_v|$ is actually the areas of the parallelogram with sides $\overline{r}_u$ and $\overline{r}_v$. Hence we replace dA with $|\overline{N}|\, du\, dv$

$$\overline{n}\, dA = \overline{n}\, |\overline{N}|\, du\, dv = \overline{N}\, du\, dv \qquad \textbf{(Eq.3*)}$$

Sometimes the surface integral is referred to as the **flux integral** as will see shortly.

We can write Eq.3 in components. Recall $\overline{F} = \langle F_1, F_2, F_3 \rangle$. We also note that we can write the unit normal vector $\overline{n}$ as

$$\overline{n} = \langle \cos(\alpha), \cos(\beta), \cos(\gamma) \rangle$$

where α, β, γ are the angles between $\overline{n}$ and the coordinate axes.

With this, we can write Eq.3 as (either in the original xyz space or in the uv space as)

$$\int\int_S \overline{F} \cdot \overline{n}\, dA = \int\int_S (F_1 \cos(\alpha) + F_2 \cos(\beta) + F_3 \cos(\gamma))\, dA$$

(Eq.4)

$$\int\int_S \overline{F} \cdot \overline{n}\, dA = \int\int_R (F_1 N_1 + F_2 N_2 + F_3 N_3)\, du\, dv$$

Note that in Eq.4, we can write $\cos(\alpha)\, dA = dy\, dz$ because α is the angle between the x axis. Similarly, $\cos(\beta)\, dA = dz\, dx$ and $\cos(\gamma)\, dA = dx\, dy$. Therefore, an alternative form of Eq.4 is

$$\iint_S \overline{F} \cdot \overline{n}\, dA = \iint_S (F_1\, dy\, dz + F_2\, dz\, dx + F_3\, dx\, dy) \qquad \textbf{(Eq.5)}$$

We can break Eq.5 into 3 separate integrals

$\iint_S F_1\, dy\, dz$ = x component of vector field across plane normal to the x direction

$\iint_S F_2\, dz\, dx$ = y component of vector field across plane normal to the y direction

$\iint_S F_3\, dx\, dy$ = z component of vector field across plane normal to the z direction

So if $\overline{F}$ represents a vector field of some fluid, the surface integral computes the amount of fluid that flows across this surface.

Example 1: Flux Through a Surface

Consider a fluid (water) flowing across a parabolic surface S defined by

$y = x^2$
$x \in [0, 2]$
$z \in [0, 3]$

If the velocity of the flow is described by the vector field

$\overline{F} = \langle 3z^2 + 3y, -2x, 6xz \rangle$ (in m/s)

In[19]:=
```
F2[x_, y_, z_] = {3 z^2 + 3 y, -2 x , 6 x z};
```

Compute the flow rate of fluid across this surface.

Solution

We can first parameterize the surface. One parametrization is

$x = u$
$y = x^2 = u^2$
$z = v$

So we have

$\overline{r}(u, v) = \langle u, u^2, v \rangle$

where $u \in [0, 2]$
$v \in [0, 3]$

In[20]:=
```
r2[u_, v_] = {u, u^2, v};
ParametricPlot3D[r2[u, v], {u, 0, 2}, {v, 0, 3}, AxesLabel → {"x", "y", "z"}]
```

Out[21]=



We can now compute the surface integral

$$\int\int_S \overline{F}\cdot\overline{n}\, dA = \int\int_R \overline{F}\left(\overline{r}(u, v)\cdot\overline{N}(u, v)\, du\, dv \qquad \textbf{\textit{(Eq.3)}}$$

We now compute the normal vector using

$$\overline{N} = \overline{r}_u \times \overline{r}_v$$

In[22]:=
```
r2u[u_, v_] = D[r2[u, v], u]
r2v[u_, v_] = D[r2[u, v], v]
N2Vector[u_, v_] = Cross[r2u[u, v], r2v[u, v]] // Simplify
```

Out[22]= {1, 2 u, 0}

Out[23]= {0, 0, 1}

Out[24]= {2 u, -1, 0}

We now compute $\overline{F}(\overline{r}(u, v))$

$$\overline{F}(\overline{r}(u, v)) = <3z^2 + 3y, -2x, 6xz> \Big|_{x=u, y=u^2, z=v}$$

$$= <3v^2 + 3u^2, -2u, 6uv>$$

In[25]:=
```
F2[r2[u, v][[1]], r2[u, v][[2]], r2[u, v][[3]]]
```

Out[25]= $\{3u^2 + 3v^2, -2u, 6uv\}$

So the integrand becomes

$$\overline{F}(\overline{r}(u, v)) \cdot \overline{N}(u, v) = <3v^2 + 3u^2, -2u, 6uv> \cdot <2u, -1, 0>$$

In[26]:=
```
integrand = Dot[F2[r2[u, v][[1]], r2[u, v][[2]], r2[u, v][[3]]], N2Vector[u, v]]
```

Out[26]= $2u + 2u\left(3u^2 + 3v^2\right)$

So the final integral becomes

$$\iint_S \overline{F} \cdot \overline{n}\, dA = \int_0^3 \int_0^2 2u + 2u\left(3u^2 + 3v^2\right) du\, dv$$

Integrating yields the final result

In[27]:=
```
Integrate[
 Integrate[integrand, {u, 0, 2}],
 {v, 0, 3}]
```

Out[27]= 192

So we see that the flow is 192 m^3/s

The value is positive because for the most part, the flux of the vector field is in the same direction as the surface normal vector. To see this, let's visualize the entire system including how the normal vector, $\overline{N}(u, v)$, appears

In[28]:=
```
Manipulate[
 rGiven = r2[uGiven, vGiven];
 NGiven = N2Vector[uGiven, vGiven];

 Show[
  (*Show the surface*)
  ParametricPlot3D[r2[u, v], {u, 0, 2}, {v, 0, 3}, AxesLabel → {"x", "y", "z"}],

  (*Plot the vector field*)
  VectorPlot3D[F2[x, y, z], {x, 0, 2}, {y, 0, 4}, {z, 0, 3}],

  (*Plot the surface normal*)
  Graphics3D[
   {
    Green, Arrow[{{rGiven⟦1⟧, rGiven⟦2⟧, rGiven⟦3⟧},
       {rGiven⟦1⟧ + NGiven⟦1⟧, rGiven⟦2⟧ + NGiven⟦2⟧, rGiven⟦3⟧ + NGiven⟦3⟧}}]
   }
  ],
  PlotRange → Full
 ],
 {uGiven, 0, 2},
 {vGiven, 0, 3}
]
```

Out[28]=



In[29]:=
```
(*Clear[NGiven,rGiven,vGiven,uGiven,NVector,rv,ru,r,F,integrand]*)
```

Orientation of Surfaces

As we see, the direction of the normal vector matters. We could just as easily have chosen $-\overline{N}$ instead of $\overline{N}$ as the surface normal vector (both are normal to the surface). If we choose a direction in a consistent fashion, we say that we have an **oriented surface**. The easiest way to change the direction of the normal vector in our calculations is to simply switch u and v

Theorem 1: Change of Orientation in a Surface Integral

The replacement of $\overline{n}$ by $-\overline{n}$ (hence of $\overline{N}$ by $-\overline{N}$) corresponds to the multiplication of the integral in Eq.3 or Eq.4 by -1.

Orientation of Smooth Surfaces

A smooth surface S is called **orientable** if the positive normal direction, when given at an arbitrary point P_0 of S, can be continued in a unique and continuous fashion on the entire surface.

Orientation of Piecewise Smooth Surfaces

See text

Theory: Nonorientable Surfaces

See text

Surface Integrals Without Regard to Orientation

See text